
1.1 About NSIS
The installer is your application's first im pression. Slow or unsuccessful
software installations is one of the m ost irritating com puter problem s. A
quick and user friendly installer is therefore an essential part of your
software product. NSIS (Nullsoft Scriptable Install System ) is a tool that
allows program m ers to create such installers for W indows. It is released
under an open source license and is com pletely free for any use.

NSIS creates installers that are capable of installing, uninstalling, setting
system  settings, extracting files, etc. Because it's based on script files
you can fully control every part of your installer. The scripting language
supports variables, functions and string m anipulation, just like a norm al
program m ing language - but designed for the creation of installers. Even
with all these features, NSIS is still the sm allest installer system
available. W ith the default options, it has an overhead of only 34 KB.

W ith NSIS 3 you can also create Unicode installers, targeting all the
languages supported by the O S without display issues.

1.2 M ain Features
Sm all overhead size

NSIS has been designed to be sm all, fast and efficient. W hile other
installers often add hundreds of kilobytes or several m egabytes to your
installer data, a fully featured NSIS installer has a overhead of only 34
KB.

Com patible w ith all m ajor W indow s versions

You can create a single installer that is com patible with W indows 95, 98,
M E, NT4, 2000, XP, 2003, Vista, Server 2008, 7, Server 2008R2, 8,
Server 2012, 8.1, Server 2012R2 and W indows 10.

Unique com pression m ethods

You can choose between three different integrated com pression m ethods
(ZLib, BZip2, LZM A). LZM A com pression gives better results than any
other com m on com pression m ethod. You don't have to use large self-
extracting archive m odules or other applications, the com pression
support is included in the 34 KB overhead.

Script based

Unlike other system s that can only generate installers based on a list of
files and registry keys, NSIS has a powerful scripting language. This
script language is designed for installers and has com m ands that help
you to perform  m any installation tasks. You can easily add custom  logic
and handle different upgrades, version checks and m ore. O n the NSIS
W iki you can find a lot m ore.

M ultiple languages in one installer

O ne installer can support m ultiple interface languages. M ore than 60
translations are already included and you can also create your own
language files. RTL (right-to-left) languages such as Arabic and Hebrew
are fully supported. Creating a Unicode native installer is also possible for



even m ore supported languages.

M any features and checks for the target system

The script language provides com m ands you can use on the target
system . From  sim ple features like folder creation and registry editing to
text/binary file m odification, m odification of environm ent variables and
system  reboots. By using one of the provided plug-ins you can even call
the W indows API directly.

Custom  dialogs and interfaces

You can create custom  wizard pages to get user input or integrate
configuration options. NSIS includes a classic and m odern wizard
interface, but it is even possible to create your own custom  interface.

Plug-in system

NSIS can be extended with plug-ins that can com m unicate with the
installer. They can be written in C, C++, Delphi or another language and
can be used to perform  installation tasks or extend the installer interface.
You can use the plug-in with a single line of script code. Plug-ins are also
be com pressed like other installation data and will only be included when
you are using their features.

Support for w eb installation, file patching

The NSIS distribution includes a set of plug-ins that allow you to
download files from  the internet, m ake internet connections, patch
existing files and m ore.

Project integration, different releases and autom atic builds

The NSIS com piler features a powerful preprocessor. This allows you to
easily integrate m ultiple projects into a single installer or autom atically
generate installer builds. You can also generate different releases such
as lite and full versions.

Easy and hum an readable file form ats



The NSIS script form at and the form at used for interface dialogs are
easy, docum ented and hum anly readable, so you can edit your files with
your favorite editor. This also m akes autom atic script generation possible.

1.3 Feature List
G enerates self contained executable installers
Support for ZLIB, BZIP2 and LZM A data com pression (files can be
com pressed individually or together)
Uninstall support (installer can generate an uninstaller)
Custom izable user interface (dialogs, fonts, backgrounds, icons, text,
checkm arks, im ages etc.)
Classic and M odern wizard interface
Fully m ultilingual, support for m ultiple languages in one installer.
M ore than 60 translations are available, but you can also create your
own. Unicode support allowing even m ore languages.
Page system : You can add standard wizard pages or custom  pages
User selection of installation com ponents, tree for com ponent
selection
M ultiple install configurations (usually M inim al, Typical, Full), and
custom  configuration
Installer self-verification using a CRC32 checksum
Sm all overhead over com pressed data size (34 KB with default
options)
Ability to display a license agreem ent in text or RTF form at
Ability to detect destination directory from  the registry
Easy to use plug-in system  (lots of plug-ins for creation of custom
dialogs, internet connections, HTTP downloading, file patching,
W in32 API calls etc. are included)
Installers can be as large as 2G B
O ptional silent m ode for autom ated installations
A preprocessor with support for defined sym bols, m acro's,
conditional com pilation, standard predefines
A lovely coding experience with elem ents of PHP and assem bly
(includes user variables, a stack, real flow control, etc.)
Installers have their own VM s that let you write code that can
support:

File extraction (with configurable overwrite param eters)
File/directory copying, renam ing, deletion, searching
Plug-in DLL calling



DLL/ActiveX control registration/deregistration
Executable execution (shell execute and wait options)
Shortcut creation
Registry key reading/setting/enum erating/deleting
INI file reading/writing
G eneric text file reading/writing
Powerful string and integer m anipulation
W indow finding based on class nam e or title
User interface m anipulation (font/text setting)
W indow m essage sending
User interaction with m essage boxes or custom  pages
Branching, com parisons, etc.
Error checking
Reboot support, including delete or renam e on reboot
Installer behaviour com m ands (such as show/hide/wait/etc)
User functions in script
Callback functions for user actions

Com pletely free for any use. See license.

1.4 Unicode installers
Starting with NSIS v3.0 you can choose to create Unicode installers by
setting the Unicode attribute. These installers will not work on W indows
95/98/M E but they will allow you to display your installer in any Unicode
language supported by the O S.

W hen building a Unicode installer NSIS variables can hold Unicode
characters (0001-FFFF). There should be no need to m odify your existing
scripts. If you want to read/write Unicode files, specific instructions have
been added to read/write UTF-16LE strings from /to disk.

2.1 Introduction
M ost software packages you download or buy com e with an installer. The
installer copies and/or updates files, writes registry keys, writes
configuration, creates shortcuts, etc. All of this is done autom atically for
the user. All the user needs to do is supply som e inform ation and the
installer will do the rest. The user goes through a wizard, m akes the
appropriate choices and waits until the installer finishes. After the installer
has finished the user is left only with the sim ple task of starting the
program . The user doesn't have to worry about things he m ight have
forgotten because all of the necessary steps were done by the installer.

NSIS is a tool for developers to create such installers. NSIS allows you to
create everything from  basic installers that just copies files to very
com plex installers that handle a lot of advanced tasks such as writing
registry keys, settings environm ent variables, downloading the latest files
from  the internet, custom izing configuration files and m ore. NSIS is very
flexible and its scripting language is easy to learn.

NSIS com piles all of the files and the installation script into one
executable file so your application will be easy to distribute. NSIS adds
only about 34KB of code of its own (for the default configuration) to the
data. NSIS boasts the sm allest overhead available while still providing a
lot of options thanks to its powerful scripting language and support of
external plug-ins.

2.2 Script Files
To create a NSIS installer you first have to write a NSIS script. A NSIS
script is just a regular text file with a special syntax. You can edit scripts
with any text editor. It's recom m ended to use a text editor that shows line
num bers because NSIS uses line num bers to indicate where errors lie,
and to warn you about where errors m ight lie. An editor that supports
syntax highlighting is also recom m ended. You can download editors
m ade especially for NSIS and files for syntax highlighting from  the NSIS
W iki.

In a NSIS script every line is treated as a com m and. If your com m and is
too long for one line you can use a back-slash - '\' - at the end of the line.
The com piler will treat the new line as an addition to the previous line and
will not expect a new com m and. For exam ple:

Messagebox MB_OK|MB_ICONINFORMATION \
"This is a sample that shows how to use line breaks for larger commands in NSIS scripts"

If you want to use a double-quote in a string you can either use $\" to
escape the quote or quote the string with a different type of quote such
as ̀ or '.

For m ore details about the script form at, see Script File Form at.

The default extension for a script file is .nsi. Header files have the .nsh
extension. Header files can help you arrange your script by dividing it to
m ore than one block of code, you can also put functions or m acros in
header files and include the header files in m ultiple installers. This m akes
updating easier and it also m akes your scripts easier to read. To include
a header file in your script use !include. Header files that reside in the
Include directory under your NSIS directory can be included just by their
nam e. For exam ple:

!include Sections.nsh

2.3 Scripting structure
A NSIS script contains Installer Attributes, Pages and Sections/Functions.
You can also use Com piler Com m ands for com pile-tim e operations. The
O utFile instruction is required and tells NSIS where to write the installer,
you also need at least one section.

2.3.1 Installer Attributes

Installer Attributes determ ine the behavior and the look and feel of your
installer. W ith these attributes you can change texts that will be shown
during the installation, the num ber of installation types etc. M ost of these
com m ands can only be set and are not changeable during runtim e.

O ther basic instructions are Nam e and InstallDir.

For m ore inform ation about installer attributes, have a look at Installer
Attributes.

2.3.2 Pages

A non-silent installer has a set of wizard pages to let the user configure
the installer. You can set which pages to display using the Page
com m and (or PageEx for m ore advanced settings). A typical set of pages
looks like this:

Page license
Page components
Page directory
Page instfiles
UninstPage uninstConfirm
UninstPage instfiles

For the installer, this typical set of pages will display a license agreem ent,
allow selection of com ponents to install, allow selection of an installation
directory, and finally install the selected com ponents in the instfiles page.
For the uninstaller, it will display a confirm ation page, and uninstall in the



instfiles page.

2.3.3 Sections

It's com m on for installers to have several things the user can install. For
exam ple in the NSIS distribution installer you can choose to install
additional tools, plug-ins, exam ples and m ore. Each of these com ponents
has its own piece of code. If the user selects to install this com ponent
then the installer will execute that code. In the script, that code is defined
in sections. Each section corresponds to one com ponent on the
com ponents page. The section's nam e is the displayed com ponent nam e
and the section code will be executed if that com ponent is selected. It is
possible to build your installer with only one section but if you want to use
the com ponents page and let the user choose what to install, you'll have
to use m ore than one section.

Uninstallers can also have m ultiple sections. Uninstaller section nam es
are prefixed with 'un.'. For exam ple:

Section "Installer Section"
SectionEnd

Section "un.Uninstaller Section"
SectionEnd

The instructions that can be used in sections are very different from  the
installer attributes instructions, they are executed at runtim e on the user's
com puter. Those instructions can extract files, read from  and write to the
registry, INI files or norm al files, create directories, create shortcuts and a
lot m ore. You can find out m ore in Instructions.

The m ost basic instructions are SetO utPath which tells the installer
where to extract files and File which extracts files.

Exam ple:

Section "My Program"
  SetOutPath $INSTDIR
  File "My Program.exe"



  File "Readme.txt"
SectionEnd

For m ore inform ation about sections see Sections.

2.3.4 Functions

Functions can contain script code, just like sections. The difference
between sections and functions is the way they are called. There are two
types of functions, user functions and callback functions.

User functions are called by the user from  within sections or other
functions using the Call instruction. User functions will not execute unless
you call them . After the code in the function has executed the installer will
continue executing the instructions that cam e after the Call instruction,
unless you have aborted the installation inside the function. User
functions are very useful if you have a set of instructions that need to be
executed at several locations in the installers. If you put the code into a
function you can save the copying tim e and you can m aintain the code
m ore easily.

Callback functions are called by the installer upon certain defined events
such as when the installer starts. Callbacks are optional. If for exam ple
you want to welcom e the user to your installer you can define a function
called .onInit. The NSIS com piler will recognize this function as a callback
function by the nam e and will call it when the installer starts.

Function .onInit
  MessageBox MB_YESNO "This will install My Program. Do you wish to continue?" IDYES gogogo
    Abort
  gogogo:
FunctionEnd

Abort has a special m eaning in callback functions. Each callback function
has its own m eaning for it, have a look at Callback Functions for m ore
inform ation. In the above exam ple Abort tells the installer to stop
initializing the installer and quit im m ediately.

For m ore inform ation about functions see Functions.



2.3.5 W orking w ith Scripts

2.3.5.1 Logical Code Structures

Conditionally executing code, or executing code in a loop can be done
using StrCm p, IntCm p, IfErrors, G oto and m ore. However, there's a m uch
easier way do this. The LogicLib provides som e very sim ple m acros that
allow easy construction of com plex logical structures. Its syntax,
explained in LogicLib.nsh, is sim ilar to other program m ing languages and
can prove to be sim pler for beginners and advanced users alike.

For exam ple, checking a value of a variable without the LogicLib can be
done as follows.

StrCmp $0 'some value' 0 +3
  MessageBox MB_OK '$$0 is some value'
  Goto done
StrCmp $0 'some other value' 0 +3
  MessageBox MB_OK '$$0 is some other value'
  Goto done
# else
  MessageBox MB_OK '$$0 is "$0"'
done:

However, with the LogicLib the code is m uch m ore readable and easy to
understand, as can be seen in the following exam ple.

${If} $0 == 'some value'
  MessageBox MB_OK '$$0 is some value'
${ElseIf} $0 == 'some other value'
  MessageBox MB_OK '$$0 is some other value'
${Else}
  MessageBox MB_OK '$$0 is "$0"'
${EndIf}

The sam e can also be done using a switch, as shown in the following
exam ple.



${Switch} $0
  ${Case} 'some value'
    MessageBox MB_OK '$$0 is some value'
    ${Break}
  ${Case} 'some other value'
    MessageBox MB_OK '$$0 is some other value'
    ${Break}
  ${Default}
    MessageBox MB_OK '$$0 is "$0"'
    ${Break}
${EndSwitch}

M ultiple conditions are also supported. The following exam ple will notify
the user, if both $0 and $1 are em pty.

${If} $0 == ''
${AndIf} $1 == ''
  MessageBox MB_OK|MB_ICONSTOP 'both are empty!'
${EndIf}

The LogicLib rem oves the need for labels and relative jum ps, thus
prevents label nam e conflicts, and rem oves the need to m anually adjust
relative jum p offsets every tim e the script is changed.

It also sim plifies looping by supporting the com m on while, do and for
loops. All of the following exam ples count to five using the LogicLib.

StrCpy $R1 0
${While} $R1 < 5
  IntOp $R1 $R1 + 1
  DetailPrint $R1
${EndWhile}

${For} $R1 1 5
  DetailPrint $R1
${Next}

StrCpy $R1 0



${Do}
  IntOp $R1 $R1 + 1
  DetailPrint $R1
${LoopUntil} $R1 >= 5

To use the LogicLib the following line needs to be added near the top of
the script.

!include LogicLib.nsh

M ore exam ples can be found in LogicLib.nsi.

2.3.5.2 Variables

You can declare your own variables ($VARNAM E) with the Var
com m and. Variables are global and can be used in any Section or
Function.

Declaring and using a user variable:

Var BLA ;Declare the variable

Section bla

  StrCpy $BLA "123" ;Now you can use the variable $BLA

SectionEnd

In addition there is a stack, which can also be used for tem porary
storage. To access the stack use the com m ands Push and Pop. Push
adds a value to the stack, Pop rem oves one and sets the variable.

For shared code, there are 20 registers available (like $0 and $R0).
These static variables don't have to be declared and you won't get any
nam e conflicts. If you want to use these variables in shared code, store
the original values on the stack and restore the original values
afterwards.



After calling the function, the variables contain the sam e value as before.
Note the order when using m ultiple variables (last-in first-out):

Function bla

  Push $R0
  Push $R1

    ...code...

  Pop $R1
  Pop $R0

FunctionEnd

2.3.5.3 Debugging Scripts

The m ore you work with NSIS the m ore com plex the scripts will becom e.
This will increase the potential of m istakes, especially when dealing with
lots of variables. There are a few possibilities to help you debugging the
code. To display the contents of variables you should use M essageBoxes
or DetailPrint. To get a brief overview about all variables you should use
the plug-in Dum pState. By default all actions of the Installer are printed
out in the Log W indow. You can access the log if you right-click in the Log
W indow and select "Copy Details To Clipboard". There is also a way to
write it directly to a file, see here.

2.3.6 Script Execution

W hen a user runs an installer or uninstaller, pages are displayed in the
order they were defined in the script. W hen the instfiles page is reached,
sections, corresponding to the selected com ponents, are executed in the
order they were defined in the script. If the com ponents page is not
displayed, all sections are executed, assum ing they were not unselected
or som ehow disabled by the script.

Beside code in sections, there's also code in callback functions. If



defined, they m ight be executed before the sections code. For exam ple,
the .onInit callback function is executed before anything else in the script.
There are also page callback functions which are executed at certain
points of the page display process.

2.3.7 Com piler Com m ands

Com piler com m ands will be executed at com pile tim e on your com puter.
They can be used for conditional com pilation, to include header files, to
execute applications, to change the working directory and m ore. The
m ost com m on usage is defines. Defines are com pile tim e constants. You
can define your product's version num ber and use it in your script. For
exam ple:

!define VERSION "1.0.3"
Name "My Program ${VERSION}"
OutFile "My Program Installer - ${VERSION}.exe"

For m ore inform ation about defines see Conditional Com pilation.

Another com m on use is m acros. M acros are used to insert code at
com pile tim e, depending on defines and using the values of the defines.
The m acro's com m ands are inserted at com pile tim e. This allows you to
write a general code only once and use it a lot of tim es but with a few
changes. For exam ple:

!macro MyFunc UN
Function ${UN}MyFunc
  Call ${UN}DoRegStuff
  ReadRegStr $0 HKLM Software\MyProgram key
  DetailPrint $0
FunctionEnd
!macroend

!insertmacro MyFunc ""
!insertmacro MyFunc "un."

This m acro helps you avoid writing the sam e code for both the installer



and the uninstaller. The two !insertm acros insert two functions, one for
the installer called M yFunc and one for the uninstaller called un.M yFunc
and both do exactly the sam e thing.

For m ore inform ation see Com pile Tim e Com m ands.

2.4 Com piler
The second thing you need to do in order to create your installer after you
have created your script is to com pile your script. M akeNSIS.exe is the
NSIS com piler. It reads your script, parses it and creates an installer for
you.

To com pile you can right-click your .nsi file and select Com pile NSIS
Script. This will cause M akeNSISW , the NSIS Com piler Interface, to
launch and call M akeNSIS to com pile your script. M akeNSISW  receives
the output of M akeNSIS and presents it to you in a window where you
can see it, copy it, test the installer and m ore. Using m akensis.exe from
the com m and prom pt is also possible.

The com piler will check your script and give you warnings or an error. If
an error occurs (i.e. 2 param eters required but only 1 given) the com piler
will abort and a short error m essage including the line num ber will be
displayed. For non-critical errors the com piler will give a warning (i.e. two
DirText com m ands in one script). If your script has no errors the com piler
will output an installer for you to distribute.

NSIS supports different com pression m ethods, as explained here. ZLIB is
the default com pression m ethod, which is fast and uses only a little bit of
m em ory. LZM A is a good m ethod for the creation of sm all installers for
internet distribution. BZIP2 usually com presses better than ZLIB but not
as good as LZM A, it is useful if you need lower m em ory usage or fast
script com pilation.

It is also possible to com pile W indows installers on Linux, BSD or M ac
O S X servers. See Building NSIS for details.

2.5 M odern UI
A popular user interface for NSIS is the M odern User Interface. It has an
interface like the wizards of recent W indows versions. The M odern UI is
not only a custom ized resource file, it has a lots of new interface
elem ents. It features a white header to describe the current step, a
description area on the com ponent page, a welcom e page, a finish page
that allows the user to run the application or reboot the system  and m ore.

For m ore inform ation, see the M odern UI 2 Readm e and the M odern UI
Exam ples.

2.6 Plug-ins
NSIS support plug-ins that can be called from  the script. Plug-ins are DLL
files written in C, C++, Delphi or another program m ing language and
therefore provide a m ore powerful code base to NSIS.

A plug-in call looks like this:

DLLName::FunctionName "parameter number 1" "parameter number 2" "parameter number 3"

Every plug-in's function has its own requirem ents when it com es to
param eters, som e will require none, som e will accept as m any
param eters as you want to send. Exam ples:

nsExec::ExecToLog '"${NSISDIR}\makensis.exe" /CMDHELP'
Pop $0 ; Process exit code or "error"
InstallOptions::dialog "$PLUGINSDIR\test.ini"
Pop $0 ; success/back/cancel/error
NSISdl::download http://download.nullsoft.com/winamp/client/winamp291_lite.exe $R0
Pop $0 ; "success" or a error code

The plug-ins that NSIS knows of are listed at the top of the com piler
output (verbose level 4). NSIS searches for plug-ins in the Plugins folder
under your NSIS directory and lists all of their available functions. You
can use !addplugindir to tell NSIS to search in other directories too.

The NSIS distribution already includes m any plug-ins. InstallO ptions is a
popular plug-in that allows you to create custom  pages, in com bination
with the NSIS Page com m ands (See Pages). The Startm enu plug-in
provides a page that allows the user to choose a Start M enu folder. There
are a lot of plug-ins for different purposes, take a look in the Docs folder
for help files and exam ples. You can find additional plug-ins online: NSIS
W iki.

You can also create a plug-in yourself. C/C++ and Delphi header files are
already available, see the exam ple plugin for how to do this. Source code
of included plug-ins can also be found in the source code package.

2.7 M ore
This tutorial has described the basic NSIS features, to learn m ore about
everything NSIS can do, take som e tim e to read the rest of this m anual.

3.1 M akeNSIS Usage
NSIS installers are generated by using the 'M akeNSIS' program  to
com pile a NSIS script (.NSI) into an installer executable. The NSIS
developm ent kit installer sets up your com puter so that you can com pile a
.nsi file by sim ply right-clicking on it in Explorer and selecting 'com pile'.

If you want to use M akeNSIS on the com m and line, the syntax of
m akensis is:

makensis [ option | script.nsi | - ] [...]

3.1.1 O ptions

/LICENSE displays license inform ation.
The /V switch followed by a num ber between 0 and 4 will set the
verbosity of output accordingly. 0=no output, 1=errors only,
2=warnings and errors, 3=info, warnings, and errors, 4=all output.
The /P switch followed by a num ber between 0 and 5 will set the
priority of the com piler process accordingly. 0=idle, 1=below norm al,
2=norm al (default), 3=above norm al, 4=high, 5=realtim e.
The /O  switch followed by a filenam e tells the com piler to print its log
to that file (instead of the screen)
/PAUSE m akes m akensis pause before quitting, which is useful
when executing directly from  W indows.
/NO CO NFIG  disables inclusion of nsisconf.nsh. W ithout this
param eter, installer defaults are set from  nsisconf.nsh.
/CM DHELP prints basic usage inform ation for com m and (if
specified), or all com m ands (if com m and is not specified).
/HDRINFO  prints inform ation about which options were used to
com pile m akensis.
/NO CD disables the current directory change to that of the .nsi file
/INPUTCHARSET allows you to specify a specific codepage for files
without a BO M . (ACP|O EM |CP#|UTF8|UTF16<LE|BE>)
/O UTPUTCHARSET allows you to specify the codepage used by
stdout when the output is redirected.
(ACP|O EM |CP#|UTF8[SIG ]|UTF16<LE|BE>[BO M ])



/PPO  or /SAFEPPO  will only run the preprocessor and print the
result to stdout. The safe version will not execute instructions like
!appendfile or !system . !packhdr and !finalize are never executed.
/W X treats warnings as errors
Using the /D switch one or m ore tim es will add to sym bols to the
globally defined list (See !define).
Using the /X switch one or m ore tim es will execute the code you
specify following it. Exam ple: "/XAutoCloseW indow false"
Specifying a dash (-) for the script nam e will tell m akensis to use the
standard input as a source.

3.1.2 Notes

Param eters are processed in order. makensis /Ddef script.nsi is
not the sam e as makensis script.nsi /Ddef.
If m ultiple scripts are specified, they are treated as one concatenated
script.

3.1.3 Environm ent variables

m akensis checks a num ber of environm ent variables that tell it where to
locate the things it needs in order to create installers. These variables
include:

NSISDIR, NSISCO NFDIR - Places where nsis data and config files
are installed. NSISDIR alters the script variable ${NSISDIR}. See
section 4.2.3 for m ore info.
APPDATA (on W indows) or HO M E (on other platform s) - Location of
the per-user configuration file.

3.1.4 Exam ples

Basic usage:

makensis.exe myscript.nsi

Q uiet m ode:

makensis.exe /V1 myscript.nsi



Force com pressor:

makensis.exe /X"SetCompressor /FINAL lzma" myscript.nsi

Change script behavior:

makensis.exe /DUSE_UPX /DVERSION=1.337 /DNO_IMAGES myscript.nsi

Param eters order:

makensis /XSection sectioncontents.nsi /XSectionEnd

3.2 Installer Usage
G enerated installers and uninstallers accept a few options on the
com m and line. These options give the user a bit m ore control over the
installation process.

3.2.1 Com m on O ptions

/NCRC disables the CRC check, unless CRCCheck force was used in
the script.
/S runs the installer or uninstaller silently. See section 4.12 for m ore
inform ation.
/D sets the default installation directory ($INSTDIR), overriding
InstallDir and InstallDirRegKey. It m ust be the last param eter used in
the com m and line and m ust not contain any quotes, even if the path
contains spaces. O nly absolute paths are supported.

3.2.2 Uninstaller Specific O ptions

_?= sets $INSTDIR. It also stops the uninstaller from  copying itself to
the tem porary directory and running from  there. It can be used along
with ExecW ait to wait for the uninstaller to finish. It m ust be the last
param eter used in the com m and line and m ust not contain any
quotes, even if the path contains spaces.

3.2.3 Exam ples

installer.exe /NCRC

installer.exe /S

installer.exe /D=C:\Program Files\NSIS

installer.exe /NCRC /S /D=C:\Program Files\NSIS

uninstaller.exe /S _?=C:\Program Files\NSIS



# uninstall old version
ExecWait '"$INSTDIR\uninstaller.exe" /S _?=$INSTDIR'

4.1 Script File Form at
A NSIS Script File (.nsi) is just a text file with script code.

Com m ands

Com m ands lines are in the form at 'com m and [param eters]'

File "myfile"

Com m ents

Lines beginning with ; or # are com m ents. You can put com m ents after
com m ands. You can also use C-style com m ents to com m ent one or m ore
lines.

; Comment
# Comment

# Comment \
    Another comment line (see `Long commands` section below)

/*
Comment
Comment
*/

Name /* comment */ mysetup

File "myfile" ; Comment

If you want a param eter to start with ; or # put it in quotes.

Plug-ins

To call a plug-in, use 'plugin::com m and [param eters]'. For m ore info see
Plug-in DLLs.



nsExec::Exec "myfile"

Num bers

For param eters that are treated as num bers, use decim al (the num ber) or
hexadecim al (with 0x prepended to it, i.e. 0x12345AB), or octal (num bers
beginning with a 0 and no x).

Colors should be set in hexadecim al RG B form at, like HTM L but without
the #.

IntCmp 1 0x1 lbl_equal

SetCtlColors $HWND CCCCCC

Strings

To represent strings that have spaces, use quotes:

MessageBox MB_OK "Hi there!"

Q uotes only have the property of containing a param eter if they surround
the rest of the param eter. They can be either single quotes, double
quotes, or the backward single quote.

You can escape quotes using $\:

MessageBox MB_OK "I'll be happy" ; this one puts a ' inside a string
MessageBox MB_OK 'And he said to me "Hi there!"' ; this one puts a " inside a string
MessageBox MB_OK `And he said to me "I'll be happy!"` ; this one puts both ' and "s inside a string
MessageBox MB_OK "$\"A quote from a wise man$\" said the wise man" ; this one shows escaping of quotes

It is also possible to put newlines, tabs etc. in a string using $\r, $\n, $\t
etc. M ore inform ation...

Variables

Variables start with $. User variables m ust be declared.

Var MYVAR



StrCpy $MYVAR "myvalue"

M ore inform ation...

Long com m ands

To extend a com m and over m ultiple lines, use a backslash (\) at the end
of the line. The next line will effectively be concatenated to the end of it.
For exam ple:

CreateShortcut "$SMPROGRAMS\NSIS\ZIP2EXE project workspace.lnk" \
    "$INSTDIR\source\zip2exe\zip2exe.dsw"

MessageBox MB_YESNO|MB_ICONQUESTION \
    "Do you want to remove all files in the folder? \
    (If you have anything you created that you want \
     to keep, click No)" \
    IDNO NoRemoveLabel

Line extension for long com m ands works for com m ents as well. It can be
a bit confusing, so it should be avoided.

# A comment \
    still a comment here...

Configuration file

If a file nam ed "nsisconf.nsh" in the config directory exists, it will be
included by default before any scripts (unless the /NO CO NFIG  com m and
line param eter is used). The config directory on W indows is the sam e
directory as m akensis.exe is in. O n other platform s this is set at install
tim e and defaults to $PREFIX/etc/. You can alter this at runtim e, see
section 3.1.3 for m ore inform ation.

4.2 Variables
All variables are global and can be used in Sections or Functions. Note
that by default, variables are lim ited to 1024 characters. To extend this
lim it, build NSIS with a bigger value of the NSIS_M AX_STRLEN build
setting or use the special build.

4.2.1 User Variables

$VARNAM E

User variables m ust be declared with the Var com m and. You can use
these variables to store values, work with string m anipulation etc.

4.2.1.1 Var

[/GLOBAL] var_name

Declare a user variable. Allowed characters for variables nam es: [a-z][A-
Z][0-9] and '_'. All defined variables are global, even if defined in a
section or a function. To m ake this clear, variables defined in a section or
a function m ust use the /G LO BAL flag. The /G LO BAL flag is not required
outside of sections and functions.

Var example

Function testVar
  Var /GLOBAL example2

  StrCpy $example "example value"
  StrCpy $example2 "another example value"
FunctionEnd

4.2.2 O ther W ritable Variables

$0, $1, $2, $3, $4, $5, $6, $7, $8, $9, $R0, $R1, $R2, $R3, $R4, $R5,



$R6, $R7, $R8, $R9

Registers. These variables can be used just like user variables, but are
usually used in shared functions or m acros. You don't have to declare
these variables so you won't get any nam e conflicts when using them  in
shared code. W hen using these variables in shared code it's
recom m ended that you use the stack to save and restore their original
values. These variables can also be used for com m unication with plug-
ins because they can be read and written by the plug-in DLLs.

$INSTDIR

The installation directory ($INSTDIR is m odifiable using StrCpy,
ReadRegStr, ReadINIStr, etc. - This could be used, for exam ple, in the
.onInit function to do a m ore advanced detection of install location).

Note that in uninstaller code, $INSTDIR contains the directory where the
uninstaller lies. It does not necessarily contain the sam e value it
contained in the installer. For exam ple, if you write the uninstaller to
$W INDIR and the user doesn't m ove it, $INSTDIR will be $W INDIR in the
uninstaller. If you write the uninstaller to another location, you should
keep the installer's $INSTDIR in the registry or an alternative storing
facility and read it in the uninstaller.

$O UTDIR

The current output directory (set im plicitly via SetO utPath or explicitly via
StrCpy, ReadRegStr, ReadINIStr, etc)

$CM DLINE

The com m and line of the installer. The form at of the com m and line can
be one of the following:

"full\path to\installer.exe" PARAM ETER PARAM ETER PARAM ETER
installer.exe PARAM ETER PARAM ETER PARAM ETER
For parsing out the PARAM ETER portion, see G etParam eters. If /D=
is specified on the com m and line (to override the install directory) it
won't show up in $CM DLINE.



$LANG UAG E

The identifier of the language that is currently used. For exam ple, English
is 1033. You can only change this variable in .onInit.

4.2.3 Constants

Constants can also be used in the InstallDir attribute.

Note that som e of the new constants will not work on every O S. For
exam ple, $CDBURN_AREA will only work on W indows XP and above. If
it's used on W indows 98, it'll be em pty. Unless m entioned otherwise, a
constant should be available on every O S.

$PRO G RAM FILES, $PRO G RAM FILES32, $PRO G RAM FILES64

The program  files directory (usually C:\Program Files but detected at
runtim e). O n W indows x64, $PRO G RAM FILES and
$PRO G RAM FILES32 point to C:\Program Files (x86) while
$PRO G RAM FILES64 points to C:\Program Files. Use
$PRO G RAM FILES64 when installing x64 applications.

$CO M M O NFILES, $CO M M O NFILES32, $CO M M O NFILES64

The com m on files directory. This is a directory for com ponents that are
shared across applications (usually C:\Program Files\Common Files but
detected at runtim e). O n W indows x64, $CO M M O NFILES and
$CO M M O NFILES32 point to C:\Program Files (x86)\Common Files while
$CO M M O NFILES64 points to C:\Program Files\Common Files. Use
$CO M M O NFILES64 when installing x64 applications.

$DESKTO P

The W indows desktop directory. The context of this constant (All Users or
Current user) depends on the SetShellVarContext setting. The default is
the current user.

$EXEDIR

The directory containing the installer executable (technically this is a



variable and you can m odify it, but it is probably not a good idea).

$EXEFILE

The base nam e of the installer executable.

$EXEPATH

The full path of the installer executable.

${NSISDIR}

A sym bol that contains the path where NSIS is installed. Useful if you
want to reference resources that are in NSIS directory e.g. Icons, UIs etc.

W hen com piled with support for keeping m akensis and the data in the
sam e place (the default on W indows), it is in the sam e place as
m akensis, on other platform s it is set at com pile tim e (See the INSTALL
file for info). In both instances you can m odify it at runtim e by setting the
NSISDIR environm ent variable. See section 3.1.3 for m ore info.

$W INDIR

The W indows directory (usually C:\Windows or C:\WinNT but detected at
runtim e).

$SYSDIR

The W indows system  directory (usually C:\Windows\System or
C:\WinNT\System32 but detected at runtim e).

$TEM P

The tem porary directory.

$STARTM ENU

The start m enu folder (useful for adding start m enu item s using
CreateShortcut). The context of this constant (All Users or Current user)
depends on the SetShellVarContext setting. The default is the current
user.



$SM PRO G RAM S

The start m enu program s folder (use this whenever you want
$STARTM ENU\Program s). The context of this constant (All Users or
Current user) depends on the SetShellVarContext setting. The default is
the current user.

$SM STARTUP

The start m enu program s / startup folder. The context of this constant (All
Users or Current user) depends on the SetShellVarContext setting. The
default is the current user.

$Q UICKLAUNCH

The quick launch folder for IE4 active desktop and above. If quick launch
is not available it sim ply returns the sam e as $TEM P.

$DO CUM ENTS

The docum ents directory. A typical path for the current user is
C:\Documents and Settings\Foo\My Documents. The context of this
constant (All Users or Current user) depends on the SetShellVarContext
setting. The default is the current user.

This constant is not available on W indows 95 unless Internet Explorer 4
is installed.

$SENDTO

The directory that contains Send To m enu shortcut item s.

$RECENT

The directory that contains shortcuts to the user's recently used
docum ents.

$FAVO RITES

The directory that contains shortcuts to the user's favorite websites,



docum ents, etc. The context of this constant (All Users or Current user)
depends on the SetShellVarContext setting. The default is the current
user.

This constant is not available on W indows 95 unless Internet Explorer 4
is installed.

$M USIC

The user's m usic files directory. The context of this constant (All Users or
Current user) depends on the SetShellVarContext setting. The default is
the current user.

This constant is available on W indows XP, M E and above.

$PICTURES

The user's picture files directory. The context of this constant (All Users
or Current user) depends on the SetShellVarContext setting. The default
is the current user.

This constant is available on W indows 2000, XP, M E and above.

$VIDEO S

The user's video files directory. The context of this constant (All Users or
Current user) depends on the SetShellVarContext setting. The default is
the current user.

This constant is available on W indows XP, M E and above.

$NETHO O D

The directory that contains link objects that m ay exist in the M y Network
Places/Network Neighborhood folder.

This constant is not available on W indows 95 unless Internet Explorer 4
with Active Desktop is installed.

$FO NTS



The system 's fonts directory.

$TEM PLATES

The docum ent tem plates directory. The context of this constant (All Users
or Current user) depends on the SetShellVarContext setting. The default
is the current user.

$APPDATA

The application data directory. Detection of the current user path requires
Internet Explorer 4 and above. Detection of the all users path requires
Internet Explorer 5 and above. The context of this constant (All Users or
Current user) depends on the SetShellVarContext setting. The default is
the current user.

This constant is not available on W indows 95 unless Internet Explorer 4
with Active Desktop is installed.

$LO CALAPPDATA

The local (non-roam ing) application data directory. The context of this
constant (All Users or Current user) depends on the SetShellVarContext
setting. The default is the current user.

This constant is available on W indows M E, 2000 and above.

$PRINTHO O D

The directory that contains link objects that m ay exist in the Printers
folder.

This constant is not available on W indows 95 and W indows 98.

$INTERNET_CACHE

Internet Explorer's tem porary internet files directory.

This constant is not available on W indows 95 nor W indows NT 4 unless
Internet Explorer 4 with Active Desktop is installed.



$CO O KIES

Internet Explorer's cookies directory.

This constant is not available on W indows 95 nor W indows NT 4 unless
Internet Explorer 4 with Active Desktop is installed.

$HISTO RY

Internet Explorer's history directory.

This constant is not available on W indows 95 nor W indows NT 4 unless
Internet Explorer 4 with Active Desktop is installed.

$PRO FILE

The user's profile directory. A typical path is C:\Documents and
Settings\Foo.

This constant is available on W indows 2000 and above.

$ADM INTO O LS

A directory where adm inistrative tools are kept. The context of this
constant (All Users or Current user) depends on the SetShellVarContext
setting. The default is the current user.

This constant is available on W indows 2000, M E and above.

$RESO URCES

The resources directory that stores them es and other W indows resources
(usually $WINDIR\Resources but detected at runtim e).

This constant is available on W indows XP and above.

$RESO URCES_LO CALIZED

The localized resources directory that stores them es and other W indows
resources (usually $WINDIR\Resources\1033 but detected at runtim e).



This constant is available on W indows XP and above.

$CDBURN_AREA

A directory where files awaiting to be burned to CD are stored.

This constant is available on W indows XP and above.

$HW NDPARENT

HW ND of the m ain window (in decim al).

$PLUG INSDIR

The path to a tem porary folder created upon the first usage of a plug-in or
a call to InitPluginsDir. This folder is autom atically deleted when the
installer exits. This m akes this folder the ideal folder to hold INI files for
InstallO ptions, bitm aps for the splash plug-in, or any other file that a plug-
in needs to work.

4.2.4 Constants Used in Strings

$$

Use to represent $.

$\r

Use to represent a carriage return (\r).

$\n

Use to represent a newline (\n).

$\t

Use to represent a tab (\t).

4.3 Labels
Labels are the targets of G oto instructions and the various branching
instructions (such as IfErrors, M essageBox, IfFileExists, and StrCm p).
Labels m ust be within a Section or a Function. Labels are local in scope,
m eaning they are only accessible from  within the Section or Function that
they reside in. To declare a label, sim ply use:

M yLabel:

Labels cannot begin with a -, +, !, $, or 0-9. W hen specifying labels for
the various instructions that require them , rem em ber that both an em pty
string ("") and 0 both represent the next instruction (m eaning no G oto will
occur). Labels beginning with a period (.) are global, m eaning you can
jum p to them  from  any function or section (though you cannot jum p to an
uninstall global label from  the installer, and vice versa).

4.4 Relative Jum ps
Unlike labels, relative jum ps are, as the nam e suggests, relative to the
place they are called from . You can use relative jum ps wherever you can
use labels. Relative jum ps are m arked by num bers. +1 jum ps to the next
instruction (the default advancem ent), +2 will skip one instruction and go
to the second instruction from  the current instruction, -2 will jum p two
instructions backward, and +10 will skip 9 instructions, jum ping to the
tenth instruction from  the current instruction.

A instruction is every com m and that is executed at run-tim e, when the
installer is running. M essageBox, G oto, G etDLLVersion, FileRead,
SetShellVarContext are all instructions. AddSize, Section, SectionG roup,
SectionEnd, SetO verwrite (and everything under Com piler Flags), Nam e,
SetFont, LangString, are not instructions because they are executed at
com pile tim e.

Exam ples:

 Goto +2
   MessageBox MB_OK "You will never ever see this message box"
 MessageBox MB_OK "The last message was skipped, this one should be shown"

 Goto +4
 MessageBox MB_OK "The following message will be skipped"
 Goto +3
 MessageBox MB_OK "You will never ever see this message box"
 Goto -3
 MessageBox MB_OK "Done"

Note that m acro insertion is not considered as one instruction when it
com es to relative jum ps. The m acro is expanded before relative jum ps
are applied, and so relative jum ps can jum p into code inside an inserted
m acro. The following code, for exam ple, will not skip the m acro. It will
show a m essage box.

!macro relative_jump_test



  MessageBox MB_OK "first macro line"
  MessageBox MB_OK "second macro line"
!macroend

Goto +2
!insertmacro relative_jump_test

Previous | Contents | Next



Previous | Contents | Next



4.5 Pages
Each (non-silent) NSIS installer has a set of pages. Each page can be a
NSIS built-in page or a custom  page created by a user's function (with
nsDialogs or InstallO ptions for exam ple).

The script controls the page order, appearance, and behavior. You can
skip pages, paint them  white, force the user to stay in a certain page until
a certain condition is m et, show a readm e page, show custom  designed
pages for input and m ore. In this section you will learn how to do all of the
above.

There are two basic com m ands regarding pages, Page and UninstPage.
The first adds a page to the installer, the second adds a page to the
uninstaller. O n top of those two there is the PageEx com m and which
allows you to add a page to either one and with greater am ount of
options. PageEx allows you to set options to the specific page you are
adding instead of using the default that's set outside of PageEx.

4.5.1 O rdering

The page order is set sim ply by the order Page, UninstPage and PageEx
appear in the script. For exam ple:

 Page license
 Page components
 Page directory
 Page instfiles
 UninstPage uninstConfirm
 UninstPage instfiles

This code will tell NSIS to first show the license page, then the
com ponents selection page, then the directory selection page and finally
the install log where sections are executed. The uninstaller will first show
the uninstall confirm ation page and then the uninstallation log.

You can specify the sam e page type m ore than once.



For backwards com patibility with old NSIS scripts, the following installer
pages will be added if no installer page com m ands are used: license (if
LicenseText and LicenseData were specified), com ponents (if
Com ponentText was specified and there is m ore than one visible
section), directory (if DirText was specified) and instfiles. W hen there are
no uninstaller page com m ands the following uninstaller pages will be
added: uninstall confirm ation page (if UninstallText was specified) and
instfiles. This m ethod is deprecated, converting scripts to use page
com m ands is highly recom m ended because you can use the new
standard language strings.

4.5.2 Page O ptions

Each page has its unique set of data that defines how it will look and act.
This section describes what data each type of page uses and how you
can set it. Callback functions are described below and are not dealt with
in this section.

The list below lists the com m ands that affect a certain page type. Unless
otherwise m entioned, these com m ands can be used both inside and
outside of a PageEx block. If used inside a PageEx block they will only
affect the current page being set by PageEx, otherwise they will set the
default for all other pages.

License page

LicenseText
LicenseData
LicenseForceSelection

Com ponents selection page

Com ponentText

Directory selection page

DirText
DirVar (can only be used in PageEx)
DirVerify



Un/Installation log page

DetailsButtonText
Com pletedText

Uninstall confirm ation page

DirVar (can only be used in PageEx)
UninstallText

Use Caption to set the page caption.

4.5.3 Callbacks

Each built-in page has three callback functions: the pre-function, the
show function and the leave-function. The pre-function is called right
before the page is created, the show-function is called right after it has
been created but before it is shown and the leave-function is called right
after the user has pressed the next button (before actually leaving the
page).

The pre-function allows you to skip the page using Abort.
The show-function allows you to tweak the page's user interface with
CreateFont, SetCtlColors, SendM essage etc.
The leave-function allows you to force the user to stay on the current
page using Abort.

A custom  page only has two callback functions, one that creates it which
is m andatory, and one leave-function that acts just like the leave-function
for built-in pages.

Exam ples:

 Page license skipLicense "" stayInLicense
 Page custom customPage "" ": custom page"
 Page instfiles

 Function skipLicense
   MessageBox MB_YESNO "Do you want to skip the license page?" IDNO no



     Abort
   no:
 FunctionEnd

 Function stayInLicense
   MessageBox MB_YESNO "Do you want to stay in the license page?" IDNO no
     Abort
   no:
 FunctionEnd

 Function customPage
   GetTempFileName $R0
   File /oname=$R0 customPage.ini
   InstallOptions::dialog $R0
   Pop $R1
   StrCmp $R1 "cancel" done
   StrCmp $R1 "back" done
   StrCmp $R1 "success" done
   error: MessageBox MB_OK|MB_ICONSTOP "InstallOptions error:$\r$\n$R1"
   done:
 FunctionEnd

4.5.4 Page

custom [creator_function] [leave_function] [caption] [/ENABLECANCEL]
  OR
internal_page_type [pre_function] [show_function] [leave_function] [/ENABLECANCEL]

Adds an installer page. See the above sections for m ore inform ation
about built-in versus custom  pages and about callback functions.

internal_page_type can be:

license - license page
com ponents - com ponents selection page
directory - installation directory selection page
instfiles - installation page where the sections are executed
uninstConfirm  - uninstall confirm ation page



The last page of the installer has its cancel button disabled to prevent
confusion. To enable it anyway, use /ENABLECANCEL.

4.5.5 UninstPage

custom [creator_function] [leave_function] [caption] [/ENABLECANCEL]
  OR
internal_page_type [pre_function] [show_function] [leave_function] [/ENABLECANCEL]

Adds an uninstaller page. See the above sections for m ore inform ation
about built-in versus custom  pages and about callback functions.

See Page for possible values of internal_page_type.

4.5.6 PageEx

[un.](custom|uninstConfirm|license|components|directory|instfiles)

Adds an installer page or an uninstaller page if the un. prefix was used.
Every PageEx m ust have a m atching PageExEnd. In a PageEx block you
can set options that are specific to this page and will not be used for
other pages. O ptions that are not set will revert to what was set outside
the PageEx block or the default if nothing was set. To set the sub-caption
for a page use Caption or SubCaption to set the default. To set the
callback functions for a page set with PageEx use PageCallbacks. See
the above sections for m ore inform ation about built-in versus custom
pages.

Exam ple usage:

 PageEx license
   LicenseText "Readme"
   LicenseData readme.rtf
 PageExEnd

 PageEx license
   LicenseData license.txt
   LicenseForceSelection checkbox



 PageExEnd

4.5.7 PageExEnd

Ends a PageEx block.

4.5.8 PageCallbacks

([creator_function] [leave_function]) | ([pre_function] [show_function] [leave_function])

Sets the callback functions for a page defined using PageEx. Can only be
used inside a PageEx block. See the above sections for m ore inform ation
about callback functions.

PageEx license
  PageCallbacks licensePre licenseShow licenseLeave
PageExEnd

4.6 Sections
Each NSIS installer contains one or m ore sections. Each of these
sections are created, m odified, and ended with the following com m ands.

Each section contains zero or m ore instructions.
Sections are executed in order by the resulting installer, and if a
com ponent page is used, the user will have the option of
disabling/enabling each visible section.
If a section's nam e is 'Uninstall' or is prefixed with 'un.', it's an
uninstaller section.

4.6.1 Section Com m ands

4.6.1.1 AddSize

size_kb

Tells the installer that the current section needs an additional "size_kb"
kilobytes of disk space. O nly valid within a section (will have no effect
outside of a section or in a function).

Section
AddSize 500
SectionEnd

4.6.1.2 Section

[/o] [([!]|[-])section_name] [section_index_output]

Begins and opens a new section. If section_nam e is em pty, om itted, or
begins with a -, then it is a hidden section and the user will not have the
option of disabling it. If the section nam e is 'Uninstall' or is prefixed with
'un.', then it is a an uninstaller section. If section_index_output is
specified, the param eter will be !defined with the section index (can be



used with SectionSetText etc). If the section nam e begins with a !, the
section will be displayed as bold. If the /o switch is specified, the section
will be unselected by default.

Section "-hidden section"
SectionEnd

Section # hidden section
SectionEnd

Section "!bold section"
SectionEnd

Section /o "optional"
SectionEnd

Section "install something" SEC_IDX
SectionEnd

To access the section index, curly brackets m ust be used and the code
m ust be located below the section in the script.

Section test1 sec1_id
SectionEnd

Section test2 sec2_id
SectionEnd

Function .onInit
  SectionGetText ${sec2_id} $0
  MessageBox MB_OK "name of ${sec2_id}:$\n$0" # will correctly display 'name of 1: test2'
FunctionEnd

Function .onInit
  SectionGetText ${sec2_id} $0
  MessageBox MB_OK "name of ${sec2_id}:$\n$0" # will incorrectly display 'name of ${sec2_id}: test1'
    # plus a warning stating:
    #   unknown variable/constant "{sec2_id}" detected, ignoring



FunctionEnd

Section test1 sec1_id
SectionEnd

Section test2 sec2_id
SectionEnd

4.6.1.3 SectionEnd

This com m and closes the current open section.

4.6.1.4 SectionIn

insttype_index [insttype_index] [RO]

This com m and specifies which install types (see InstType) the current
section defaults to the enabled state in. M ultiple SectionIn com m ands
can be specified (they are com bined). If you specify RO  as a param eter,
then the section will be read-only, m eaning the user won't be able to
change its state. The first install type defined using InstType is indexed 1,
the next 2 and so on.

InstType "full"
InstType "minimal"

Section "a section"
SectionIn 1 2
SectionEnd

Section "another section"
SectionIn 1
SectionEnd

4.6.1.5 SectionG roup



[/e] section_group_name [index_output]

This com m and inserts a section group. The section group m ust be closed
with SectionG roupEnd, and should contain 1 or m ore sections. If the
section group nam e begins with a !, its nam e will be displayed with a bold
font. If /e is present, the section group will be expanded by default. If
index_output is specified, the param eter will be !defined with the section
index (can be used with SectionSetText etc). If the nam e is prefixed with
'un.' the section group is an uninstaller section group.

SectionGroup "some stuff"
Section "a section"
SectionEnd
Section "another section"
SectionEnd
SectionGroupEnd

4.6.1.6 SectionG roupEnd

Closes a section group opened with SectionG roup.

4.6.2 Uninstall Section

A special Section nam ed 'Uninstall' m ust be created in order to generate
an uninstaller. This section should rem ove all files, registry keys etc etc
that were installed by the installer, from  the system . Here is an exam ple
of a sim ple uninstall section:

Section "Uninstall"
  Delete $INSTDIR\Uninst.exe ; delete self (see explanation below why this works)
  Delete $INSTDIR\myApp.exe
  RMDir $INSTDIR
  DeleteRegKey HKLM SOFTWARE\myApp
SectionEnd

The first Delete instruction works (deleting the uninstaller), because the
uninstaller is transparently copied to the system  tem porary directory for



the uninstall.

Note that in uninstaller code, $INSTDIR contains the directory where the
uninstaller lies. It does not necessarily contain the sam e value it
contained in the installer.

4.7 Functions
Functions are sim ilar to Sections in that they contain zero or m ore
instructions. User functions are not called by the installer directly, instead
they are called from  Sections using the Call instruction. Callback
functions will be called by the installer when a certain event occurs.

Functions m ust be declared outside of Sections or other Functions.

4.7.1 Function Com m ands

4.7.1.1 Function

[function_name]

Begins and opens a new function. Function nam es beginning with "."
(e.g. ".W hatever") are generally reserved for callback functions. Function
nam es beginning with "un." are functions that will be generated in the
Uninstaller. Hence, norm al install Sections and functions cannot call
uninstall functions, and the Uninstall Section and uninstall functions
cannot call norm al functions.

Function func
  # some commands
FunctionEnd

Section
  Call func
SectionEnd

4.7.1.2 FunctionEnd

This com m and closes the current open function.

4.7.2 Callback Functions



You can create callback functions (which have special nam es), that will
be called by the installer at certain points in the install. Below is a list of
available callbacks:

4.7.2.1 Install Callbacks

4.7.2.1.1 .onG UIInit

This callback will be called just before the first page is loaded and the
installer dialog is shown, allowing you to tweak the user interface.

Exam ple:

 !include "WinMessages.nsh"

 Function .onGUIInit
   # 1028 is the id of the branding text control
   GetDlgItem $R0 $HWNDPARENT 1028
   CreateFont $R1 "Tahoma" 10 700
   SendMessage $R0 ${WM_SETFONT} $R1 0
   # set background color to white and text color to red
   SetCtlColors $R0 FFFFFF FF0000
 FunctionEnd

4.7.2.1.2 .onInit

This callback will be called when the installer is nearly finished initializing.
If the '.onInit' function calls Abort, the installer will quit instantly.

Here are two exam ples of how this m ight be used:

 Function .onInit
   MessageBox MB_YESNO "This will install. Continue?" IDYES NoAbort
     Abort ; causes installer to quit.
   NoAbort:
 FunctionEnd

or:



 Function .onInit
   ReadINIStr $INSTDIR $WINDIR\wincmd.ini Configuration InstallDir
   StrCmp $INSTDIR "" 0 NoAbort
     MessageBox MB_OK "Windows Commander not found. Unable to get install path."
     Abort ; causes installer to quit.
   NoAbort:
 FunctionEnd

4.7.2.1.3 .onInstFailed

This callback is called when the user hits the 'cancel' button after the
install has failed (if it could not extract a file, or the install script used the
Abort com m and).

Exam ple:

  Function .onInstFailed
    MessageBox MB_OK "Better luck next time."
  FunctionEnd

4.7.2.1.4 .onInstSuccess

This callback is called when the install was successful, right before the
install window closes (which m ay be after the user clicks 'Close' if
AutoCloseW indow or SetAutoClose is set to false).

Exam ple:

  Function .onInstSuccess
    MessageBox MB_YESNO "Congrats, it worked. View readme?" IDNO NoReadme
      Exec notepad.exe ; view readme or whatever, if you want.
    NoReadme:
  FunctionEnd

4.7.2.1.5 .onG UIEnd

This callback is called right after the installer window closes. Use it to free
any user interface related plug-ins if needed.



4.7.2.1.6 .onM ouseO verSection

This callback is called whenever the m ouse position over the sections
tree has changed. This allows you to set a description for each section
for exam ple. The section id on which the m ouse is over currently is
stored, tem porarily, in $0.

Exam ple:

  Function .onMouseOverSection
    FindWindow $R0 "#32770" "" $HWNDPARENT
    GetDlgItem $R0 $R0 1043 ; description item (must be added to the UI)

    StrCmp $0 0 "" +2
      SendMessage $R0 ${WM_SETTEXT} 0 "STR:first section description"

    StrCmp $0 1 "" +2
      SendMessage $R0 ${WM_SETTEXT} 0 "STR:second section description"
  FunctionEnd

4.7.2.1.7 .onRebootFailed

This callback is called if Reboot fails. W riteUninstaller, plug-ins, File and
W riteRegBin should not be used in this callback.

Exam ple:

 Function .onRebootFailed
   MessageBox MB_OK|MB_ICONSTOP "Reboot failed. Please reboot manually." /SD IDOK
 FunctionEnd

4.7.2.1.8 .onSelChange

Called when the selection changes on the com ponent page. Useful for
using with SectionSetFlags and SectionG etFlags.

Selection changes include both section selection and installation type
changes. The section id of the changed section is stored in $0. $0 is -1 if
the installation type changed. You only get notifications for changes



initiated by the user and only one notification per action even if the action
also affected child sections and/or parent groups.

4.7.2.1.9 .onUserAbort

This callback is called when the user hits the 'cancel' button, and the
install hasn't already failed. If this function calls Abort, the install will not
be aborted.

Exam ple:

 Function .onUserAbort
   MessageBox MB_YESNO "Abort install?" IDYES NoCancelAbort
     Abort ; causes installer to not quit.
   NoCancelAbort:
 FunctionEnd

4.7.2.1.10 .onVerifyInstDir

This callback enables control over whether or not an installation path is
valid for your installer. This code will be called every tim e the user
changes the install directory, so it shouldn't do anything crazy with
M essageBox or the like. If this function calls Abort, the installation path in
$INSTDIR is deem ed invalid.

Exam ple:

  Function .onVerifyInstDir
    IfFileExists $INSTDIR\Winamp.exe PathGood
      Abort ; if $INSTDIR is not a winamp directory, don't let us install there
    PathGood:
  FunctionEnd

4.7.2.2 Uninstall Callbacks

4.7.2.2.1 un.onG UIInit

This callback will be called just before the first page is loaded and the



installer dialog is shown, allowing you to tweak the user interface.

Have a look at .onG UIInit for an exam ple.

4.7.2.2.2 un.onInit

This callback will be called when the uninstaller is nearly finished
initializing. If the ' un.onInit' function calls Abort, the uninstaller will quit
instantly. Note that this function can verify and/or m odify $INSTDIR if
necessary.

Here are two exam ples of how this m ight be used:

  Function un.onInit
    MessageBox MB_YESNO "This will uninstall. Continue?" IDYES NoAbort
      Abort ; causes uninstaller to quit.
    NoAbort:
  FunctionEnd

or:

  Function un.onInit
    IfFileExists $INSTDIR\myfile.exe found
      Messagebox MB_OK "Uninstall path incorrect"
      Abort
    found:
  FunctionEnd

4.7.2.2.3 un.onUninstFailed

This callback is called when the user hits the 'cancel' button after the
uninstall has failed (if it used the Abort com m and or otherwise failed).

Exam ple:

  Function un.onUninstFailed
    MessageBox MB_OK "Better luck next time."
  FunctionEnd



4.7.2.2.4 un.onUninstSuccess

This callback is called when the uninstall was successful, right before the
install window closes (which m ay be after the user clicks 'Close' if
SetAutoClose is set to false)..

Exam ple:

  Function un.onUninstSuccess
    MessageBox MB_OK "Congrats, it's gone."
  FunctionEnd

4.7.2.2.5 un.onG UIEnd

This callback is called right after the uninstaller window closes. Use it to
free any user interface related plug-ins if needed.

4.7.2.2.6 un.onRebootFailed

This callback is called if Reboot fails. W riteUninstaller, plug-ins, File and
W riteRegBin should not be used in this callback.

Exam ple:

 Function un.onRebootFailed
   MessageBox MB_OK|MB_ICONSTOP "Reboot failed. Please reboot manually." /SD IDOK
 FunctionEnd

4.7.2.2.7 un.onSelChange

Called when the selection changes on the com ponent page. Useful for
using with SectionSetFlags and SectionG etFlags.

Selection changes include both section selection and installation type
changes. The section id of the changed section is stored in $0. $0 is -1 if
the installation type changed. You only get notifications for changes
initiated by the user and only one notification per action even if the action
also affected child sections and/or parent groups.



4.7.2.2.8 un.onUserAbort

This callback is called when the user hits the 'cancel' button and the
uninstall hasn't already failed. If this function calls Abort, the install will
not be aborted.

Exam ple:

  Function un.onUserAbort
    MessageBox MB_YESNO "Abort uninstall?" IDYES NoCancelAbort
      Abort ; causes uninstaller to not quit.
    NoCancelAbort:
  FunctionEnd

4.8 Installer Attributes

4.8.1 G eneral Attributes

The com m ands below all adjust attributes of the installer. These
attributes control how the installer looks and functions, including which
pages are present in the installer, which text is displayed in each part of
each page, the nam e of the installer, the icon the installer uses, the
default installation directory and m ore. Note that these attributes can be
set anywhere in the file except in a Section or Function.

Defaults are bold and underlined

4.8.1.1 AddBrandingIm age

(left|right|top|bottom) (width|height) [padding]

Adds a branding im age on the top, bottom , left, or right of the installer. Its
size will be set according to the width/height specified, the installer
width/height and the installers font. The final size will not always be what
you requested; have a look at the output of the com m and for the actual
size. Because this depends on the installers font, you should use SetFont
before AddBrandingIm age. The default padding value is 2.

AddBrandingIm age only adds a placeholder for an im age. To set the
im age itself at runtim e, use SetBrandingIm age.

AddBrandingImage left 100
AddBrandingImage right 50
AddBrandingImage top 20
AddBrandingImage bottom 35
AddBrandingImage left 100 5

4.8.1.2 Allow RootDirInstall



true|false

Controls whether or not installs are allowed in the root directory of a
drive, or directly into a network share. Set to 'true' to change the safe
behavior, which prevents users from  selecting C:\ or \\Server\Share as an
install (and later on, uninstall) directory. For additional directory selection
page custom izability, see .onVerifyInstDir.

4.8.1.3 AutoCloseW indow

true|false

Sets whether or not the install window autom atically closes when
com pleted. This is overrideable from  a section using SetAutoClose.

4.8.1.4 BG Font

[font_face [height [weight] [/ITALIC] [/UNDERLINE] [/STRIKE]]]

Specifies the font used to show the text on the background gradient. To
set the color use BG G radient. The default font will be used if no
param eters are specified. The default font is bold and italic Tim es New
Rom an.

4.8.1.5 BG G radient

[off|(topc botc [textcolor|notext])]

Specifies whether or not to use a gradient background window. If 'off', the
installer will not show a background window, if no param eters are
specified, the default black to blue gradient is used, and otherwise the
top_color or bottom _color are used to m ake a gradient. Top_color and
bottom _color are specified using the form  RRG G BB (in hexadecim al, as
in HTM L, only m inus the leading '#', since # can be used for com m ents).
'textcolor' can be specified as well, or 'notext' can be specified to turn the
big background text off.



4.8.1.6 BrandingText

/TRIM(LEFT|RIGHT|CENTER) text

Sets the text that is shown at the bottom  of the install window (by default
it is 'Nullsoft Install System  vX.XX'). Setting this to an em pty string ("")
uses the default; to set the string to blank, use " " (a space). If it doesn't
m atter to you, leave it the default so that everybody can know why the
installer didn't suck :). Use /TRIM LEFT, /TRIM RIG HT or /TRIM CENTER
to trim  down the size of the control to the size of the string.

Accepts variables. If variables are used, they m ust be initialized on
.onInit.

4.8.1.7 Caption

caption

W hen used outside a PageEx block: Sets the text for the titlebar of the
installer. By default it is '$(̂ N am e) Setup', where Nam e is specified by the
Nam e instruction. You can however override it with 'M yApp Installer' or
whatever. If you specify an em pty string (""), the default will be used (you
can specify " " to sim ulate a em pty string).

W hen used inside a PageEx block: Sets the subcaption of the current
page.

Accepts variables. If variables are used, they m ust be initialized on .onInit
or .onG UIInit.

4.8.1.8 ChangeUI

dialog ui_file.exe

Replaces dialog (IDD_LICENSE, IDD_DIR, IDD_SELCO M , IDD_INST,
IDD_INSTFILES, IDD_UNINST or IDD_VERIFY) with a dialog from
ui_file.exe with the sam e resource ID. You can also specify 'all' as the



dialog if you wish to replace all 7 of the dialogs at once from  the sam e UI
file. For som e exam ple UIs look at Contrib\UIs under your NSIS directory.

IDD_LICENSE m ust contain IDC_EDIT1 (RICHEDIT control).
IDD_DIR m ust contain IDC_DIR (edit box), IDC_BRO W SE (button)
and IDC_CHECK1 (checkbox).
IDD_SELCO M  m ust contain IDC_TREE1 (SysTreeView32 control),
and IDC_CO M BO 1 (com bo box).
IDD_INST m ust contain IDC_BACK (button), IDC_CHILDRECT
(static control the size of all other dialogs), IDC_VERSTR (static),
IDO K (button), and IDCANCEL (button). If an im age control (static
with SS_BITM AP style) will be found in this dialog it will be used as
the default for SetBrandingIm age.
IDD_INSTFILES m ust contain IDC_LIST1 (SysListView32 control),
IDC_PRO G RESS (m sctls_progress32 control), and
IDC_SHO W DETAILS (button).
IDD_UNINST m ust contain IDC_EDIT1 (edit box).
IDD_VERIFY m ust contain IDC_STR (static).

ChangeUI all "${NSISDIR}\Contrib\UIs\sdbarker_tiny.exe"

4.8.1.9 CheckBitm ap

bitmap.bmp

Specifies the bitm ap with the checkbox im ages used in the com ponent-
selection page treeview.

This bitm ap should have a size of 96x16 pixels, no m ore than 8bpp (256
colors) and contain six 16x16 im ages for the different states (in order:
selection m ask, not checked, checked, greyed out, unchecked & read-
only, checked & read-only). Use m agenta as m ask color (this area will be
transparent).

4.8.1.10 Com pletedText

text



Replaces the default text ("Com pleted") that is printed at the end of the
install if param eter is specified. O therwise, the default is used.

Accepts variables. If variables are used, they m ust be initialized before
the m essage is printed.

4.8.1.11 Com ponentText

[text [subtext] [subtext2]]

Used to change the default text on the com ponent page.

text: Text above the controls, to the right of the installation icon.

subtext: Text next to the installation type selection.

subtext2: Text to the left of the com ponents list and below the installation
type.

The default string will be used if a string is em pty ("").

Accepts variables. If variables are used, they m ust be initialized before
the com ponents page is created.

4.8.1.12 CRCCheck

on|off|force

Specifies whether or not the installer will perform  a CRC on itself before
allowing an install. Note that if the user uses /NCRC on the com m and line
when executing the installer, and you didn't specify 'force', the CRC will
not occur, and the user will be allowed to install a (potentially) corrupted
installer.

4.8.1.13 DetailsButtonText

show_details_text



Replaces the default details button text of "Show details", if param eter is
specified (otherwise the default is used).

Accepts variables. If variables are used, they m ust be initialized before
the install log (instfiles) page is created.

4.8.1.14 DirText

[text] [subtext] [browse_button_text] [browse_dlg_text]

Used to change the default text on the directory page.

text: Text above the controls, to the right of the installation icon.

subtext: Text on the directory selection fram e.

browse_button_text: Text on the Browse button.

browse_dlg_text: Text on the "Browse For Folder" dialog, appears after
clicking on "Browse" button.

The default string will be used if a string is em pty ("").

Accepts variables. If variables are used, they m ust be initialized before
the directory page is created.

4.8.1.15 DirVar

user_var(dir input/output)

Specifies which variable is to be used to contain the directory selected.
This variable should be initialized with a default value. This allows you to
easily create two different directory pages that will not require you to
m ove values in and out of $INSTDIR. The default variable is $INSTDIR.
This can only be used in PageEx for directory and uninstConfirm  pages.

Var ANOTHER_DIR
PageEx directory



  DirVar $ANOTHER_DIR
PageExEnd

Section
  SetOutPath $INSTDIR
  File "a file.dat"
  SetOutPath $ANOTHER_DIR
  File "another file.dat"
SectionEnd

4.8.1.16 DirVerify

auto|leave

If ̀D irVerify leave' is used, the Next button will not be disabled if the
installation directory is not valid or there is not enough space. A flag that
you can read in the leave function using G etInstDirError will be set
instead.

PageEx directory
  DirVerify leave
  PageCallbacks "" "" dirLeave
PageExEnd

4.8.1.17 FileErrorText

file_error_text [noignore_file_error_text]

Replaces the default text that com es up when a file cannot be written to.
This string can contain a reference to $0, which is the filenam e ($0 is
tem porarily changed to this value). Exam ple: "Can not write to file
$\r$\n$0$\r$\ngood luck.".

Accepts variables. If variables are used, they m ust be initialized before
File is used.



4.8.1.18 Icon

[path\]icon.ico

Sets the icon of the installer. Every im age in the icon file will be included
in the installer. Use UninstallIcon to set the uninstaller icon.

4.8.1.19 InstallButtonText

install_button_text

If param eter is specified, overrides the default install button text (of
"Install") with the specified text.

Accepts variables. If variables are used, they m ust be initialized before
the install button shows.

4.8.1.20 InstallColors

/windows | (foreground_color background_color)

Sets the colors to use for the install info screen (the default is 00FF00
000000. Use the form  RRG G BB (in hexadecim al, as in HTM L, only m inus
the leading '#', since # can be used for com m ents). Note that if
"/windows" is specified as the only param eter, the default windows colors
will be used.

4.8.1.21 InstallDir

definstdir

Sets the default installation directory. See the variables section for
variables that can be used to m ake this string (especially
$PRO G RAM FILES). Note that the part of this string following the last \
will be used if the user selects 'browse', and m ay be appended back on
to the string at install tim e (to disable this, end the directory with a \



(which will require the entire param eter to be enclosed with quotes). If
this doesn't m ake any sense, play around with the browse button a bit.

4.8.1.22 InstallDirRegKey

root_key subkey key_name

This attribute tells the installer to check a string in the registry and use it
as the install dir if that string is valid. If this attribute is present, it will
override the InstallDir attribute if the registry key is valid, otherwise it will
fall back to the InstallDir value. W hen querying the registry, this com m and
will autom atically rem ove any quotes. If the string ends in ".exe", it will
autom atically rem ove the filenam e com ponent of the string (i.e. if the
string is "C:\Program  Files\Foo\app.exe", it will know to use "C:\Program
Files\Foo"). For m ore advanced install directory configuration, set
$INSTDIR in .onInit.

Language strings and variables cannot be used with InstallDirRegKey.

InstallDirRegKey HKLM Software\NSIS ""
InstallDirRegKey HKLM Software\ACME\Thingy InstallLocation

4.8.1.23 InstProgressFlags

[flag [...]]

Valid values for flag are "sm ooth" (sm ooth the progress bar) or "colored"
(color the progress bar with the colors set by InstallColors. Exam ples:
"InstProgressFlags" (default old-school windows look),
"InstProgressFlags sm ooth" (new sm ooth look), "InstProgressFlags
sm ooth colored" (colored sm ooth look whee). Note: neither "sm ooth" or
"colored" work with XPStyle on when the installer runs on W indows XP
with a m odern them e.

4.8.1.24 InstType



install_type_name | /NOCUSTOM | /CUSTOMSTRING=str | /COMPONENTSONLYONCUSTOM

Adds an install type to the install type list, or disables the custom  install
type. There can be as m any as 32 types, each one specifying the nam e
of the install type. If the nam e is prefixed with 'un.' it is an uninstaller
install type. The nam e can contain variables which will be processed at
runtim e before the com ponents page shows. Another way of changing
the InstType nam e during runtim e is the InstTypeSetText com m and. The
difference is that with InstTypeSetText you are saving your precious user
variables. The first type is the default (generally 'Typical'). If the
/NO CUSTO M  switch is specified, then the "custom " install type is
disabled, and the user has to choose one of the pre-defined install types.
Alternatively, if the /CUSTO M STRING  switch is specified, the param eter
will override the "Custom " install type text. Alternatively, if the
/CO M PO NENTSO NLYO NCUSTO M  flag is specified, the com ponent list
will only be shown if the "Custom " install type is selected.

Accepts variables for type nam es. If variables are used, they m ust be
initialized before the com ponents page is created.

4.8.1.25 LicenseBkColor

color | /gray | /windows

Sets the background color of the license data. Color is specified using the
form  RRG G BB (in hexadecim al, as in HTM L, only m inus the leading '#',
since # can be used for com m ents). Default is '/gray'. You can also use
the W indows O S defined color by using '/windows'.

4.8.1.26 LicenseData

licdata.(txt|rtf)

Specifies a text file or a RTF file to use for the license that the user can
read. O m it this to not have a license displayed. Note that the file m ust be
in DO S text form at (\r\n). To define a m ultilingual license data use
LicenseLangString.



If you are using a RTF file it is recom m ended that you edit it with
W ordPad and not M S W ord. Using W ordPad will result in a m uch sm aller
file.

Use LicenseLangString to show a different license for every language.

4.8.1.27 LicenseForceSelection

(checkbox [accept_text] | radiobuttons [accept_text] [decline_text] | 

Specifies if the displayed license m ust be explicitly accepted or not. This
can be done either by a checkbox or by radiobuttons. By default the "next
button" is disabled and will only be enabled if the checkbox is enabled or
the correct radio button is selected. If off is specified the "next button" is
enabled by default.

LicenseForceSelection checkbox
LicenseForceSelection checkbox "i accept"
LicenseForceSelection radiobuttons
LicenseForceSelection radiobuttons "i accept"
LicenseForceSelection radiobuttons "i accept" "i decline"
LicenseForceSelection radiobuttons "" "i decline"
LicenseForceSelection off

4.8.1.28 LicenseText

[text [button_text]]

Used to change the default text on the license page.

text: Text above the controls, to the right of the installation icon.

button_text: Text on the "I Agree" button.

The default string will be used if a string is em pty ("").

Accepts variables. If variables are used, they m ust be initialized before



the license page is created.

4.8.1.29 M anifestDPIAw are

notset|true|false

Declare that the installer is DPI-aware. A DPI-aware application is not
scaled by the DW M  (DPI virtualization) so the text is never blurry. NSIS
does not scale the bitm ap used by the tree control on the com ponent
page and som e plugins m ight have com patibility issues so m ake sure
that you test your installer at different DPI settings if you select true.

See M SDN for m ore inform ation about DPI-aware applications.

4.8.1.30 M anifestSupportedO S

none|all|WinVista|Win7|Win8|Win8.1|Win10|{GUID} [...]

Declare that the installer is com patible with the specified W indows
version(s). This adds a SupportedO S entry in the com patibility section of
the application m anifest. The default is W in7+8+8.1+10. none is the
default if RequestExecutionLevel is set to none for com patibility reasons.

W indows 8.1 and later will fake its version num ber if you don't declare
support for that particular version. You can read m ore about the other
changes in behavior on M SDN.

4.8.1.31 M iscButtonText

[back_button_text [next_button_text] [cancel_button_text] [close_button_text]]

Replaces the default text strings for the four buttons (< Back, Next >,
Cancel and Close). If param eters are om itted, the defaults are used.

Accepts variables. If variables are used, they m ust be initialized in .onInit.



4.8.1.32 Nam e

name [name_doubled_ampersands]

Sets the nam e of the installer. The nam e is usually sim ply the product
nam e such as 'M yApp' or 'CrapSoft M yApp'. If you have one or m ore
am persands (&) in the nam e, set the second param eter to the sam e
nam e, only with doubled am persands. For exam ple, if your product's
nam e is "Foo & Bar", use:

 Name "Foo & Bar" "Foo && Bar"

If you have am persands in the nam e and use a LangString for the nam e,
you will have to create another one with doubled am persands to use as
the second param eter.

Accepts variables. If variables are used, they m ust be initialized in .onInit.

4.8.1.33 O utFile

[path\]install.exe

Specifies the output file that the M akeNSIS should write the installer to.
This is just the file that M akeNSIS writes, it doesn't affect the contents of
the installer.

4.8.1.34 RequestExecutionLevel

none|user|highest|admin

Specifies the requested execution level for W indows Vista and higher.
The value is em bedded in the installer and uninstaller's XM L m anifest
and tells W indows which privilege level the installer requires. user
requests the user's norm al level with no adm inistrative privileges. highest
will request the highest execution level available for the current user and
will cause W indows to prom pt the user to verify privilege escalation if
they are a m em ber of the adm inistrators group. The prom pt m ight



request for the user's password. adm in, which is also the default,
requests adm inistrator level and will cause W indows to prom pt the user
as well. Specifying none will keep the m anifest em pty and let W indows
decide which execution level is required. W indows autom atically
identifies NSIS installers and decides adm inistrator privileges are
required. Because of this, none and adm in have virtually the sam e effect.

It's recom m ended that every application is m arked with a required
execution level. Unm arked installers are subject to com patibility m ode.
W orkarounds of this m ode include autom atically m oving any shortcuts
created in the user's start m enu to all users' start m enu. Installers that
don't install anything into system  folders nor write to the local m achine
registry (HKLM ) should specify user execution level.

M ore inform ation about this topic can be found on M SDN.

4.8.1.35 SetFont

[/LANG=lang_id] font_face_name font_size

Sets the installer font. Please rem em ber that the font you choose m ust
be present on the user's m achine as well. Don't use rare fonts that only
you have.

Use the /LANG  switch if you wish to set a different font for each
language. For exam ple:

 SetFont /LANG=${LANG_ENGLISH} "English Font" 9
 SetFont /LANG=${LANG_FRENCH} "French Font" 10

There are two LangStrings nam ed ̂ Font and ̂ FontSize which contain the
font and font size for every language.

4.8.1.36 Show InstDetails

hide|show|nevershow

Sets whether or not the details of the install are shown. Can be 'hide' to



hide the details by default, allowing the user to view them , or 'show' to
show them  by default, or 'nevershow', to prevent the user from  ever
seeing them . Note that sections can override this using SetDetailsView.

4.8.1.37 Show UninstDetails

hide|show|nevershow

Sets whether or not the details of the uninstall are shown. Can be 'hide'
to hide the details by default, allowing the user to view them , or 'show' to
show them  by default, or 'nevershow', to prevent the user from  ever
seeing them . Note that sections can override this using SetDetailsView.

4.8.1.38 SilentInstall

normal|silent|silentlog

Specifies whether or not the installer should be silent. If it is 'silent' or
'silentlog', all sections that have the SF_SELECTED flag are installed
quietly (you can set this flag using SectionSetFlags), with no screen
output from  the installer itself (the script can still display whatever it
wants, use M essageBox's /SD to specify a default for silent installers).
Note that if this is set to 'norm al' and the user runs the installer with /S
(case sensitive) on the com m and line, it will behave as if SilentInstall
'silent' was used. Note: see also LogSet.

See section 4.12 for m ore inform ation.

4.8.1.39 SilentUnInstall

normal|silent

Specifies whether or not the uninstaller should be silent. If it is 'silent' the
uninstall sections will run quietly, with no screen output from  the
uninstaller itself (the script can still display whatever it wants, use
M essageBox's /SD to specify a default for silent uninstallers). Note that if



this is set to 'norm al' and the user runs the uninstaller with /S on the
com m and line, it will behave as if SilentUnInstall 'silent' was used.

See section 4.12 for m ore inform ation.

4.8.1.40 SpaceTexts

[req_text [avail_text]]

If param eters are specified, overrides the space required and space
available text ("Space required: " and "Space available: " by default). If
'none' is specified as the required text no space texts will be shown.

Accepts variables. If variables are used, they m ust be initialized before
the com ponents page is created.

4.8.1.41 SubCaption

[page_number subcaption]

O verrides the subcaptions for each of the installer pages (0=": License
Agreem ent",1=": Installation O ptions",2=": Installation Directory", 3=":
Installing Files", 4=": Com pleted"). If you specify an em pty string (""), the
default will be used (you can however specify " " to achieve a blank
string).

You can also set a subcaption (or override the default) using Caption
inside a PageEx block.

Accepts variables. If variables are used, they m ust be initialized before
the relevant page is created.

4.8.1.42 UninstallButtonText

text

Changes the text of the button that by default says "Uninstall" in the



uninstaller. If no param eter is specified, the default text is used.

Accepts variables. If variables are used, they m ust be initialized before
the uninstall button shows.

4.8.1.43 UninstallCaption

caption

Sets what the titlebars of the uninstaller will display. By default it is
'$(̂ N am e) Uninstall', where Nam e is specified with the Nam e com m and.
You can, however, override it with 'M yApp uninstaller' or whatever. If you
specify an em pty string (""), the default will be used (you can specify " " to
sim ulate a em pty string).

Accepts variables. If variables are used, they m ust be initialized in
un.onInit.

4.8.1.44 UninstallIcon

[path\]icon.ico

Sets the icon of the uninstaller.

4.8.1.45 UninstallSubCaption

page_number subcaption

Sets the default subcaptions for the uninstaller pages (0=":
Confirm ation",1=": Uninstalling Files",2=": Com pleted"). If you specify an
em pty string (""), the default will be used (you can specify " " to sim ulate
a em pty string).

You can also set a subcaption (or override the default) using Caption
inside a PageEx block.

Accepts variables. If variables are used, they m ust be initialized before



the relevant page is created.

4.8.1.46 UninstallText

text [subtext]

Specifies the texts on the uninstaller confirm  page.

text: Text above the controls

subtext: Text next to the uninstall location

Accepts variables. If variables are used, they m ust be initialized before
the uninstaller confirm  page is created.

4.8.1.47 W indow Icon

on|off

Sets whether or not the installer's icon is displayed on certain pages.

4.8.1.48 XPStyle

on|off

Sets whether or not a XP visual style m anifest will be added to the
installer. This m anifest m akes the installers controls use the new visual
styles when running on W indows XP and later. This affects the uninstaller
too.

4.8.2 Com piler Flags

The following com m ands affect how the com piler generates code and
com presses data. Unless otherwise noted, these com m ands are valid
anywhere in the script and affect every line below where each one is
placed (until overridden by another com m and). They cannot be jum ped
over using flow control instructions.



For exam ple, in the following script, blah.dat will never be overwritten.

${If} $0 == 0
  SetOverwrite on
${Else}
  SetOverwrite off
${EndIf}
File blah.dat # overwrite is always off here!

Instead, the following should be used.

${If} $0 == 0
  SetOverwrite on
  File blah.dat
${Else}
  SetOverwrite off
  File blah.dat
${EndIf}

4.8.2.1 Allow SkipFiles

on|off

This com m and specifies whether the user should be able to skip a file or
not. A user has an option to skip a file if SetO verwrite is set to on (default)
and the installer fails to open a file for writing when trying to extract a file.
If off is used the ignore button which allows the user to skip the file will
not be shown and the user will only have an option to abort the
installation (Cancel button) or retry opening the file for writing (Retry
button). If on is used the user will have an option to skip the file (error flag
will be set - see SetO verwrite).

4.8.2.2 FileBufSize

buffer_size_in_mb

This com m and sets the size of the com piler's internal file buffers. This



com m and allows you to control the com piler's m em ory usage by lim iting
how m uch of a given file it will load into m em ory at once. Since the
com piler needs both input and output, twice the m em ory size specified
could be used at any given tim e for file buffers. This com m and does not
lim it the com pression buffers which could take another couple of M B,
neither does it lim it the com piler's other internal buffers, but those
shouldn't norm ally top 1M B anyway. Specifying a very sm all num ber
could decrease perform ance. Specifying a very large num ber could
exhaust system  resources and force the com piler to cancel the
com pilation process. The default value is 32M B.

4.8.2.3 SetCom press

auto|force|off

This com m and sets the com press flag which is used by the installer to
determ ine whether or not data should be com pressed. Typically the
SetCom press flag will affect the com m ands after it, and the last
SetCom press com m and in the file also determ ines whether or not the
install info section and uninstall data of the installer is com pressed. If
com pressflag is 'auto', then files are com pressed if the com pressed size
is sm aller than the uncom pressed size. If com pressflag is set to 'force',
then the com pressed version is always used. If com pressflag is 'off' then
com pression is not used (which can be faster).

Note that this option has no effect when solid com pression is used.

4.8.2.4 SetCom pressor

[/SOLID] [/FINAL] zlib|bzip2|lzma

This com m and sets the com pression algorithm  used to com press
files/data in the installer. It can only be used outside of sections and
functions and before any data is com pressed. Different com pression
m ethods can not be used for different files in the sam e installer. It is
recom m ended to use it at the very top of the script to avoid com pilation
errors.



Three com pression m ethods are supported: ZLIB, BZIP2 and LZM A.

ZLIB (the default) uses the deflate algorithm , it is a quick and sim ple
m ethod. W ith the default com pression level it uses about 300 KB of
m em ory.

BZIP2 usually gives better com pression ratios than ZLIB, but it is a bit
slower and uses m ore m em ory. W ith the default com pression level it
uses about 4 M B of m em ory.

LZM A is a new com pression m ethod that gives very good com pression
ratios. The decom pression speed is high (10-20 M B/s on a 2 G Hz CPU),
the com pression speed is lower. The m em ory size that will be used for
decom pression is the dictionary size plus a few KBs, the default is 8 M B.

If /FINAL is used, subsequent calls to SetCom pressor will be ignored.

If /SO LID is used, all of the installer data is com pressed in one block.
This results in greater com pression ratios.

4.8.2.5 SetCom pressorDictSize

dict_size_mb

Sets the dictionary size in m egabytes (M B) used by the LZM A
com pressor (default is 8 M B).

4.8.2.6 SetDatablockO ptim ize

on|off

This com m and tells the com piler whether or not to do datablock
optim izations. Datablock optim izations causes the com piler to check to
see if any data being added to the data block is already in the data block,
and if so, it is sim ply referenced as opposed to added (can save a little bit
of size). It is highly recom m ended to leave this option on.



4.8.2.7 SetDateSave

on|off

This com m and sets the file date/tim e saving flag which is used by the File
com m and to determ ine whether or not to save the last write date and
tim e of the file, so that it can be restored on installation. Valid flags are
'on' and 'off'. 'on' is the default.

4.8.2.8 SetO verw rite

on|off|try|ifnewer|ifdiff|lastused

This com m and sets the overwrite flag which is used by the File com m and
to determ ine whether or not the file should overwrite any existing files
that are present. If overwriteflag is 'on', files are overwritten (this is the
default). If overwriteflag is 'off', files that are already present are not
overwritten. If overwriteflag is 'try', files are overwritten if possible
(m eaning that if the file is not able to be written to, it is skipped without
any user interaction). If overwriteflag is 'ifnewer', then files are only
overwritten if the existing file is older than the new file. If overwriteflag is
'ifdiff', then files are only overwritten if the existing file is older or newer
than the new file. Note that when in 'ifnewer' or 'ifdiff' m ode, the
destination file's date is set, regardless of what SetDateSave is set to.

SetOverwrite off
File program.cfg # config file we don't want to overwrite
SetOverwrite on

4.8.2.9 Unicode

true|false

G enerate a Unicode installer. It can only be used outside of sections and
functions and before any data is com pressed.



4.8.3 Version Inform ation

4.8.3.1 VIAddVersionKey

 [/LANG=lang_id] keyname value

Adds a string entry to the version inform ation stored in the installer and
uninstaller. These can be viewed in the File Properties Version or Details
tab. keynam e can either be a special nam e known by W indows or a user
defined nam e. /LANG =0 can be used to indicate a language neutral
language id. The following nam es are known by W indows:

ProductNam e
Com m ents
Com panyNam e
LegalCopyright
FileDescription
FileVersion
ProductVersion
InternalNam e
LegalTradem arks
O riginalFilenam e
PrivateBuild
SpecialBuild

The displayed nam e of these special entries are translated on the target
system , whereas user defined keynam es rem ain untranslated.

VIAddVersionKey /LANG=${LANG_ENGLISH} "ProductName" "Test Application"
VIAddVersionKey /LANG=${LANG_ENGLISH} "Comments" "A test comment"
VIAddVersionKey /LANG=${LANG_ENGLISH} "CompanyName" "Fake company"
VIAddVersionKey /LANG=${LANG_ENGLISH} "LegalTrademarks" "Test Application is a trademark of Fake company"
VIAddVersionKey /LANG=${LANG_ENGLISH} "LegalCopyright" "© Fake company"
VIAddVersionKey /LANG=${LANG_ENGLISH} "FileDescription" "Test Application"
VIAddVersionKey /LANG=${LANG_ENGLISH} "FileVersion" "1.2.3"

4.8.3.2 VIProductVersion



version_string_X.X.X.X

Sets the Product Version in the VS_FIXEDFILEINFO  version inform ation
block.

VIProductVersion 1.2.3.4

4.8.3.3 VIFileVersion

version_string_X.X.X.X

Sets the File Version in the VS_FIXEDFILEINFO  version inform ation
block (You should also set the FileVersion string with VIAddVersionKey
so the inform ation is displayed at the top of the Version Tab in the
Properties of the file). If you don't provide a File Version the Product
Version is used in the VS_FIXEDFILEINFO  block.

VIFileVersion 1.2.3.4

Previous | Contents | Next

4.9 Instructions

4.9.1 Basic Instructions

The instructions that NSIS uses for scripting are sort of a cross between
PHP and assem bly. There are no real high level language constructs but
the instructions them selves are (for the m ost part) high level, and you
have handy string capability (i.e. you don't have to worry about
concatenating strings, etc). You essentially have 25 registers (20 general
purpose, 5 special purpose), and a stack.

4.9.1.1 Delete

[/REBOOTOK] file

Delete file (which can be a file or wildcard, but should be specified with a
full path) from  the target system . If /REBO O TO K is specified and the file
cannot be deleted then the file is deleted when the system  reboots -- if
the file will be deleted on a reboot, the reboot flag will be set. The error
flag is set if files are found and cannot be deleted. The error flag is not
set when trying to delete a file that does not exist.

Delete $INSTDIR\somefile.dat

W arning: The /REBO O TO K switch requires adm inistrator rights on
W indows NT and later.

4.9.1.2 Exec

command

Execute the specified program  and continue im m ediately. Note that the
file specified m ust exist on the target system , not the com piling system .
$O UTDIR is used as the working directory. The error flag is set if the
process could not be launched. Note, if the com m and could have spaces,



you should put it in quotes to delim it it from  param eters. e.g.: Exec
'"$INSTDIR\com m and.exe" param eters'. If you don't put it in quotes it will
not work on W indows 9x with or without param eters.

Exec '"$INSTDIR\someprogram.exe"'
Exec '"$INSTDIR\someprogram.exe" some parameters'

4.9.1.3 ExecShell

[/INVOKEIDLIST] action command [parameters] [SW_SHOWDEFAULT | SW_SHOWNORMAL | SW_SHOWMAXIMIZED | SW_SHOWMINIMIZED | SW_HIDE]

Execute the specified program  using ShellExecuteEx. Note that action is
usually "open", "print", etc, but can be an em pty string to use the default
action. Param eters and the show type are optional. $O UTDIR is used as
the working directory. The error flag is set if the process could not be
launched.

ExecShell "open" "http://nsis.sf.net/"
ExecShell "open" "$INSTDIR\readme.txt"
ExecShell "print" "$INSTDIR\readme.txt"
ExecShell /INVOKEIDLIST "properties" "$TEMP"

4.9.1.4 ExecShellW ait

[/INVOKEIDLIST] action command [parameters] [SW_SHOWDEFAULT | SW_SHOWNORMAL | SW_SHOWMAXIMIZED | SW_SHOWMINIMIZED | SW_HIDE]

Execute the specified program  using ExecShell and wait for executed
process to quit. It will only wait for executable files and not other file types
nor URLs.

4.9.1.5 ExecW ait

command [user_var(exit code)]

Execute the specified program  and wait for the executed process to quit.
See Exec for m ore inform ation. If no output variable is specified ExecW ait



sets the error flag if the program  executed returns a nonzero error code,
or if there is an error. If an output variable is specified, ExecW ait sets the
variable with the exit code (and only sets the error flag if an error occurs;
if an error occurs the contents of the user variable are undefined). Note, if
the com m and could have spaces, you should put it in quotes to delim it it
from  param eters. e.g.: ExecW ait '"$INSTDIR\com m and.exe" param eters'.
If you don't put it in quotes it will not work on W indows 9x with or without
param eters.

ExecWait '"$INSTDIR\someprogram.exe"'
ExecWait '"$INSTDIR\someprogram.exe"' $0
DetailPrint "some program returned $0"

4.9.1.6 File

[/nonfatal] [/a] ([/r] [/x file|wildcard [...]] (file|wildcard) [...] | /oname=file.dat infile.dat)

Adds file(s) to be extracted to the current output path ($O UTDIR).

Note that the output file nam e is $O UTDIR\filenam e_portion_of_file.
Use /onam e=X switch to change the output nam e. X m ay contain
variables and can be a fully qualified path or a relative path in which
case it will be appended to $O UTDIR set by SetO utPath. W hen
using this switch, only one file can be specified. If the output nam e
contains spaces, quote the entire param eter, including /onam e, as
shown in the exam ples below.
W ildcards are supported.
If the /r switch is used, m atching files and directories are recursively
searched for in subdirectories. If just one path segm ent is specified
(e.g. File /r something), the current directory will be recursively
searched. If m ore than one segm ent is specified (e.g. File /r
something\*.*), the last path segm ent will be used as the m atching
condition and anything before it specifies which directory to search
recursively. If a directory nam e m atches, all of its contents is added
recursively. Directory structure is preserved.
Use the /x switch to exclude files and directories.
If the /a switch is used, the attributes of the file(s) added will be
preserved.



The File com m and sets the error flag if overwrite m ode is set to 'try'
and the file could not be overwritten, or if the overwrite m ode is set to
'on' and the file could not be overwritten and the user selects ignore.
If the /nonfatal switch is used and no files are found, a warning will
be issued instead of an error.

File something.exe
File /a something.exe
File *.exe
File /r *.dat
File /r data
File /oname=temp.dat somefile.ext
File /oname=$TEMP\temp.dat somefile.ext
File "/oname=$TEMP\name with spaces.dat" somefile.ext
File /nonfatal "a file that might not exist"
File /r /x CVS myproject\*.*
File /r /x *.res /x *.obj /x *.pch source\*.*

Note: when using the /r switch, both m atching directories and files will be
searched. This is always done with or without the use of wildcards, even
if the given path perfectly m atches one directory. That m eans, the
following directory structure:

<DIR> something
  file.dat
  another.dat
<DIR> dir
  something
  <DIR> dir2
    file2.dat
<DIR> another
  <DIR> something
    readme.txt

with the following File usage:

File /r something



will m atch the directory nam ed som ething in the root directory, the file
nam ed som ething in the directory nam ed dir and the directory nam ed
som ething in the directory nam ed another. To m atch only the directory
nam ed som ething in the root directory, use the following:

File /r something\*.*

W hen adding \*.*, it will be used as the m atching condition and som ething
will be used as the directory to search. W hen only som ething is specified,
the current directory will be recursively searched for every file and
directory nam ed som ething and another\som ething will be m atched.

4.9.1.7 Renam e

[/REBOOTOK] source_file dest_file

Renam e source_file to dest_file. You can use it to m ove a file from
anywhere on the system  to anywhere else and you can m ove a directory
to som ewhere else on the sam e drive. The destination file m ust not exist
or the m ove will fail (unless you are using /REBO O TO K). If /REBO O TO K
is specified, and the file cannot be m oved (if, for exam ple, the destination
exists), then the file is m oved when the system  reboots. If the file will be
m oved on a reboot, the reboot flag will be set. The error flag is set if the
file cannot be renam ed (and /REBO O TO K is not used) or if the source
file does not exist.

If no absolute path is specified the current folder will be used. The current
folder is the folder set using the last SetO utPath instruction. If you have
not used SetO utPath the current folder is $EXEDIR.

Rename $INSTDIR\file.ext $INSTDIR\file.dat

W arning: The /REBO O TO K switch requires adm inistrator rights on
W indows NT and later.

4.9.1.8 ReserveFile



[/nonfatal] [/r] [/x file|wildcard [...]] file [file...] | [/nonfatal] /plugin file.dll

Reserves a file in the data block for later use. Files are added to the
com pressed data block in the order they appear in the script. Functions,
however, are not necessarily called in the order they appear in the script.
Therefore, if you add a file in a function called early but put the function at
the end of the script, all of the files added earlier will have to be
decom pressed to get to the required file. This process can take a long
tim e if there a lot of files. .onInit is one such function. It is called at the
very beginning, before anything else appears. If you put it at the very end
of the script, extract som e files in it and have lots of files added before it,
the installer m ight take a very long tim e to load. This is where this
com m and com es useful, allowing you to speed up the loading process by
including the file at the top of the data block instead of letting NSIS seek
all the way down to the bottom  of the com pressed data block.

Use /plugin to reserve a plugin in ${NSISDIR}\Plugins\*.

See File for m ore inform ation about the param eters.

4.9.1.9 RM Dir

[/r] [/REBOOTOK] directory_name

Rem ove the specified directory (fully qualified path with no wildcards).
W ithout /r, the directory will only be rem oved if it is com pletely em pty. If /r
is specified the directory will be rem oved recursively, so all directories
and files in the specified directory will be rem oved. If /REBO O TO K is
specified, any file or directory which could not be rem oved during the
process will be rem oved on reboot -- if any file or directory will be
rem oved on a reboot, the reboot flag will be set. The error flag is set if
any file or directory cannot be rem oved.

RMDir $INSTDIR
RMDir $INSTDIR\data
RMDir /r /REBOOTOK $INSTDIR
RMDir /REBOOTOK $INSTDIR\DLLs



Note that the current working directory can not be deleted. The current
working directory is set by SetO utPath. For exam ple, the following
exam ple will not delete the directory.

SetOutPath $TEMP\dir
RMDir $TEMP\dir

The next exam ple will succeed in deleting the directory.

SetOutPath $TEMP\dir
SetOutPath $TEMP
RMDir $TEMP\dir

W arning: Using RM Dir /r $INSTDIR in the uninstaller is not safe.
Though it is unlikely, the user m ight select to install to the root of the
Program  Files folder and this com m and would wipe out the entire
Program  Files folder, including all other installed program s! The user
can also put other files in the installation folder and wouldn't expect
them  to get deleted along with the program . Solutions are available for
easily uninstalling only files which were installed by the installer.

W arning: The /REBO O TO K switch requires adm inistrator rights on
W indows NT and later.

4.9.1.10 SetO utPath

outpath

Sets the output path ($O UTDIR) and creates it (recursively if necessary),
if it does not exist. M ust be a full pathnam e, usually is just $INSTDIR (you
can specify $INSTDIR with a single "-" if you are lazy).

SetOutPath $INSTDIR
File program.exe

4.9.2 Registry, INI, File Instructions



In all of the below registry instructions use an em pty string (just two
quotes with nothing between them  - "") as the key nam e to specify the
default key which is shown as (Default) in regedit.exe.

Use SetRegView on 64-bit W indows to choose which registry view is
used.

If a full path is not specified for any of the INI handling instructions, the
W indows directory will be used.

4.9.2.1 DeleteINISec

ini_filename section_name

Deletes the entire section [section_nam e] from  ini_filenam e. If the section
could not be rem oved from  the ini file, the error flag is set. It does not set
the error flag if the section could not be found.

WriteINIStr $TEMP\something.ini section1 something 123
WriteINIStr $TEMP\something.ini section1 somethingelse 1234
WriteINIStr $TEMP\something.ini section2 nsis true
DeleteINISec $TEMP\something.ini section1

4.9.2.2 DeleteINIStr

ini_filename section_name str_name

Deletes the string str_nam e from  section [section_nam e] from
ini_filenam e. If the string could not be rem oved from  the ini file, the error
flag is set. It does not set the error flag if the string could not be found.

WriteINIStr $TEMP\something.ini section1 something 123
WriteINIStr $TEMP\something.ini section1 somethingelse 1234
DeleteINIStr $TEMP\something.ini section1 somethingelse

4.9.2.3 DeleteRegKey



[/ifempty] root_key subkey

Deletes a registry key. If /ifem pty is specified, the registry key will only be
deleted if it has no subkeys (otherwise, the whole registry tree will be
rem oved). Valid values for root_key are listed under W riteRegStr. The
error flag is set if the key could not be rem oved from  the registry (or if it
didn't exist to begin with).

DeleteRegKey HKLM "Software\My Company\My Software"
DeleteRegKey /ifempty HKLM "Software\A key that might have subkeys"

4.9.2.4 DeleteRegValue

root_key subkey key_name

Deletes a registry value. Valid values for root_key are listed under
W riteRegStr. The error flag is set if the value could not be rem oved from
the registry (or if it didn't exist to begin with).

DeleteRegValue HKLM "Software\My Company\My Software" "some value"

4.9.2.5 Enum RegKey

user_var(output) root_key subkey index

Set user variable $x with the nam e of the 'index'th registry key in
root_key\Subkey. Valid values for root_key are listed under W riteRegStr.
Returns an em pty string if there are no m ore keys, and returns an em pty
string and sets the error flag if there is an error.

StrCpy $0 0
loop:
  EnumRegKey $1 HKLM Software $0
  StrCmp $1 "" done
  IntOp $0 $0 + 1
  MessageBox MB_YESNO|MB_ICONQUESTION "$1$\n$\nMore?" IDYES loop
done:



4.9.2.6 Enum RegValue

user_var(output) root_key subkey index

Set user variable $x with the nam e of the 'index'th registry value in
root_key\Subkey. Valid values for root_key are listed under W riteRegStr.
Returns an em pty string and sets the error flag if there are no m ore
values or if there is an error.

StrCpy $0 0
loop:
  ClearErrors
  EnumRegValue $1 HKLM Software\Microsoft\Windows\CurrentVersion $0
  IfErrors done
  IntOp $0 $0 + 1
  ReadRegStr $2 HKLM Software\Microsoft\Windows\CurrentVersion $1
  MessageBox MB_YESNO|MB_ICONQUESTION "$1 = $2$\n$\nMore?" IDYES loop
done:

4.9.2.7 ExpandEnvStrings

user_var(output) string

Expands environm ent variables in string into the user variable $x. If an
environm ent variable doesn't exist, it will not be replaced. For exam ple, if
you use "% var% " and var doesn't exists, the result will be "% var% ". If
there is an error, the variable is set to em pty, and the error flag is set.

ExpandEnvStrings $0 "WINDIR=%WINDIR%$\nTEMP=%TEMP%"

4.9.2.8 FlushINI

ini_filename

Flushes the INI file's buffers. W indows 9x keeps all changes to the INI file
in m em ory. This com m and causes the changes to be written to the disk



im m ediately. Use it if you edit the INI m anually, delete it, m ove it or copy it
right after you change it with W riteINIStr, DeleteINISec or DeleteINStr.

WriteINIStr $TEMP\something.ini test test test
FlushINI $TEMP\something.ini
Delete $TEMP\something.ini

4.9.2.9 ReadEnvStr

user_var(output) name

Reads from  the environm ent string "nam e" and sets the value into the
user variable $x. If there is an error reading the string, the user variable is
set to em pty, and the error flag is set.

ReadEnvStr $0 WINDIR
ReadEnvStr $1 TEMP

4.9.2.10 ReadINIStr

user_var(output) ini_filename section_name entry_name

Reads from  entry_nam e in [section_nam e] of ini_filenam e and stores the
value into user variable $x. The error flag will be set and $x will be
assigned to an em pty string if the entry is not found.

ReadINIStr $0 $INSTDIR\winamp.ini winamp outname

4.9.2.11 ReadRegDW O RD

user_var(output) root_key sub_key name

Reads a 32-bit DW O RD from  the registry into the user variable $x. Valid
values for root_key are listed under W riteRegStr. The error flag will be
set and $x will be set to an em pty string ("" which is interpreted as 0 in
m ath operations) if the DW O RD is not present. If the value is present, but



is not a DW O RD, it will be read as a string and the error flag will be set.

ReadRegDWORD $0 HKLM Software\NSIS VersionBuild

4.9.2.12 ReadRegStr

user_var(output) root_key sub_key name

Reads from  the registry into the user variable $x. Valid values for
root_key are listed under W riteRegStr. The error flag will be set and $x
will be set to an em pty string ("") if the string is not present. If the value is
present, but is of type REG _DW O RD, it will be read and converted to a
string and the error flag will be set.

ReadRegStr $0 HKLM Software\NSIS ""
DetailPrint "NSIS is installed at: $0"

4.9.2.13 W riteINIStr

ini_filename section_name entry_name value

W rites entry_nam e=value into [section_nam e] of ini_filenam e. The error
flag is set if the string could not be written to the ini file.

WriteINIStr $TEMP\something.ini section1 something 123
WriteINIStr $TEMP\something.ini section1 somethingelse 1234
WriteINIStr $TEMP\something.ini section2 nsis true

4.9.2.14 W riteRegBin

root_key subkey key_name valuedata

This com m and writes a block of binary data to the registry. Valid values
for root_key are listed under W riteRegStr. Valuedata is in hexadecim al
(e.g. DEADBEEF01223211151). The error flag is set if the binary data
could not be written to the registry. If the registry key doesn't exist it will



be created.

WriteRegBin HKLM "Software\My Company\My Software" "Binary Value" DEADBEEF01223211151

4.9.2.15 W riteRegDW O RD

root_key subkey key_name value

This com m and writes a DW O RD (32-bit integer) to the registry (a user
variable can be specified). Valid values for root_key are listed under
W riteRegStr. The error flag is set if the dword could not be written to the
registry. If the registry key doesn't exist it will be created.

WriteRegDWORD HKLM "Software\My Company\My Software" "DWORD Value" 0xDEADBEEF

4.9.2.16 W riteRegStr

root_key subkey key_name value

W rite a string to the registry. See W riteRegExpandStr for m ore details.

WriteRegStr HKLM "Software\My Company\My Software" "String Value" "dead beef"

4.9.2.17 W riteRegExpandStr

root_key subkey key_name value

W rite a string to the registry. root_key m ust be one of:

HKCR or HKEY_CLASSES_RO O T
HKLM  or HKEY_LO CAL_M ACHINE
HKCU or HKEY_CURRENT_USER
HKU or HKEY_USERS
HKCC or HKEY_CURRENT_CO NFIG
HKDD or HKEY_DYN_DATA
HKPD or HKEY_PERFO RM ANCE_DATA



SHCTX or SHELL_CO NTEXT
HKCR32 or HKCR64
HKCU32 or HKCU64
HKLM 32 or HKLM 64

If root_key is SHCTX or SHELL_CO NTEXT, it will be replaced with
HKLM  if SetShellVarContext is set to all and with HKCU if
SetShellVarContext is set to current.

The error flag is set if the string could not be written to the registry. The
type of the string will be REG _SZ for W riteRegStr, or
REG _EXPAND_STR for W riteRegExpandStr. If the registry key doesn't
exist it will be created.

WriteRegExpandStr HKLM "Software\My Company\My Software" "Expand String Value" "%WINDIR%\notepad.exe"

4.9.2.18 W riteRegM ultiStr

/REGEDIT5 root_key subkey key_name value

W rites a m ulti-string value. The /REG EDIT5 switch m ust be used and
specifies that the data is in the hex form at used by .reg files on W indows
2000 and later.

WriteRegMultiStr /REGEDIT5 HKCU "Software\NSIS\Test" "Multi Value" 66,00,6f,00,6f,00,00,00,62,00,61,00,72,00,00,00,00,00

4.9.2.19 SetRegView

32|64|default|lastused

Sets the registry view affected by registry com m ands (root keys with a
32/64 suffix are not affected). O n 64-bit versions of W indows there are
two views; one for 32-bit applications and one for 64-bit applications. By
default, 32-bit applications running on 64-bit system s (W O W 64) only
have access to the 32-bit view. Using SetRegView 64 allows the installer to
access keys in the 64-bit view of the registry. Registry operations will fail
if the selected view is not supported by W indows.



Affects DeleteRegKey, DeleteRegValue, Enum RegKey, Enum RegValue,
ReadRegDW O RD, ReadRegStr, W riteRegBin, W riteRegDW O RD,
W riteRegStr and W riteRegExpandStr.

Does not affect InstallDirRegKey. Instead, the registry m ust be read
using ReadRegStr in .onInit.

SetRegView 32
ReadRegStr $0 HKLM Software\Microsoft\Windows\CurrentVersion ProgramFilesDir
DetailPrint $0 # prints C:\Program Files (x86)
!include x64.nsh
${If} ${RunningX64}
  SetRegView 64
  ReadRegStr $0 HKLM Software\Microsoft\Windows\CurrentVersion ProgramFilesDir
  DetailPrint $0 # prints C:\Program Files
${EndIf}

Function .onInit
  ${If} ${RunningX64}
    SetRegView 64
    ReadRegStr $INSTDIR HKLM Software\NSIS ""
    SetRegView Default
  ${EndIf}
FunctionEnd

4.9.3 G eneral Purpose Instructions

4.9.3.1 CallInstDLL

dllfile function_name

Calls a function nam ed function_nam e inside a NSIS extension DLL, a
plug-in. See the exam ple plugin for how to m ake one. Extension DLLs
can access the stack and variables. Note: To autom atically extract and
call plug-in DLLs, use a plug-in com m and instead of CallInstDLL.

Push "a parameter"



Push "another parameter"
CallInstDLL $INSTDIR\somedll.dll somefunction

For easier plug-in handling, use the new plug-in call syntax.

4.9.3.2 CopyFiles

[/SILENT] [/FILESONLY] filespec_on_destsys destination_path [size_of_files_in_kb]

Copies files from  the source to the destination on the installing system .
Useful with $EXEDIR if you want to copy from  installation m edia, or to
copy from  one place to another on the system . You m ight see a W indows
status window of the copy operation if the operation takes a lot of tim e (to
disable this, use /SILENT). The last param eter can be used to specify the
size of the files that will be copied (in kilobytes), so that the installer can
approxim ate the disk space requirem ents. O n error, or if the user cancels
the copy (only possible when /SILENT was om itted), the error flag is set.
If /FILESO NLY is specified, only files are copied.

Fully-qualified path nam es should always be used with this instruction.
Using relative paths will have unpredictable results.

CreateDirectory $INSTDIR\backup
CopyFiles $INSTDIR\*.dat $INSTDIR\backup

4.9.3.3 CreateDirectory

path_to_create

Creates (recursively if necessary) the specified directory. The error flag is
set if the directory couldn't be created.

You should always specify an absolute path.

CreateDirectory $INSTDIR\some\directory

4.9.3.4 CreateShortcut



[/NoWorkingDir] link.lnk target.file [parameters [icon.file [icon_index_number [start_options [keyboard_shortcut [description]]]]]]

Creates a shortcut 'link.lnk' that links to 'target.file', with optional
param eters 'param eters'. The icon used for the shortcut is
'icon.file,icon_index_num ber'; for default icon settings use em pty strings
for both icon.file and icon_index_num ber. start_options should be one of:
SW _SHO W NO RM AL, SW _SHO W M AXIM IZED, SW _SHO W M INIM IZED,
or an em pty string. keyboard_shortcut should be in the form  of 'flag|c'
where flag can be a com bination (using |) of: ALT, CO NTRO L, EXT, or
SHIFT. c is the character to use (a-z, A-Z, 0-9, F1-F24, etc). Note that no
spaces are allowed in this string. A good exam ple is "ALT|CO NTRO L|F8".
$O UTDIR is used as the working directory. You can change it by using
SetO utPath before creating the Shortcut or use /NoW orkingDir if you
don't need to set the working directory. description should be the
description of the shortcut, or com m ent as it is called under XP. The error
flag is set if the shortcut cannot be created (i.e. either of the paths (link or
target) does not exist, or som e other error).

CreateDirectory "$SMPROGRAMS\My Company"
CreateShortcut "$SMPROGRAMS\My Company\My Program.lnk" "$INSTDIR\My Program.exe" \
  "some command line parameters" "$INSTDIR\My Program.exe" 2 SW_SHOWNORMAL \
  ALT|CONTROL|SHIFT|F5 "a description"

4.9.3.5 G etDLLVersion

filename user_var(high dword output) user_var(low dword output)

G ets the version inform ation from  the DLL (or any other executable
containing version inform ation) in "filenam e". Sets the user output
variables with the high and low dwords of version inform ation on success;
on failure the outputs are em pty and the error flag is set. The following
exam ple reads the DLL version and copies a hum an readable version of
it into $0:

GetDllVersion "$INSTDIR\MyDLL.dll" $R0 $R1
IntOp $R2 $R0 / 0x00010000
IntOp $R3 $R0 & 0x0000FFFF
IntOp $R4 $R1 / 0x00010000



IntOp $R5 $R1 & 0x0000FFFF
StrCpy $0 "$R2.$R3.$R4.$R5"

4.9.3.6 G etDLLVersionLocal

localfilename user_var(high dword output) user_var(low dword output)

This is sim ilar to G etDLLVersion, only it acts on the system  building the
installer (it actually com piles into two StrCpy com m ands). Sets the two
output variables with the DLL version inform ation of the DLL on the build
system . Use !getdllversion if you need to use the values with
VIProductVersion.

4.9.3.7 G etFileTim e

filename user_var(high dword output) user_var(low dword output)

G ets the last write tim e of "filenam e". Sets the user output variables with
the high and low dwords of the FILETIM E tim estam p on success; on
failure the outputs are em pty and the error flag is set.

4.9.3.8 G etFileTim eLocal

localfilename user_var(high dword output) user_var(low dword output)

This is sim ilar to G etFileTim e, only it acts on the system  building the
installer (it actually com piles into two StrCpy com m ands). Sets the two
output variables with the file tim estam p of the file on the build system .

4.9.3.9 G etFullPathNam e

[/SHORT] user_var(output) path_or_file

Assign the full path of the file specified to user variable $x. If the path
portion of the param eter is not found, the error flag will be set and $x will



be em pty. If /SHO RT is specified, the path is converted to the short
filenam e form . However, if /SHO RT is not specified, the path isn't
converted to its long filenam e form . To get the long filenam e, call
G etLongPathNam e using the System  plug-in. Note that
G etLongPathNam e is only available on W indows 98, W indows 2000 and
above.

StrCpy $INSTDIR $PROGRAMFILES\NSIS
SetOutPath $INSTDIR
GetFullPathName $0 ..
DetailPrint $0 # will print C:\Program Files
GetFullPathName /SHORT $0 $INSTDIR
DetailPrint $0 # will print C:\Progra~1\NSIS

StrCpy $0 C:\Progra~1\NSIS
System::Call 'kernel32::GetLongPathName(t r0, t .r1, i ${NSIS_MAX_STRLEN}) i .r2'
StrCmp $2 error +2
StrCpy $0 $1
DetailPrint $0 # will print C:\Program Files\NSIS, where supported

4.9.3.10 G etTem pFileNam e

user_var(output) [base_dir]

Assign to the user variable $x, the nam e of a tem porary file. The file will
be created for you and it will be em pty. The nam e of the tem porary file is
guaranteed to be unique. If to want the tem porary file to be created in
another directory other than the W indows tem p directory, specify a
base_dir. You should Delete the file when you are done with it.

GetTempFileName $0
File /oname=$0 something.dat
# do something with something.dat
Delete $0

4.9.3.11 SearchPath



user_var(output) filename

Assign to the user variable $x, the full path of the file nam ed by the
second param eter. The error flag will be set and $x will be em pty if the
file cannot be found. Uses SearchPath() to search the system  paths for
the file.

4.9.3.12 SetFileAttributes

filename attribute1|attribute2|...

Sets the file attributes of 'filenam e'. Valid attributes can be com bined with
| and are:

NO RM AL or FILE_ATTRIBUTE_NO RM AL (you can use 0 to
abbreviate this)
ARCHIVE or FILE_ATTRIBUTE_ARCHIVE
HIDDEN or FILE_ATTRIBUTE_HIDDEN
O FFLINE or FILE_ATTRIBUTE_O FFLINE
READO NLY or FILE_ATTRIBUTE_READO NLY
SYSTEM  or FILE_ATTRIBUTE_SYSTEM
TEM PO RARY or FILE_ATTRIBUTE_TEM PO RARY

The error flag will be set if the file's attributes cannot be set (i.e. the file
doesn't exist, or you don't have the right perm issions). You can only set
attributes. It's not possible to unset them . If you want to rem ove an
attribute use NO RM AL. This way all attributes are erased. This com m and
doesn't support wildcards.

4.9.3.13 RegDLL

dllfile [entrypoint_name]

Loads the specified DLL and calls DllRegisterServer (or entrypoint_nam e
if specified). The error flag is set if an error occurs (i.e. it can't load the
DLL, initialize O LE, find the entry point, or the function returned anything
other than ERRO R_SUCCESS (=0)).



Use SetO utPath to set the current directory for DLLs that depend on
other DLLs that are now in the path or in the W indows directory. For
exam ple, if foo.dll depends on bar.dll which is located in $INSTDIR use:

 SetOutPath $INSTDIR
 RegDLL $INSTDIR\foo.dll

4.9.3.14 UnRegDLL

dllfile

Loads the specified DLL and calls DllUnregisterServer. The error flag is
set if an error occurs (i.e. it can't load the DLL, initialize O LE, find the
entry point, or the function returned anything other than
ERRO R_SUCCESS (=0)).

4.9.4 Flow  Control Instructions

4.9.4.1 Abort

[user_message]

Cancels the install, stops execution of script, and displays user_m essage
in the status display. Note: you can use this from  Callback functions to do
special things. Page callbacks also uses Abort for special purposes.

Abort
Abort "can't install"

4.9.4.2 Call

function_name | :label_name | user_var(input)

Calls the function nam ed function_nam e, the label nam ed label_nam e, or
a variable that specifies an address. An address is returned by
G etCurrentAddress, G etFunctionAddress or G etLabelAddress. A call



returns when it encounters a Return instruction. Sections and functions
are autom atically ended with a Return instruction. Uninstall functions
cannot be called from  installer functions and sections, and vice-versa.

Function func
  Call :label
  DetailPrint "#1: This will only appear 1 time."
label:
  DetailPrint "#2: This will appear before and after message #1."
  Call :.global_label
FunctionEnd

Section
  Call func
  Return

.global_label:
  DetailPrint "#3: The global label was called"
SectionEnd

4.9.4.3 ClearErrors

Clears the error flag.

ClearErrors
IfErrors 0 +2
  MessageBox MB_OK "this message box will never show"

4.9.4.4 G etCurrentAddress

user_var(output)

G ets the address of the current instruction (the G etCurrentAddress) and
stores it in the output user variable. This user variable then can be
passed to Call or G oto.



Function func
  DetailPrint "function"
  IntOp $0 $0 + 2 ; Calculate the address after of the instruction after "Goto callFunc" in the Section
  Call $0
  DetailPrint "function end"
FunctionEnd

Section
  DetailPrint "section"
  GetCurrentAddress $0
  Goto callFunc

  DetailPrint "back in section"
  Return

callFunc:
  Call func
  DetailPrint "section end"
SectionEnd

4.9.4.5 G etFunctionAddress

user_var(output) function_name

G ets the address of the function and stores it in the output user variable.
This user variable then can be passed to Call or G oto. Note that if you
G oto an address which is the output of G etFunctionAddress, your
function will never be returned to (when the function you G oto'd to
returns, you return instantly).

Function func
  DetailPrint "function"
FunctionEnd

Section
  GetFunctionAddress $0 func
  Call $0



SectionEnd

4.9.4.6 G etLabelAddress

user_var(output) label

G ets the address of the label and stores it in the output user variable.
This user variable then can be passed to Call or G oto. Note that you m ay
only call this with labels accessible from  your function, but you can call it
from  anywhere (which is potentially dangerous). Note that if you Call the
output of G etLabelAddress, code will be executed until it Return's
(explicitly or im plicitly at the end of a function), and then you will be
returned to the statem ent after the Call.

label:
DetailPrint "label"
GetLabelAddress $0 label
IntOp $0 $0 + 4
Goto $0
DetailPrint "done"

4.9.4.7 G oto

label_to_jump_to | +offset| -offset| user_var(target)

If label is specified, goto the label 'label_to_jum p_to:'.

If +offset or -offset is specified, jum p is relative by offset instructions.
G oto +1 goes to the next instruction, G oto -1 goes to the previous
instruction, etc.

If a user variable is specified, jum ps to absolute address (generally you
will want to get this value from  a function like G etLabelAddress).
Com piler flag com m ands and SectionIn aren't instructions so jum ping
over them  has no effect.

Goto label



Goto +2
Goto -2
Goto $0

4.9.4.8 IfAbort

label_to_goto_if_abort [label_to_goto_if_no_abort]

W ill "return" true if the installation has been aborted. This can happen if
the user chose abort on a file that failed to create (or overwrite) or if the
user aborted by hand. This function can only be called from  the leave
function of the instfiles page.

Page instfiles "" "" instfilesLeave

Function instfilesLeave
  IfAbort 0 +2
    MessageBox MB_OK "user aborted"
FunctionEnd

4.9.4.9 IfErrors

jumpto_iferror [jumpto_ifnoerror]

Checks and clears the error flag, and if it is set, it will goto jum pto_iferror,
otherwise it will goto jum pto_ifnoerror. The error flag is set by other
instructions when a recoverable error (such as trying to delete a file that
is in use) occurs.

ClearErrors
File file.dat
IfErrors 0 +2
  Call ErrorHandler

4.9.4.10 IfFileExists



file_to_check_for jump_if_present [jump_otherwise]

Checks for existence of file(s) file_to_check_for (which can be a wildcard,
or a directory), and G otos jum p_if_present if the file exists, otherwise
G otos jum p_otherwise. If you want to check to see if a file is a directory,
use IfFileExists DIRECTO RY\*.*

IfFileExists $WINDIR\notepad.exe 0 +2
  MessageBox MB_OK "notepad is installed"

4.9.4.11 IfRebootFlag

jump_if_set [jump_if_not_set]

Checks the reboot flag, and jum ps to jum p_if_set if the reboot flag is set,
otherwise jum ps to jum p_if_not_set. The reboot flag can be set by Delete
and Renam e, or m anually with SetRebootFlag.

IfRebootFlag 0 noreboot
  MessageBox MB_YESNO "A reboot is required to finish the installation. Do you wish to reboot now?" IDNO noreboot
    Reboot
noreboot:

4.9.4.12 IfSilent

jump_if_silent [jump_if_not]

Checks the silent flag, and jum ps to jum p_if_silent if the installer is silent,
otherwise jum ps to jum p_if_not. The silent flag can be set by SilentInstall,
SilentUninstall, SetSilent and by the user passing /S on the com m and
line.

IfSilent +2
  ExecWait '"$INSTDIR\nonsilentprogram.exe"'

4.9.4.13 IntCm p



val1 val2 jump_if_equal [jump_if_val1_less] [jump_if_val1_more]

Com pares two integers val1 and val2. If val1 and val2 are equal, G otos
jum p_if_equal, otherwise if val1 < val2, G otos jum p_if_val1_less,
otherwise if val1 > val2, G otos jum p_if_val1_m ore.

IntCmp $0 5 is5 lessthan5 morethan5
is5:
  DetailPrint "$$0 == 5"
  Goto done
lessthan5:
  DetailPrint "$$0 < 5"
  Goto done
morethan5:
  DetailPrint "$$0 > 5"
  Goto done
done:

4.9.4.14 IntCm pU

val1 val2 jump_if_equal [jump_if_val1_less] [jump_if_val1_more]

Sam e as IntCm p, but treats the values as unsigned integers.

4.9.4.15 M essageBox

mb_option_list messagebox_text [/SD return] [return_check jumpto [return_check_2 jumpto_2]]

Displays a M essageBox containing the text "m essagebox_text".
m b_option_list m ust be one or m ore of the following, delim ited by |s (e.g.
M B_YESNO |M B_ICO NSTO P).

M B_O K - Display with an O K button
M B_O KCANCEL - Display with an O K and a cancel button
M B_ABO RTRETRYIG NO RE - Display with abort, retry, ignore
buttons
M B_RETRYCANCEL - Display with retry and cancel buttons



M B_YESNO  - Display with yes and no buttons
M B_YESNO CANCEL - Display with yes, no, cancel buttons
M B_ICO NEXCLAM ATIO N - Display with exclam ation icon
M B_ICO NINFO RM ATIO N - Display with inform ation icon
M B_ICO NQ UESTIO N - Display with question m ark icon
M B_ICO NSTO P - Display with stop icon
M B_USERICO N - Display with installer's icon
M B_TO PM O ST - M ake m essagebox topm ost
M B_SETFO REG RO UND - Set foreground
M B_RIG HT - Right align text
M B_RTLREADING  - RTL reading order
M B_DEFBUTTO N1 - Button 1 is default
M B_DEFBUTTO N2 - Button 2 is default
M B_DEFBUTTO N3 - Button 3 is default
M B_DEFBUTTO N4 - Button 4 is default

Return_check can be 0 (or em pty, or left off), or one of the following:

IDABO RT - Abort button
IDCANCEL - Cancel button
IDIG NO RE - Ignore button
IDNO  - No button
IDO K - O K button
IDRETRY - Retry button
IDYES - Yes button

If the return value of the M essageBox is return_check, the installer will
G oto jum pto.

Use the /SD param eter with one of the return_check values above to
specify the option that will be used when the installer is silent. See
section 4.12 for m ore inform ation.

MessageBox MB_OK "simple message box"
MessageBox MB_YESNO "is it true?" IDYES true IDNO false
true:
  DetailPrint "it's true!"
  Goto next
false:



  DetailPrint "it's false"
next:
MessageBox MB_YESNO "is it true? (defaults to yes on silent installations)" /SD IDYES IDNO false2
  DetailPrint "it's true (or silent)!"
  Goto next2
false2:
  DetailPrint "it's false"
next2:

4.9.4.16 Return

Returns from  a function or section.

Function func
  StrCmp $0 "return now" 0 +2
    Return
  # do stuff
FunctionEnd

Section
  Call func
  ;"Return" will return here
SectionEnd

4.9.4.17 Q uit

Causes the installer to exit as soon as possible. After Q uit is called, the
installer will exit (no callback functions will get a chance to run).

4.9.4.18 SetErrors

Sets the error flag.

SetErrors
IfErrors 0 +2



  MessageBox MB_OK "this message box will always show"

4.9.4.19 StrCm p

str1 str2 jump_if_equal [jump_if_not_equal]

Com pares (case insensitively) str1 to str2. If str1 and str2 are equal,
G otos jum p_if_equal, otherwise G otos jum p_if_not_equal.

StrCmp $0 "a string" 0 +3
  DetailPrint '$$0 == "a string"'
  Goto +2
  DetailPrint '$$0 != "a string"'

4.9.4.20 StrCm pS

str1 str2 jump_if_equal [jump_if_not_equal]

Sam e as StrCm p, but case sensitive.

4.9.5 File Instructions

4.9.5.1 FileClose

handle

Closes a file handle opened with FileO pen.

4.9.5.2 FileO pen

user_var(handle output) filename openmode

O pens a file nam ed "filenam e" and sets the handle output variable with
the handle. The openm ode should be one of "r" (read) "w" (write, all
contents of file are destroyed) or "a" (append, m eaning opened for both



read and write, contents preserved). In all open m odes, the file pointer is
placed at the beginning of the file. If the file cannot be opened the handle
output is set to em pty and the error flag is set.

If no absolute path is specified the current folder will be used. The current
folder is the folder set using the last SetO utPath instruction. If you have
not used SetO utPath the current folder is $EXEDIR.

FileOpen $0 $INSTDIR\file.dat r
FileClose $0

4.9.5.3 FileRead

handle user_var(output) [maxlen]

Reads a string (ANSI characters) from  a file opened with FileO pen. The
string is read until either a newline (or carriage return newline pair)
occurs, or until a null byte is read, or until m axlen is m et (if specified). By
default, strings are lim ited to 1024 characters (a special build with larger
NSIS_M AX_STRLEN can be com piled or downloaded). If the end of file
is reached and no m ore data is available, the output string will be em pty
and the error flag will be set.

Unicode: DBCS text is supported but conversion output is lim ited to
UCS-2/BM P, surrogate pairs are not supported. The system  default
ANSI codepage (ACP) is used during the conversion.

ClearErrors
FileOpen $0 $INSTDIR\file.dat r
IfErrors done
FileRead $0 $1
DetailPrint $1
FileClose $0
done:

4.9.5.4 FileReadUTF16LE



handle user_var(output) [maxlen]

This function is only available when building a Unicode installer.

Reads a string (UTF-16LE characters) from  a file opened with FileO pen.
The string is read until either a newline (or carriage return newline pair)
occurs, or until a null wide-character is read, or until m axlen is m et (if
specified). By default, strings are lim ited to 1024 characters (a special
build with larger NSIS_M AX_STRLEN can be com piled or downloaded).
If the end of file is reached and no m ore data is available, the output
string will be em pty and the error flag will be set. If present, the BO M  at
the start of the file is skipped.

ClearErrors
FileOpen $0 $INSTDIR\file.dat r
IfErrors done
FileReadUTF16LE $0 $1
DetailPrint $1
FileClose $0
done:

4.9.5.5 FileReadByte

handle user_var(output)

Reads a byte from  a file opened with FileO pen. The byte is stored in the
output as an integer (0-255). If the end of file is reached and no m ore
data is available, the output will be em pty and the error flag will be set.

ClearErrors
FileOpen $0 $INSTDIR\file.dat r
IfErrors done
FileReadByte $0 $1
FileReadByte $0 $2
DetailPrint "$1 $2"
FileClose $0
done:



4.9.5.6 FileReadW ord

handle user_var(output)

This function is only available when building a Unicode installer.

Reads a word (2-bytes) from  a file opened with FileO pen. The word is
stored in the output as an integer (0-65535). If the end of file is reached
and no m ore data is available, the output will be em pty and the error flag
will be set.

ClearErrors
FileOpen $0 $INSTDIR\file.dat r
IfErrors done
FileReadWord $0 $1
FileReadWord $0 $2
DetailPrint "$1 $2"
FileClose $0
done:

4.9.5.7 FileSeek

handle offset [mode] [user_var(new position)]

Seeks a file opened with FileO pen. If m ode is om itted or specified as
SET, the file is positioned to "offset", relative to the beginning of the file. If
m ode is specified as CUR, then the file is positioned to "offset", relative to
the current file position. If m ode is specified as END, then the file is
positioned to "offset", relative to the end of the file. If the final param eter
"new position" is specified, the new file position will be stored in that
variable.

ClearErrors
FileOpen $0 $INSTDIR\file.dat r
IfErrors done
FileSeek $0 -5 END
FileRead $0 $1



DetailPrint $1
FileClose $0
done:

4.9.5.8 FileW rite

handle string

W rites an ANSI string to a file opened with FileO pen. If an error occurs
writing, the error flag will be set.

(If you are building a Unicode installer, the function converts the string to
ANSI/M BCS. The system  default ANSI codepage (ACP) is used during
the conversion)

ClearErrors
FileOpen $0 $INSTDIR\file.dat w
IfErrors done
FileWrite $0 "some text"
FileClose $0
done:

4.9.5.9 FileW riteUTF16LE

[/BOM] handle string

This function is only available when building a Unicode installer.

W rites a Unicode (UTF-16LE) string to a file opened with FileO pen. If an
error occurs, the error flag will be set. A BO M  can be added to em pty files
with /BO M .

ClearErrors
FileOpen $0 $INSTDIR\file.dat w
IfErrors done
FileWriteUTF16LE $0 "some text"
FileClose $0



done:

4.9.5.10 FileW riteByte

handle string

W rites the integer interpretation of 'string' to a file opened with FileO pen.
The error flag is set if an error occurs while writing. The following code
writes a "Carriage Return / Line Feed" pair to the file.

FileWriteByte file_handle "13"
FileWriteByte file_handle "10"

Note that only the low byte of the integer is used, i.e. writing 256 is the
sam e as writing 0, etc.

4.9.5.11 FileW riteW ord

handle string

This function is only available when building a Unicode installer.

W rites the integer interpretation of 'string' as a W O RD (2-bytes, range: 0-
65535) to a file opened with FileO pen. The error flag is set if an error
occurs while writing. The following code writes a "Carriage Return / Line
Feed" pair to the file.

FileWriteWord file_handle "13"
FileWriteWord file_handle "10"

Note that only the low W O RD of the integer is used, i.e. writing 65536 is
the sam e as writing 0, etc.

4.9.5.12 FindClose

handle



Closes a search opened with FindFirst.

4.9.5.13 FindFirst

user_var(handle output) user_var(filename output) filespec

Perform s a search for 'filespec', placing the first file found in
filenam e_output (a user variable). It also puts the handle of the search
into handle_output (also a user variable). If no files are found, both
outputs are set to em pty and the error flag is set. FindClose m ust be
used to close the handle. Note that the filenam e output is without path.

FindFirst $0 $1 $INSTDIR\*.txt
loop:
  StrCmp $1 "" done
  DetailPrint $1
  FindNext $0 $1
  Goto loop
done:
FindClose $0

4.9.5.14 FindNext

handle user_var(filename_output)

Continues a search began with FindFirst. handle should be the
handle_output_variable returned by FindFirst. If the search is com pleted
(there are no m ore files), filenam e_output is set to em pty and the error
flag is set. Note that the filenam e output is without path.

4.9.6 Uninstaller Instructions

4.9.6.1 W riteUninstaller

[Path\]exename.exe



W rites the uninstaller to the filenam e (and optionally path) specified. O nly
valid from  within an install section or function and requires that you have
an uninstall section in your script. You can call this one or m ore tim es to
write out one or m ore copies of the uninstaller.

WriteUninstaller $INSTDIR\uninstaller.exe

4.9.7 M iscellaneous Instructions

4.9.7.1 G etErrorLevel

user_var(error level output)

Returns the last error level set by SetErrorLevel or -1 if it has never been
set.

GetErrorLevel $0
IntOp $0 $0 + 1
SetErrorLevel $0

4.9.7.2 G etInstDirError

user_var(error output)

Use in the leave function of a directory page. Reads the flag set if
'DirVerify leave' is used. Possible values:

0: No error

1: Invalid installation directory

2: Not enough space on installation drive

!include LogicLib.nsh
PageEx directory
  DirVerify leave
  PageCallbacks "" "" dirLeave



PageExEnd

Function dirLeave
  GetInstDirError $0
  ${Switch} $0
    ${Case} 0
      MessageBox MB_OK "valid installation directory"
      ${Break}
    ${Case} 1
      MessageBox MB_OK "invalid installation directory!"
      Abort
      ${Break}
    ${Case} 2
      MessageBox MB_OK "not enough free space!"
      Abort
      ${Break}
  ${EndSwitch}
FunctionEnd

4.9.7.3 InitPluginsDir

Initializes the plug-ins dir ($PLUG INSDIR) if not already initialized.

InitPluginsDir
File /oname=$PLUGINSDIR\image.bmp image.bmp

4.9.7.4 Nop

Does nothing.

4.9.7.5 SetErrorLevel

error_level

Sets the error level of the installer or uninstaller to error_level. See Error



Levels for m ore inform ation.

IfRebootFlag 0 +2
  SetErrorLevel 4

4.9.7.6 SetShellVarContext

current|all

Sets the context of $SM PRO G RAM S and other shell folders. If set to
'current' (the default), the current user's shell folders are used. If set to
'all', the 'all users' shell folder is used. The all users folder m ay not be
supported on all O Ses. If the all users folder is not found, the current user
folder will be used. Please take into consideration that a "norm al user"
has no rights to write in the all users area. O nly adm ins have full access
rights to the all users area. You can check this by using the UserInfo
plug-in. See Contrib\UserInfo\UserInfo.nsi for an exam ple.

Note that, if used in installer code, this will only affect the installer, and if
used in uninstaller code, this will only affect the uninstaller. To affect both,
it needs to be used in both.

SetShellVarContext current
StrCpy $0 $DESKTOP
SetShellVarContext all
StrCpy $1 $DESKTOP
MessageBox MB_OK $0$\n$1

4.9.7.7 Sleep

sleeptime_in_ms

Pauses execution in the installer for sleeptim e_in_m s m illiseconds.
sleeptim e_in_m s can be a variable, e.g. "$0" or a num ber, i.e. "666".

DetailPrint "sleeping..."
Sleep 3000



DetailPrint "back to work"

4.9.8 String M anipulation Instructions

4.9.8.1 StrCpy

user_var(destination) str [maxlen] [start_offset]

Sets the user variable $x with str. str can contain variables (including the
user variable being set (concatenating strings this way is possible, etc)).
If m axlen is specified, the string will be a m axim um  of m axlen characters
(if m axlen is negative, the string will be truncated abs(m axlen) characters
from  the end). If start_offset is specified, the source is offset by it (if
start_offset is negative, it will start abs(start_offset) from  the end of the
string).

StrCpy $0 "a string" # = "a string"
StrCpy $0 "a string" 3 # = "a s"
StrCpy $0 "a string" -1 # = "a strin"
StrCpy $0 "a string" "" 2 # = "string"
StrCpy $0 "a string" "" -3 # = "ing"
StrCpy $0 "a string" 3 -4 # = "rin"
StrCpy $0 "$0$0" # = "rinrin"

4.9.8.2 StrLen

user_var(length output) str

Sets user variable $x to the length of str.

StrLen $0 "123456" # = 6

4.9.9 Stack Support

4.9.9.1 Exch



[user_var | stack_index]

W hen no param eter is specified, exchanges the top two elem ents of the
stack. W hen a param eter is specified and is a user variable, exchanges
the top elem ent of the stack with the param eter. W hen a param eter is
specified and is a positive integer, Exch will swap the item  on the top of
the stack with the item  that is specified by the offset from  the top of the
stack in the param eter. If there are not enough item s on the stack to
accom plish the exchange, a fatal error will occur (to help you debug your
code :).

Push 1
Push 2
Exch
Pop $0 # = 1

Push 1
Push 2
Push 3
Exch 2
Pop $0 # = 1

StrCpy $0 1
Push 2
Exch $0 # = 2
Pop $1 # = 1

4.9.9.2 Pop

user_var(out)

Pops a string off of the stack into user variable $x. If the stack is em pty,
the error flag will be set.

Push 1
Pop $0 # = 1



4.9.9.3 Push

string

Pushes a string onto the stack. The string can then be Pop'ed off of the
stack.

Push "a string"

4.9.10 Integer Support

4.9.10.1 IntFm t

user_var(output) format numberstring

Form ats the num ber in "num berstring" using the form at "form at", and sets
the output to user variable $x. Exam ple form at strings include "% 08X"
"% u"

IntFmt $0 "0x%08X" 195948557
IntFmt $0 "%c" 0x41

4.9.10.2 IntO p

user_var(output) value1 OP [value2]

Com bines value1 and (depending on O P) value2 into the specified user
variable (user_var). O P is defined as one of the following:

+ ADDs value1 and value2
- SUBTRACTs value2 from  value1
* M ULTIPLIEs value1 and value2
/ DIVIDEs value1 by value2
%  M O DULUSs value1 by value2
| BINARY O Rs value1 and value2
& BINARY ANDs value1 and value2
 ̂BIN ARY XO Rs value1 and value2



>> RIG HT SHIFTs value1 by value2
<< LEFT SHIFTs value1 by value2
~ BITW ISE NEG ATEs value1 (i.e. 7 becom es 4294967288)
! LO G ICALLY NEG ATEs value1 (i.e. 7 becom es 0)
|| LO G ICALLY O Rs value1 and value2
&& LO G ICALLY ANDs value1 and value2

IntOp $0 1 + 1
IntOp $0 $0 + 1
IntOp $0 $0 << 2
IntOp $0 $0 ~
IntOp $0 $0 & 0xF

4.9.11 Reboot Instructions

4.9.11.1 Reboot

Reboots the com puter. Be careful with this one. If it fails, .onRebootFailed
is called. In any case, this instruction never returns, just like Q uit.

MessageBox MB_YESNO|MB_ICONQUESTION "Do you wish to reboot the system?" IDNO +2
  Reboot

4.9.11.2 SetRebootFlag

true|false

Sets the reboot flag to either true or false. The flag's value can be read
using IfRebootFlag.

SetRebootFlag true
IfRebootFlag 0 +2
  MessageBox MB_OK "this message box will always show"

4.9.12 Install Logging Instructions



4.9.12.1 LogSet

on|off

Sets whether install logging to $INSTDIR\install.log will happen.
$INSTDIR m ust have a value before you call this function or it will not
work. Note that the NSIS_CO NFIG _LO G  build setting m ust be set (scons
NSIS_CONFIG_LOG=yes) when building (it is not set by default) to support
this. See Building NSIS for m ore inform ation about recom piling NSIS.

4.9.12.2 LogText

text

If installer logging is enabled, inserts text "text" into the log file.

IfFileExists $WINDIR\notepad.exe 0 +2
  LogText "$$WINDIR\notepad.exe exists"

4.9.13 Section M anagem ent

4.9.13.1 SectionSetFlags

section_index section_flags

Sets the section's flags. The flag is a 32-bit integer. The first bit (lowest)
represents whether the section is currently selected, the second bit
represents whether the section is a section group (don't m odify this
unless you really know what you are doing), the third bit represents
whether the section is a section group end (again, don't m odify), the
fourth bit represents whether the section is shown in bold or not, the fifth
bit represents whether the section is read-only, the sixth bit represents
whether the section group is to be autom atically expanded, the seventh
bit is set for section groups which are partially selected, the eighth bit is
internally used for partially selected section group toggling and the ninth
bit is used for reflecting section nam e changes. The error flag will be set



if an out of range section is specified.

Each flag has a nam e, prefixed with ̀SF_ :̀

!define SF_SELECTED   1
!define SF_SECGRP     2
!define SF_SECGRPEND  4
!define SF_BOLD       8
!define SF_RO         16
!define SF_EXPAND     32
!define SF_PSELECTED  64

For an exam ple of usage please see the one-section.nsi exam ple.

For m ore useful m acros and definitions, see Include\Sections.nsh.

Section test test_section_id
SectionEnd

Function .onInit
  # set section 'test' as selected and read-only
  IntOp $0 ${SF_SELECTED} | ${SF_RO}
  SectionSetFlags ${test_section_id} $0
FunctionEnd

4.9.13.2 SectionG etFlags

section_index user_var(output)

Retrieves the section's flags. See SectionSetFlags for a description of the
flags. The error flag will be set if an out of range section is specified.

Section test test_section_id
SectionEnd

Function .onSelChange
  # keep section 'test' selected
  SectionGetFlags ${test_section_id} $0



  IntOp $0 $0 | ${SF_SELECTED}
  SectionSetFlags ${test_section_id} $0
FunctionEnd

4.9.13.3 SectionSetText

section_index section_text

Sets the description for the section section_index. If the text is set to ""
then the section will be hidden. The error flag will be set if an out of range
section is specified.

Section "" test_section_id
SectionEnd

Function .onInit
  # change section's name to $WINDIR
  SectionSetText ${test_section_id} $WINDIR
FunctionEnd

4.9.13.4 SectionG etText

section_index user_var(output)

Stores the text description of the section section_index into the output. If
the section is hidden, stores an em pty string. The error flag will be set if
an out of range section is specified.

Section test test_section_id
SectionEnd

Function .onInit
  # append $WINDIR to section's name
  SectionGetText ${test_section_id} $0
  StrCpy $0 "$0 - $WINDIR"
  SectionSetText ${test_section_id} $0



FunctionEnd

4.9.13.5 SectionSetInstTypes

section_index inst_types

Sets the install types the section specified by section_index defaults to
the enabled state in. Note that the section index starts with zero. Every
bit of inst_types is a flag that tells if the section is in that install type or
not. For exam ple, if you have 3 install types and you want the first section
to be included in install types 1 and 3, then the com m and should look like
this:

SectionSetInstTypes 0 5

because the binary value for 5 is "...00101". The error flag will be set if
the section index specified is out of range.

Section test test_section_id
SectionEnd

Function .onInit
  # associate section 'test' with installation types 3 and 4
  SectionSetInstTypes ${test_section_id} 12
FunctionEnd

4.9.13.6 SectionG etInstTypes

section_index user_var(output)

Retrieves the install types flags array of a section. See above explanation
about SectionSetInstTypes for a description of how to deal with the
output. The error flag will be set if the section index is out of range.

Section test test_section_id
SectionEnd



Function .onInit
  # associate section 'test' with installation types 5, on top of its existing associations
  SectionGetInstTypes ${test_section_id} $0
  IntOp $0 $0 | 16
  SectionSetInstTypes ${test_section_id} $0
FunctionEnd

4.9.13.7 SectionSetSize

section_index new_size

Sets the size of the section specified by section_index. Note that the
index starts with zero. The Value for Size m ust be entered in KiloByte
and supports only whole num bers.

Section test test_section_id
SectionEnd

Function .onInit
  # set required size of section 'test' to 100 bytes
  SectionSetSize ${test_section_id} 100
FunctionEnd

4.9.13.8 SectionG etSize

section_index user_var

G ets the size of the section specified by section_index and stores the
value in the given user variable. Note that the index starts with zero. The
error flag will be set if the section index is out of range.

Section test test_section_id
SectionEnd

Function .onInit
  # increase required size of section 'test' by 100 KiB



  SectionGetSize ${test_section_id} $0
  IntOp $0 $0 + 100
  SectionSetSize ${test_section_id} $0
FunctionEnd

4.9.13.9 SetCurInstType

inst_type_idx

Sets the current InstType. inst_type_idx should be between 0 and 31.
The error flag is not set if an out of range InstType was used.

4.9.13.10 G etCurInstType

user_var

G et the current InstType and stores it in user_var. If the first install type is
selected, 0 will be put in user_var. If the second install type is selected, 1
will be put in user_var, and so on. The value of
${NSIS_M AX_INST_TYPES} (32 by default) m eans that the user
selected a custom  set of sections (Sim ply selecting "Custom " in the drop-
down m enu is not enough to trigger this, the value is calculated by the
sections actually selected).

4.9.13.11 InstTypeSetText

inst_type_idx text

Sets the text of the specified InstType. If the text is em pty then the
InstType is rem oved. By using a previously unused inst_type_idx num ber
you can create new InstTypes. To add/rem ove Sections to this new
InstType see SectionSetInstTypes. Unlike SectionIn the index is zero
based, which m eans the first install type's index is 0.

InstType a
InstType b



Function .onInit
  # set first installation type's name to $WINDIR
  InstTypeSetText 0 $WINDIR
  # set second installation type's name to $TEMP
  InstTypeSetText 1 $TEMP
FunctionEnd

4.9.13.12 InstTypeG etText

inst_type_idx user_var

G ets the text of the specified InstType.

InstType a
InstType b

Function .onInit
  InstTypeGetText 0 $0
  DetailPrint $0 # prints 'a'
  InstTypeGetText 1 $0
  DetailPrint $0 # prints 'b'
FunctionEnd

4.9.14 User Interface Instructions

4.9.14.1 BringToFront

M akes the installer window visible and brings it to the top of the window
list. If an application was executed that shows itself in front of the
installer, BringToFront would bring the installer back in focus.

Recent W indows versions restrict the setting of foreground windows. If
the user is working with another application during installation, the user
m ay be notified using a different m ethod.



4.9.14.2 CreateFont

user_var(handle output) face_name [height] [weight] [/ITALIC] [/UNDERLINE] [/STRIKE]

Creates a font and puts its handle into user_var. For m ore inform ation
about the different param eters have a look at M SDN's page about the
W in32 API function CreateFont().

You can get the current font used by NSIS using the ̂ Font and ̂ FontSize
LangStrings.

!include WinMessages.nsh
GetDlgItem $0 $HWNDPARENT 1
CreateFont $1 "Times New Roman" "7" "700" /UNDERLINE
SendMessage $0 ${WM_SETFONT} $1 1

4.9.14.3 DetailPrint

user_message

Adds the string "user_m essage" to the details view of the installer.

DetailPrint "this message will be shown in the installation window"

4.9.14.4 EnableW indow

hwnd state(1|0)

Enables or disables m ouse and keyboard input to the specified window
or control. Possible states are 0 (disabled) or 1 (enabled).

GetDlgItem $0 $HWNDPARENT 1
EnableWindow $0 0
Sleep 1000
EnableWindow $0 1



4.9.14.5 FindW indow

user_var(hwnd output) windowclass [windowtitle] [windowparent] [childafter]

Searches for a window. Behaves like W in32's FindW indowEx(). Searches
by windowclass (and/or windowtitle if specified). If windowparent or
childafter are specified, the search will be restricted as such. If
windowclass or windowtitle is specified as "", they will not be used for the
search. If the window is not found the user variable is set to 0.

FindWindow $1 "#32770" "" $HWNDPARENT # Finds the inner dialog
FindWindow $2 "EDIT" "" $1 # Finds the first edit control in the inner dialog

4.9.14.6 G etDlgItem

user_var(output) dialog item_id

Retrieves the handle of a control identified by item _id in the specified
dialog box dialog. If you want to get the handle of a control in the inner
dialog, first use FindW indow to get the handle of the inner dialog.

GetDlgItem $0 $HWNDPARENT 1 # next/install button

4.9.14.7 HideW indow

Hides the installer window.

4.9.14.8 IsW indow

HWND jump_if_window [jump_if_not_window]

If HW ND is a window, G otos jum p_if_window, otherwise, G otos
jum p_if_not_window (if specified).

GetDlgItem $0 $HWNDPARENT 1
IsWindow $0 0 +3



  MessageBox MB_OK "found a window"
  Goto +2
  MessageBox MB_OK "no window"

4.9.14.9 LockW indow

on|off

LockW indow on prevents the m ain window from  redrawing itself upon
changes. W hen LockW indow off is used, all controls that weren't redrawn
since LockW indow on will be redrawn. This m akes the pages flickering
look nicer because now it flickers a group of controls at the sam e tim e,
instead of one control at a tim e. The individual control flickering is m ore
noticeable on old com puters.

4.9.14.10 SendM essage

HWND msg wparam lparam [user_var(return value)] [/TIMEOUT=time_in_ms]

Sends a m essage to HW ND. If a user variable $x is specified as the last
param eter (or one before the last if you use /TIM EO UT), the return value
from  SendM essage will be stored in it. Note that when specifying 'm sg'
you m ust just use the integer value of the m essage. Include
W inM essages.nsh to have all W indows m essages defined in your script.
If you wish to send strings use "STR:a string" as wParam  or lParam
where needed. Use /TIM EO UT=tim e_in_m s to specify the duration, in
m illiseconds, of the tim e-out period.

!include WinMessages.nsh
FindWindow $0 "Winamp v1.x"
SendMessage $0 ${WM_CLOSE} 0 0

GetDlgItem $1 $HWNDPARENT 2
SendMessage $1 ${WM_SETTEXT} 0 "STR:Goodbye"

4.9.14.11 SetAutoClose



true|false

O verrides the default auto window-closing flag (specified for the installer
using AutoCloseW indow, and false for the uninstaller). Specify 'true' to
have the install window im m ediately disappear after the install has
com pleted, or 'false' to m ake it require a m anual close.

4.9.14.12 SetBrandingIm age

[/IMGID=item_id_in_dialog] [/RESIZETOFIT] path_to_bitmap_file.bmp

Sets the current bitm ap file displayed as the branding im age. If no IM G ID
is specified, the first im age control found will be used, or the im age
control created by AddBrandingIm age. Note that this bitm ap m ust be
present on the user's m achine. Use File first to put it there. If
/RESIZETO FIT is specified the im age will be autom atically resized (very
poorly) to the im age control size. If you used AddBrandingIm age you can
get this size by com piling your script and watching for AddBrandingIm age
output, it will tell you the size. SetBrandingIm age will not work when
called from  .onInit!

4.9.14.13 SetDetailsView

show|hide

Shows or hides the details, depending on which param eter you pass.
O verrides the default details view, which is set via ShowInstDetails.

4.9.14.14 SetDetailsPrint

none|listonly|textonly|both|lastused

Sets m ode at which com m ands print their status. None has com m ands
be quiet, listonly has status text only added to the listbox, textonly has
status text only printed to the status bar, and both enables both (the
default). For extracting m any sm all files, textonly is recom m ended



(especially on win9x with sm ooth scrolling enabled).

SetDetailsPrint none
File "secret file.dat"
SetDetailsPrint both

4.9.14.15 SetCtlColors

hwnd [/BRANDING] [text_color|SYSCLR:text_color_id] [transparent|bg_color|SYSCLR:bg_color_id]

Sets the text and background color of a static control, edit control, button
or a dialog. text_color and bg_color don't accept variables. Use
G etDlgItem  to get the handle (HW ND) of the control. To m ake the control
transparent specify transparent as the background color value. Prefix the
color value with SYSCLR: to specify a W indows COLOR_* constant. You can
also specify /BRANDING with or without text color and background color to
m ake the control com pletely gray (or any other color you choose). This is
used by the branding text control in the M UI.

Page Components "" CmpntPageShow
Function CmpntPageShow
FindWindow $1 "#32770" "" $HWNDPARENT
GetDlgItem $0 $1 1006
SetCtlColors $0 0xFF0000 0x00FF00 ; Red on Green
GetDlgItem $0 $1 1022
SetCtlColors $0 SYSCLR:23 SYSCLR:24 ; COLOR_INFOTEXT on COLOR_INFOBK
FunctionEnd

W arning: Setting the background color of check boxes to transparent
m ay not function properly when using XPStyle on. The background m ay
be com pletely black instead of transparent when using certain W indows
them es.

4.9.14.16 SetSilent

silent | normal



Sets the installer to silent m ode or norm al m ode. See SilentInstall for
m ore inform ation about silent installations. Can only be used in .onInit.

4.9.14.17 Show W indow

hwnd show_state

Sets the visibility of a window. Possible show_states are the sam e as the
W indows ShowW indow function. SW _* constants are defined in
Include\W inM essages.nsh.

!include WinMessages.nsh
GetDlgItem $0 $HWNDPARENT 1
ShowWindow $0 ${SW_HIDE}
Sleep 1000
ShowWindow $0 ${SW_SHOW}

4.9.15 M ultiple Languages Instructions

4.9.15.1 LoadLanguageFile

language_file.nlf

Loads a language file for the construction of a language table. All of the
language files that ship with NSIS are in Contrib\Language Files

After you have inserted the language file ${LANG _langfile} will be defined
as the language id (for exam ple, ${LANG _ENG LISH} will be defined as
1033). Use it with LangString, LicenseLangString, LangDLL and
VIAddVersionKey.

4.9.15.2 LangString

name language_id|0 string

Defines a m ultilingual string. This m eans its value m ay be different (or



not, it's up to you) for every language. It allows you to easily m ake your
installer m ultilingual without the need to add m assive switches to the
script.

Each language string has a nam e that identifies it and a value for each
language used by the installer. They can be used in any runtim e string in
the script. To use a language string all you need to add to the string is
$(LangString_nam e_here) where you want the LangString to be inserted.

Notes:

Unlike defines that use curly braces - {}, language strings use
parenthesis - ().
If you change the language in the .onInit function, note that language
strings in .onInit will still use the detected language based on the
user's default W indows language because the language is initialized
after .onInit.
Always set language strings for every language in your script.
If you set the language ID to 0 the last used language by LangString
or LoadLanguageFile will be used.

Exam ple of usage:

 LangString message ${LANG_ENGLISH} "English message"
 LangString message ${LANG_FRENCH} "French message"
 LangString message ${LANG_KOREAN} "Korean message"

 MessageBox MB_OK "A translated message: $(message)"

4.9.15.3 LicenseLangString

name language_id|0 license_path

Does the sam e as LangString only it loads the string from  a text/RTF file
and defines a special LangString that can only be used by LicenseData.

LicenseLangString license ${LANG_ENGLISH} license-english.txt
LicenseLangString license ${LANG_FRENCH} license-french.txt



LicenseLangString license ${LANG_GERMAN} license-german.txt

LicenseData $(license)

4.10 M ultiple Languages
As of version 2 NSIS fully supports m ultiple languages. The interface of
one installer can support m ultiple languages.

Use LoadLanguageFile for every language to load the default interface
texts and language properties. Visit the NSIS translations forum  for m ore
inform ation about creating new language files.

The default interface texts can easily be changed using instructions like
Com ponentText etc.

You can also use the contents of the standard language strings in your
own strings (for exam ple, $(̂ N am e) contains the installer's nam e set
using the Nam e instruction). The nam es of all standard language strings
are listed as com m ents just above the strings in the language files. The
language files are located in Contrib\Language Files.

To create your own language strings, use LangString.

For an exam ple of an installer with m ultiple languages, see
languages.nsi.

4.10.1 Language Selection

W hen the installer starts up it goes through these steps to select the
interface language:

1. G et user's default W indows UI language
2. Find a perfect m atch for the language
3. If there is no perfect m atch, find a prim ary language m atch
4. If there is no m atch, use the first language defined in the script

(m ake sure your first language is a com m on one like English)
5. If the language variable $LANG UAG E has changed during .onInit,

NSIS goes through steps 2 to 4 again.

4.10.2 LangDLL Plug-in



The LangDLL plug-in allows you to give the user an option to choose the
language of the installer. Just push the language id (${LANG _langfile})
and its nam e for every language in your installer, then the num ber of
languages pushed, the caption, and the text that tells the user to select
the language, call the plug-in function nam ed LangDialog, pop the
returned value into $LANG UAG E and you're good to go. If the user clicks
on the cancel button the return value will be "cancel".

For an exam ple of usage see languages.nsi.

4.10.3 RTL Languages

RTL languages are languages that are written from  right to left (e.g.
Arabic and Hebrew). NSIS fully supports RTL languages. In the language
file there is a place to specify if the language is RTL or not. To find out at
runtim e if the current language is RTL or not, check the value of the
$(̂ RTL) language string. It will be 1 if the language is RTL and 0
otherwise. This can be useful when using plug-ins that create dialogs,
they usually have RTL settings too.

4.11 Plug-in DLLs
The abilities of the NSIS scripting language can be extended by utilising
functionality provided in a DLL file. Probably the best known exam ple of
this is the InstallO ptions.dll bundled with every NSIS release.

W hen the NSIS com piler starts it scans the plug-ins directory for DLLs
and m akes a list of the plug-ins found and their exported functions.
During com pilation, if a sequence such as fred::flintstone is encountered
where the com piler expected to find a language keyword the com piler will
look through this list. If a list entry specifies that fred.dll exports function
flintstone NSIS will pack the fred.dll file into the created installer binary.

During execution of a plug-in com m and NSIS will unpack the necessary
DLL to a tem porary folder ($PLUG INSDIR), push all of the argum ents
specified (right-to-left order), and then execute the DLL function.

4.11.1 Using Plug-in Com m ands

A plug-in call looks like this:

InstallOptions::dialog "ini_file_location.ini"

All param eters are pushed onto the stack (in this case, the plug-in
function only needs one param eter). Som e plug-in com m ands m ay not
need any param eters on the stack, others m ight require m ore of them . To
use a plug-in com m and you will need to read the docum entation for the
plug-in so that you know what param eters its functions require.

4.11.2 Calling plug-ins m anually

If you want to call a plug-in that is stored on user's hard drive or
som ewhere else, use CallInstDLL. Alm ost all plug-ins provide installer
functionality, so using plug-in com m ands is way easier. Using CallInstDLL
can be useful when you have created plug-ins that are linked to a certain
version of your application and are being copied to the installation folder.



Previous | Contents | Next



Previous | Contents | Next



4.12 Silent Installers/Uninstallers
Silent installers are installers which require no user intervention and have
no user interface. The user doesn't see any dialog and isn't asked any
questions. This is useful for network adm inistrators who wish to install or
uninstall som ething without user intervention so they can perform  the
operation quickly over any num ber of com puters. It is also useful for other
developers who wish to em bed another installer in their own and collect
all of the required inform ation on their installer instead of showing two
installers.

NSIS installers and uninstallers can be both silent and not silent. W hen
an installer or an uninstaller is silent, not all callback functions are called.
.onG UIInit, .onG UIEnd, their uninstaller equivalents and any callback
related to a specific page or page type will not be called.

There are several m ethods to m ake an installer or an uninstaller silent:

1. SilentInstall and SilentUninstall
2. SetSilent
3. Passing /S on the com m and line (case sensitive)

To check if the installer/uninstaller is silent use IfSilent.

To m ake sure your installer will be silent when it needs to, you should
check with IfSilent before each com m and that m ight require user
intervention or create a window. The M essageBox com m and, which is
the m ost com m on culprit in silent installers, has the /SD switch to set a
default answer for silent installers. If you want your installer/uninstaller to
be able to be com pletely silent you should use this switch. All internal
NSIS m essage boxes have defaults for silent installers. The silent.nsi
exam ple dem onstrates all aspects of this topic.

Since the directory page is not shown in silent installers the user has an
option to specify the installation directory on the com m and line (this also
works on non-silent installers/uninstallers). To do that, the user uses the
/D switch as in the following exam ple:



foo.exe /S /D=C:\Program Files\Foo

If your installer/uninstaller requires som e m ore inform ation that can not
be gathered when silent, you can allow the user to specify that
inform ation on the com m and line and process it in .onInit. You can use
G etO ptions.

!include FileFunc.nsh
!insertmacro GetParameters
!insertmacro GetOptions

Function .onInit
  ${GetParameters} $R0
  ClearErrors
  ${GetOptions} $R0 /USERNAME= $0
FunctionEnd

The above exam ple will copy the value the user passes on after
/USERNAM E= into $0. This allows the user to specify the required
inform ation on the com m and line instead of using the interactive user
interface. The user can use:

foo.exe /S /USERNAME=Bar /D=C:\Program Files\Foo

or:

foo.exe /S /USERNAME=string with spaces /D=C:\Program Files\Foo

or:

foo.exe /S /USERNAME="string with spaces" /D=C:\Program Files\Foo

If your installer/uninstaller requires a lot of inform ation and you want it to
be able to be silent, you should allow the user to pass on a path to an
answers file. This would be m uch m ore com fortable than writing all of the
inform ation on the com m and line.

5.1 Com piler Utility Com m ands
These com m ands are sim ilar to the C preprocessor in term s of purpose
and functionality. They allow file inclusion, conditional com pilation,
executable header packing and process execution during the build
process. Note: None of these com m ands allow the use of variables.

Num ber literals support the 0b, 0o, 0n and 0x radix prefixes (base 2, 8, 10
and 16 respectively). Note: The deprecated plain 0 octal prefix is also
supported in som e places but its usage is discouraged.

5.1.1 !include

[/NONFATAL] [/CHARSET=ACP|OEM|CP#|UTF8|UTF16LE|UTF16BE] file

This com m and will include 'file' as if it was part of the original script. Note
that if a file is included in another directory, the current directory is still
where the script was com piled from  (not where the included file resides).
If the com piler can't find the file it will look for it in every include directory.
See !addincludedir for m ore inform ation. If the /nonfatal switch is used
and no files are found, a warning will be issued instead of an error.
/charset can be used to specify a codepage for plain text files without a
BO M .

!include WinMessages.nsh
!include Library.nsh
!include /CHARSET=CP1252 C:\MyConfig.nsi
!include ..\MyConfig.nsh
!include /NONFATAL file_that_may_exist_or_not.nsh

5.1.2 !addincludedir

directory

Adds another include directory to the include directories list. This list is
searched when !include is used. This list's initial value is
${NSISDIR}\Include.



!addincludedir ..\include
!include something.nsh

5.1.3 !addplugindir

[/x86-ansi | /x86-unicode] directory

Causes the NSIS com piler to scan the given directory for plug-in DLLs. If
you don't specify the plug-in architecture it is assum ed to m atch the
current target architecture. If the architecture does not m atch the installer
will probably crash!

!addplugindir ..\myplugin
MyPlugin::SomeFunction

5.1.4 !appendfile

[/CHARSET=ACP|OEM|CP#|UTF8[SIG]|UTF16<LE|BE>[BOM]] [/RawNL] file text

Appends text to file. The text is written as ANSI (ACP) unless the file
already has a BO M . Using /CHARSET will force a specific character
encoding. $\n will be translated to $\r$\n on W indows unless you specify
/RawNL.

!tempfile FILE
!appendfile "${FILE}" "XPStyle on$\n"
!appendfile "${FILE}" "Name 'test'$\n"
!include "${FILE}"
!delfile "${FILE}"
!undef FILE

5.1.5 !cd

new_path

This com m and will change the com piler to the new directory, new_path.
new_path can be relative or absolute.



!cd ..\more-scripts\new

5.1.6 !delfile

[/nonfatal] file

This com m and deletes a file.

!tempfile FILE
!delfile "${FILE}"
!undef FILE

5.1.7 !echo

message

This com m and will echo a m essage to the user com piling the script.

!echo "hello world"

5.1.8 !error

[message]

This com m and will issue an error to the script com piler and will stop
execution of the script. You can also add a m essage to this error.

!ifdef VERSION & NOVERSION
  !error "both VERSION and NOVERSION are defined"
!endif

5.1.9 !execute

command [compare comparevalue | symbol]

This com m and will execute 'com m and' using a call to CreateProcess().
Unlike !system , it does not use the com m and line processor, so
input/output redirection and com m ands like 'cd', 'dir' and 'type' can not be



used. Currently, the only known advantage of !execute over !system  is
that it does not give trouble when the current working directory is
specified using UNC.

O n PO SIX platform s, !execute will use system () just like !system .

!execute '"$%WINDIR%\notepad.exe" /P "${NSISDIR}\COPYING"'

5.1.10 !m akensis

parameters [compare comparevalue | symbol]

This com m and will !execute a new instance of M akeNSIS with the
param eters you specify.

!makensis '-DGENERATEUNINST "${__FILE__}"' = 0
!system '"signtool" sign ...' = 0

5.1.11 !packhdr

tempfile command

This option m akes the com piler use an external EXE packer (such as
Petite or UPX) to com press the executable header. Specify a tem porary
file nam e (such as "tem p.dat") and a com m and line (such as "C:\program
files\upx\upx -9 tem p.dat") to com press the header.

!packhdr "$%TEMP%\exehead.tmp" '"C:\Program Files\UPX\upx.exe" "$%TEMP%\exehead.tmp"'

5.1.12 !finalize

command [compare comparevalue]

This option will execute 'com m and' using a call to system () after the
output EXE has been generated. You can typically use it to sign
(Authenticode) your installer. If 'com m and' contains a '% 1' it will be
replaced by the executables filenam e.



!finalize 'sign.bat "%1" "Product Installer" http://example.com'

5.1.13 !system

command [compare comparevalue | symbol]

This com m and will execute 'com m and' using a call to system (). You can
store the return value in a define ('sym bol') or halt execution if the return
value com pared (using 'com pare') to 'com parevalue' is false. 'com pare'
can be '<' or '>' or '<>' or '='.

!system '"%WINDIR%\notepad.exe" "${NSISDIR}\COPYING"'
!system 'echo !define something > newinclude.nsh'
!include newinclude.nsh
!ifdef something
  !echo "something is defined"
!endif

5.1.14 !tem pfile

symbol

This com m and creates a tem porary file. It puts its path into a define,
nam ed sym bol.

!tempfile PACKHDRTEMP
!packhdr "${PACKHDRTEMP}" '"C:\Program Files\UPX\upx.exe" "${PACKHDRTEMP}"'

!tempfile FILE
!define /date DATE "%H:%M:%S %d %b, %Y"
!system 'echo built on ${DATE} > "${FILE}"'
!undef DATE
File /oname=build.txt "${FILE}"
!delfile "${FILE}"
!undef FILE

5.1.15 !getdllversion



localfilename define_basename

This is sim ilar to G etDLLVersionLocal, only it stores the version num ber
in defines and can therefore be used anywhere, not just inside functions
and sections.

!getdllversion "$%WINDIR%\Explorer.exe" expv_
!echo "Explorer.exe version is ${expv_1}.${expv_2}.${expv_3}.${expv_4}"

5.1.16 !w arning

[message]

This com m and will issue a warning to the script com piler. You can also
add a m essage to this warning.

!ifdef USE_DANGEROUS_STUFF
  !warning "using dangerous stuff"
!endif

5.1.17 !pragm a

warning <enable|disable|default> code
warning <push|pop>

The pragm a com m ands allows you to change com piler features and
behavior.

!pragma warning disable 9000 ; Disable warning about using "Setup.exe" as the name
OutFile "Setup.exe"

5.1.18 !verbose

level | push | pop

This com m and will set the level of verbosity. 4=all, 3=no script, 2=no info,
1=no warnings, 0=none.



Passing push will cause !verbose to push the current verbosity level on a
special stack. Passing pop will cause !verbose to pop the current
verbosity level from  the sam e stack and use it.

!verbose push
!verbose 1
!include WinMessages.nsh
!verbose pop

5.2 Predefines
You can use these standard predefines to autom atically add the build
tim e to the title of developm ent versions, add the date to the version
num ber, etc.

5.2.1 ${__CO UNTER__}

Expands to a num ber (Starting at 0 and increm enting by 1 every tim e it is
used)

5.2.2 ${__FILE__}

Current script nam e.

5.2.3 ${__FILEDIR__}

Current script directory.

5.2.4 ${__LINE__}

Current line num ber.

5.2.5 ${__DATE__}

Date when the script started com piling according to the current locale.

5.2.6 ${__TIM E__}

Tim e when the script started com piling according to the current locale.

5.2.7 ${__TIM ESTAM P__}

Date & tim e of the last m odification to the script file according to the
current locale.



5.2.8 ${NSIS_VERSIO N}

NSIS version used to build the script.

5.2.9 ${NSIS_PACKEDVERSIO N}

NSIS version as a 32-bit num ber.

!if 0x3014000 >= "${NSIS_PACKEDVERSION}"
  !error "NSIS 3.15 or higher is required to build this installer!"
!endif

5.2.10 ${NSIS_CHAR_SIZE}

The size of a character code unit (in bytes). 1 in ANSI installers and 2 in
Unicode installers.

A graphem e cluster consists of a base character plus optional com bining
characters and diacritics and is defined as one or m ore code points. O ne
or m ore code units is required to encode a single code point.

5.2.11 ${NSIS_PTR_SIZE}

The size of a pointer (in bytes) in the generated installer.

5.2.12 ${U+1}...${U+10FFFF}

A Unicode (UCS-4) character.

DetailPrint "${U+2115}SIS" # DOUBLE-STRUCK CAPITAL N + "SIS"

5.2.13 Scope Predefines

Standard predefines that contain inform ation about the current code
scope.

5.2.13.1 ${__G LO BAL__}



Defined in the global scope.

Section test
  !ifdef ${__GLOBAL__}
    !error "this shouldn't be here!"
  !endif
SectionEnd

PageEx instfiles
  !ifdef ${__GLOBAL__}
    !error "this shouldn't be here!"
  !endif
PageExEnd

5.2.13.2 ${__SECTIO N__}

Defined as the section nam e, without any prefixes, in section scope.

!ifdef __SECTION__
  !error "this shouldn't be here!"
!endif

Section test
  !ifndef __SECTION__
    !error "missing predefine!"
  !endif

  !if ${__SECTION__} != test
    !error "wrong predefine value!"
  !endif
SectionEnd

Section !test
  !if ${__SECTION__} != test
    !error "wrong predefine value!"
  !endif
SectionEnd



Section un.test
  !if ${__SECTION__} != test
    !error "wrong predefine value!"
  !endif
SectionEnd

5.2.13.3 ${__FUNCTIO N__}

Defined as the function nam e, without any prefixes, in function scope.

!ifdef __FUNCTION__
  !error "this shouldn't be here!"
!endif

Function test
  !ifndef __FUNCTION__
    !error "missing predefine!"
  !endif

  !if ${__FUNCTION__} != test
    !error "wrong predefine value!"
  !endif
FunctionEnd

Function un.test
  !if ${__FUNCTION__} != test
    !error "wrong predefine value!"
  !endif
FunctionEnd

5.2.13.4 ${__PAG EEX__}

Defined as the page type in PageEx scope.

!ifdef __PAGEEX__



  !error "this shouldn't be here!"
!endif

PageEx instfiles
  !ifndef __PAGEEX__
    !error "missing predefine!"
  !endif

  !if ${__PAGEEX__} != instfiles
    !error "wrong page type"
  !endif
PageExEnd

5.2.13.5 ${__UNINSTALL__}

Defined in section, function or PageEx scopes of the uninstaller.

!ifdef __UNINSTALL__
  !error "this shouldn't be here!"
!endif

Function test
  !ifdef __UNINSTALL__
    !error "this shouldn't be here!"
  !endif
FunctionEnd

Function un.test
  !ifndef __UNINSTALL__
    !error "missing predefine!"
  !endif
FunctionEnd

5.2.13.6 ${__M ACRO __}

Defined as the nam e of the current m acro.

5.3 Read environm ent variables

5.3.1 $% envVarNam e%

$% envVarNam e%  will be replaced at com pile tim e by the environm ent
variable envVarNam e.

5.4 Conditional Com pilation
The com piler m aintains a list of defined sym bols, which can be defined
using !define or the /D com m and line switch. These defined sym bols can
be used for conditional com pilation (using !ifdef) or for sym bol
replacem ent (a sim ple form  of m acros). To replace a sym bol with its
value, use ${SYM BO L} (if SYM BO L is not defined, no translation will
occur). The translation is first-com e-first-served, m eaning if you do:

!define symbol_one ${symbol_two}

If sym bol_two is defined when that line occurs, it will be replaced.
O therwise, any replacing will occur when ${sym bol_one} is referenced.

Define/conditional com pilation related com m ands:

5.4.1 !define

[/ifndef | /redef] ([/date|/utcdate] gflag [value]) | (/math gflag val1 OP val2) | (/file gflag filename.txt)

This com m and will add gflag to the global define list. This will have a
sim ilar effect as using the /D switch on the com m and line (the define only
becom es effective after the !define com m and).

If /date or /utcdate are used, value will be passed to strftim e() and the
result will be used as the value of gflag. strftim e converts special sym bols
into certain parts of the current tim e or date. For exam ple, % H will be
converted into the current hour in 24-hour form at. For a com plete list of
available sym bols, search for strftim e on M SDN. O n PO SIX, you can get
the list by using man strftime.

If /m ath is used, the result of 'val1 O P val2', where O P m ay be +,-
,*,&,|,̂ ,/,<<,>>,>>> or %  , will be used as the value of gflag. Note that
val1 AND val2 M UST be integer values!

If /file is used, the entire text file specified (including whitespace and
newlines) will be read and stuffed into gflag.



!define USE_SOMETHING
!define VERSION 1.2
!define /date NOW "%H:%M:%S %d %b, %Y"
!define /math RESULT 3 + 10
!define /math REST 15 % ${RESULT}
!define /file BunchaStuff somesourcefile.cpp
!define /redef USE_SOMETHING ${RESULT} ;redefine USE_SOMETHING

5.4.2 !undef

gflag

Rem oves an item  from  the global define list. Note that ${SYM BO L} where
SYM BO L is undefined will be translated to "${SYM BO L}".

!define SOMETHING
!undef SOMETHING

5.4.3 !ifdef

gflag [bcheck gflag [...]]

This com m and, when paired with an !endif com m and, will tell the
com piler whether or not to com pile the lines in between the two lines. If
gflag is globally defined (using !define or the /D switch), then the
contained lines will be com piled. O therwise, they will be skipped. 'bcheck'
can be specified as & (boolean and) or | (boolean or) along with m ore
gflags -- precedence is sim ple, left to right.

!define SOMETHING
!ifdef SOMETHING
  !echo "SOMETHING is defined"
!endif
!undef SOMETHING
!ifdef SOMETHING
  !echo "SOMETHING is defined" # will never be printed
!endif



5.4.4 !ifndef

gflag [bcheck gflag [...]]]

The opposite of !ifdef. The lines will be com piled when the gflag has not
been defined.

5.4.5 !if

[!] value [op value2]
[!] /FileExists "c:\path\file.exe"

This com m and, when paired with an !endif com m and, will tell the
com piler whether or not to com pile the lines in between the two lines. If
value is non-zero, or the com parison of value and value2 depending on
the operator results in true, the contained lines will be com piled.
O therwise, they will be skipped. op can be either == or != (case-
insensitive string com parison), S== or S!= (case-sensitive string
com parison), =, <>, <=, <, > or >= (int/hex/float com parison), & (bitwise
AND com parison), && or || (boolean com parison). If [!] is set, the result
will be flipped from  true to false and vice versa.

!if 1 < 0x2
  !echo "1 is smaller than 2!!"
!else if ! 3.1 > 1.99
  !error "this line should never appear"
!else
  !error "neither should this"
!endif

!if /FileExists ".\cert.pfx"
  !finalize '".\sign.bat" "%1"'
!endif

5.4.6 !ifm acrodef

gflag [bcheck gflag [...]]]



This com m and, when paired with an !endif com m and, will tell the
com piler whether or not to com pile the lines in between the two lines. If
the m acro gflag exists, then the contained lines will be com piled.
O therwise, they will be skipped. 'bcheck' can be specified as & (boolean
and) or | (boolean or) along with m ore gflags -- precedence is sim ple, left
to right.

!macro SomeMacro
!macroend
!ifmacrodef SomeMacro
  !echo "SomeMacro is defined"
!endif

5.4.7 !ifm acrondef

gflag [bcheck gflag [...]]]

The opposite of !ifm acrodef. The lines will be com piled when the m acro
gflag does not exist.

5.4.8 !else

[if|ifdef|ifndef|ifmacrodef|ifmacrondef [...]]

This com m and allows to easily insert different code when different
defines or m acros are set. You can create blocks like !ifdef/!else/!endif,
!ifdef/!else ifdef/!else/!endif etc.

!ifdef VERSION
OutFile installer-${VERSION}.exe
!else
OutFile installer.exe
!endif

5.4.9 !endif

This com m and closes a block started with !if, !ifdef, !ifndef, !ifm acrodef or
!ifm acrondef.



5.4.10 !insertm acro

macro_name [parameter] [...]

Inserts the contents of a m acro that was created with !m acro. If the
m acro was created with param eters, then you m ust pass as m any
param eters to the m acro as it requires.

!macro Print text
  DetailPrint "${text}"
!macroend
!insertmacro Print "some text"
!insertmacro Print "some more text"

5.4.11 !m acro

macro_name [parameter][...]

Creates a m acro nam ed 'm acro_nam e'. All lines between the !m acro and
the !m acroend will be saved. To insert the m acro later on, use
!insertm acro. !m acro definitions can have one or m ore param eters
defined. The param eters m ay be accessed the sam e way a !define would
(e.g. ${PARM NAM E}) from  inside the m acro.

!macro SomeMacro parm1 parm2 parm3
  DetailPrint "${parm1}"
  MessageBox MB_OK "${parm2}"
  File "${parm3}"
!macroend

5.4.12 !m acroend

Ends a m acro that was started with !m acro.

5.4.13 !m acroundef

macro_name



Deletes a m acro.

5.4.14 !searchparse

[/ignorecase] [/noerrors] [/file] source_string_or_file substring_start OUTPUTSYMBOL1 [substring [OUTPUTSYMBOL2 [substring ...]]]

Parses source_string_or_file (which is treated as a string, or as a
filenam e if /file is set), looking for substring_start. If substring_start is
found, then O UTPUTSYM BO L1 is defined to the rest of the string (m inus
any other substring that m ay be found). Any num ber of
O UTPUTSYM BO Lx m ay be specified, and the final substring is optional.

If /noerrors is specified, m atching less than the full num ber of strings is
allowed (all O UTPUTSYM BO Lx after the not-found substring will be
ignored).

If /file is specified, the file is treated as a series of lines. The file is
searched until all substrings are m atched. If /noerrors is specified and not
all strings are m atched, the first line with the m ost sym bols m atched is
used.

# search filename.cpp for a line '#define APP_VERSION "2.5"' and set ${VER_MAJOR} to 2, ${VER_MINOR} to 5.
!searchparse /file filename.cpp `#define APP_VERSION "` VER_MAJOR `.` VER_MINOR `"`

5.4.15 !searchreplace

[/ignorecase] symbol_out source_string searchfor replacewith

Searches source_string, looking for searchfor and replacing all instances
of it with replacewith. Unlike !define, !searchreplace allows you to
redefine sym bol_out without warning or error.

# defines ${blah} to "i like ponies"
!searchreplace blah "i love ponies" "love" "like"

Appendix A: M odern User Interface
NSIS 2 m akes it is possible to create installers with a custom  user
interface. The M odern UI is a interface with a style like the wizards of
recent W indows versions. This new interface also features new pages
(W elcom e, Finish, Start M enu) and a description area on the com ponents
page. The interface and the graphics can be custom ized using the
provided settings. Using the M odern UI m acros and language files,
writing scripts with a m odern interface is easy.

For m ore inform ation and docum entation see the M odern UI 2 Readm e.

NSIS 2.34 brought with it a new version of M odern UI - version 2. It is
faster and m ore extendible. It allows plug-ins to add new types of pages
and even change existing pages using a sim ple NSH file. It also uses
nsDialogs which faster than its elder sibling - InstallO ptions.

For m ore inform ation and docum entation of the old version see the
M odern UI Readm e.

Appendix B: DLL/TLB Library Setup

Introduction
Param eters
O ptions

Exam ple

Library Uninstallation

Param eters
O ptions
Exam ple

Visual Basic 6 Run-Tim e Files

B.1 Introduction
The Library header file can be used to setup dynam ic link libraries (DLL)
and type libraries (TLB). If necessary, the following actions will be
perform ed:

File copying
File copying on reboot
Version checks
Registration and unregistration
Registration and unregistration on reboot
Shared DLL counting
W indows File Protection checks

The m acros are stored in the header file Library.nsh, which should be
included in scripts using this system :

!include Library.nsh

Note that the library m acros are lim ited on non-W indows platform s. DLL
version inform ation is required when com piling on non-W indows
platform s.

B.2 Library Installation

B.2.1 Introduction

The InstallLib m acro allows you to install a library. It sets the error flag if
som ething went wrong during library setup.

To ask the user for a reboot, if required, use the M odern UI with a Finish
page or use IfRebootFlag and m ake your own page or m essage box.

B.2.2 Param eters

libtype shared install localfile destfile tempbasedir

libtype

The type of the library

DLL - Dynam ic link library (DLL)
REG DLL - DLL that has to be registered
REG EXE - EXE CO M  server that has to be registered using /regserver
TLB - Type library or DLL that contains a type library
REG DLLTLB - DLL that has to be registered and contains a type library

shared

Specify whether the library is shared with other applications

NO TSHARED - The library is not shared
$VARNAM E - Variable that is em pty when the application is installed for
the first tim e, which is when the shared library count will be increased.

install

Specify the installation m ethod

REBO O T_PRO TECTED



Upgrade the library on reboot when in use (required for system  files).
Upgrade the library if the file is not protected by W indows File
Protection.

NO REBO O T_PRO TECTED

W arns the user when the library is in use. The user will have to close
applications using the library.
Upgrade the library if the file is not protected by W indows File
Protection.

REBO O T_NO TPRO TECTED

Upgrade the library on reboot when in use (required for system  files).
Upgrade the library without checking for W indows File Protection.

NO REBO O T_NO TPRO TECTED

W arns the user when the library is in use. The user will have to close
applications using the library.
Upgrade the library without checking for W indows File Protection.

localfile

Location of the library on the com piler system

destfile

Location to store the library on the user's system

tem pbasedir

Directory on the user's system  to store a tem porary file when the system
has to be rebooted.

For W indows 9x/M E support, this directory should be on the sam e
volum e as the destination file (destfile). The W indows tem p directory
could be located on any volum e, so you cannot use this directory.

B.2.3 O ptions



Define any of the following before inserting a InstallLib m acro to m odify
its behavior as specified.

B.2.3.1 LIBRARY_X64

Installs a DLL built for W indows x64.
W arning: This resets file system  redirection.

B.2.3.2 LIBRARY_SHELL_EXTENSIO N

Define this before inserting InstallLib m acro to call SHChangeNotify
with SHCNE_ASSO CCHANG ED after registration.
Use this to refresh the shell when installing a shell extension or when
changing file associations.

B.2.3.3 LIBRARY_CO M

Define this before inserting InstallLib m acro to call
CoFreeUnusedLibraries after registration.
Use this for unloading all unnecessary libraries from  m em ory when
installing CO M  libraries.

B.2.3.4 LIBRARY_IG NO RE_VERSIO N

Define this before inserting InstallLib m acro to ignore version
inform ation in the file and always install it, even if it already exists.
Use this when an older or specific version is required.
Not recom m ended for DLLs installed to $SYSDIR.

B.2.4 Notes

If you need to support W indows 9x/M E, you can only use short
filenam es (8.3).
W arning: Always use redistributable files when deploying DLLs,
never copy files from  your system  directory!



B.2.5 Exam ple

B.2.5.1 Unshared DLL

 !insertmacro InstallLib REGDLL NOTSHARED REBOOT_NOTPROTECTED dllname.dll $SYSDIR\dllname.dll $SYSDIR

B.2.5.2 Shared DLL

 ;Add code here that sets $ALREADY_INSTALLED to a non-zero value if the application is
 ;already installed. For example:

 IfFileExists "$INSTDIR\MyApp.exe" 0 new_installation ;Replace MyApp.exe with your application filename
   StrCpy $ALREADY_INSTALLED 1
 new_installation:

 !insertmacro InstallLib REGDLL $ALREADY_INSTALLED REBOOT_NOTPROTECTED dllname.dll $SYSDIR\dllname.dll $SYSDIR

B.3 Library Uninstallation

B.3.1 Introduction

The UnInstallLib m acro allows you to uninstall a library. It sets the error
flag if som ething went wrong during library rem oval.

B.3.2 Param eters

libtype shared uninstall file

libtype

The type of the library

DLL - Dynam ic link library (DLL)
REG DLL - DLL that has to be unregistered
REG EXE - EXE CO M  server that has to be unregistered using
/unregserver
TLB - Type library or DLL that contains a type library
REG DLLTLB - DLL that has to be unregistered and contains a type
library

shared

Specify whether the library is shared with other applications

NO TSHARED - The library is not shared
SHARED - The library is shared and should be rem oved if the shared
library count indicates that the file is not in use anym ore..

uninstall

Specify the uninstallation m ethod

NO REM O VE

The library should not be rem oved. You should use this option for



com m on or im portant system  files such as the Visual
Basic/C++/M FC runtim es.

REBO O T_PRO TECTED

Rem ove the library on reboot when in use (required for system  files).
Rem ove the library if the file is not protected by W indows File
Protection.

NO REBO O T_PRO TECTED

W arns the user when the library is in use. The user will have to close
applications using the library.
Rem ove the library if the file is not protected by W indows File
Protection.

REBO O T_NO TPRO TECTED

Rem ove the library on reboot when in use (required for system  files).
Rem ove the library without checking for W indows File Protection.

NO REBO O T_NO TPRO TECTED

W arns the user when the library is in use. The user will have to close
applications using the library.
Rem ove the library without checking for W indows File Protection.

file

Location of the library

B.3.3 O ptions

Define any of the following before inserting a UnInstallLib m acro to
m odify its behavior as specified.

B.3.3.1 LIBRARY_X64

Uninstalls a DLL built for W indows x64.



W arning: This resets SetRegView and file system  redirection.

B.3.3.2 LIBRARY_SHELL_EXTENSIO N

Define this before inserting UninstallLib m acro to call
SHChangeNotify with SHCNE_ASSO CCHANG ED after
unregistration. Use this to refresh the shell when uninstalling a shell
extension or when changing file associations.

B.3.3.3 LIBRARY_CO M

Define this before inserting UninstallLib m acro to call
CoFreeUnusedLibraries after unregistration. Use this for unloading
all unnecessary libraries from  m em ory when uninstalling CO M
libraries.

B.3.4 Exam ple

 !insertmacro UnInstallLib REGDLL SHARED REBOOT_NOTPROTECTED $SYSDIR\dllname.dll

B.4 Visual Basic 6 Run-Tim e Files
A new VB6RunTim e.nsh header file is available for the setup of the VB6
run-tim e files. To obtain the latest run-tim e files, download vb6runtim e.zip
and extract this file.

 !include VB6RunTime.nsh

 Var AlreadyInstalled

 Section "-Install VB6 run-time files"

   ;Add code here that sets $AlreadyInstalled to a non-zero value if the application is already installed. For example:
   IfFileExists "$INSTDIR\MyApp.exe" 0 new_installation ;Replace MyApp.exe with your application filename
     StrCpy $AlreadyInstalled 1
   new_installation:

   !insertmacro VB6RunTimeInstall C:\vb6runtimes $AlreadyInstalled ;Replace C:\vb6runtimes with the location of the files

 SectionEnd

 Section "-un.Uninstall VB6 run-time files"

   !insertmacro VB6RunTimeUnInstall

 SectionEnd

Rem arks:

You m ay have to install additional files for such Visual Basic
application to work, such as O CX files for user interface controls.
Installation of the run-tim e files requires Adm inistrator or Power User
privileges. Use the M ulti-User header file to verify whether these
privileges are available.
Add a M odern UI finish page or another check (see IfRebootFlag) to
allow the user to restart the com puter when necessary.

Appendix C: Useful Scripts
G et Internet Explorer version
Is .NET Fram ework installed?
Is M acrom edia Flash Player installed?
Connect to the Internet
G et Installer Filenam e
Prevent M ultiple Instances
M ore

C.1 G et Internet Explorer version

 ; GetIEVersion
 ;
 ; Based on Yazno's function, http://yazno.tripod.com/powerpimpit/
 ; Returns 1-6 (IE Version) or '' (IE is not installed) on top of the stack
 ;
 ; Usage:
 ;   Call GetIEVersion
 ;   Pop $R0 ; at this point $R0 is "5" or whatnot

 Function GetIEVersion
 Push $R0
   ClearErrors
   ReadRegStr $R0 HKLM "Software\Microsoft\Internet Explorer" "Version"
   IfErrors lbl_123 lbl_456

   lbl_456: ; ie 4+
     Strcpy $R0 $R0 1
   Goto lbl_done

   lbl_123: ; older ie version
     ClearErrors
     ReadRegStr $R0 HKLM "Software\Microsoft\Internet Explorer" "IVer"
     IfErrors lbl_error

       StrCpy $R0 $R0 3
       StrCmp $R0 '100' lbl_ie1
       StrCmp $R0 '101' lbl_ie2
       StrCmp $R0 '102' lbl_ie2

       StrCpy $R0 '3' ; default to ie3 if not 100, 101, or 102.
       Goto lbl_done
         lbl_ie1:
           StrCpy $R0 '1'
         Goto lbl_done



         lbl_ie2:
           StrCpy $R0 '2'
         Goto lbl_done
     lbl_error:
       StrCpy $R0 ''
   lbl_done:
   Exch $R0
 FunctionEnd

C.2 Is .NET Fram ew ork installed?

 ; IsDotNETInstalled
 ;
 ; Based on GetDotNETVersion
 ;   http://nsis.sourceforge.net/Get_.NET_Version
 ;
 ; Usage:
 ;   Call IsDotNETInstalled
 ;   Pop $0
 ;   StrCmp $0 1 found_dotNETFramework no_dotNETFramework

 Function IsDotNETInstalled
   Push $0
   Push $1

   StrCpy $0 1
   System::Call "mscoree::GetCORVersion(w, i ${NSIS_MAX_STRLEN}, *i) i .r1"
   StrCmp $1 0 +2
     StrCpy $0 0

   Pop $1
   Exch $0
 FunctionEnd

C.3 Is M acrom edia Flash Player installed?

 ; IsFlashInstalled
 ;
 ; By Yazno, http://yazno.tripod.com/powerpimpit/
 ; Returns the result on top of the stack
 ;
 ; Usage:
 ;   Call IsFlashInstalled
 ;   Pop $R0 ; $R0 is "1" or "0" at this point 

 Function IsFlashInstalled
  Push $R0
  ClearErrors
  ReadRegStr $R0 HKCR "CLSID\{D27CDB6E-AE6D-11cf-96B8-444553540000}" ""
  IfErrors lbl_na
    StrCpy $R0 1
  Goto lbl_end
  lbl_na:
    StrCpy $R0 0
  lbl_end:
  Exch $R0
 FunctionEnd

C.4 Connect to the Internet

 ; ConnectInternet (uses Dialer plug-in)
 ; Written by Joost Verburg 
 ;
 ; This function attempts to make a connection to the internet if there is no
 ; connection available. If you are not sure that a system using the installer
 ; has an active internet connection, call this function before downloading
 ; files with NSISdl.
 ; 
 ; The function requires Internet Explorer 3, but asks to connect manually if
 ; IE3 is not installed.
 
 Function ConnectInternet
 
   Push $R0
     
     ClearErrors
     Dialer::AttemptConnect
     IfErrors noie3
     
     Pop $R0
     StrCmp $R0 "online" connected
       MessageBox MB_OK|MB_ICONSTOP "Cannot connect to the internet."
       Quit ;This will quit the installer. You might want to add your own error handling.
     
     noie3:
   
     ; IE3 not installed
     MessageBox MB_OK|MB_ICONINFORMATION "Please connect to the internet now."
     
     connected:
   
   Pop $R0
   
 FunctionEnd

C.5 G et Installer Filenam e

 System::Call 'kernel32::GetModuleFileName(p 0, t .R0, i ${NSIS_MAX_STRLEN}) i.r1'
 ;$R0 will contain the installer filename

C.6 Prevent M ultiple Instances
Put the following code in your .onInit function:

 System::Call 'kernel32::CreateMutex(p 0, i 0, t "myMutex") p .r1 ?e'
 Pop $R0
 
 StrCmp $R0 0 +3
   MessageBox MB_OK|MB_ICONEXCLAMATION "The installer is already running."
   Abort

'm yM utex' m ust be replaced by a unique value!

C.7 M ore
You can find m ore useful scripts on the NSIS W iki, the NSIS forum  and
the NSIS developm ent page.

Appendix D: Useful Inform ation
Error Levels
Add uninstall inform ation to Add/Rem ove Program s
Calling an external DLL using the System .dll plug-in
Dum p Content of Log W indow to File
How to Read REG _M ULTI_SZ Values
Predefined M acros for Unicode support

D.1 Error Levels
Like other applications, installers m ade by NSIS return error levels as a
result of their execution. Checking the error level can be useful if you call
an NSIS installer from  another application or installer.

0 - Norm al execution (no error)
1 - Installation aborted by user (cancel button)
2 - Installation aborted by script

You can set the error level to other values using SetErrorLevel.

Note that uninstallers copy them selves to the tem porary directory and
execute from  there so the original uninstaller can be deleted. This m eans
the error level the uninstaller sets is not available to the executing
process, unless it sim ulates this copy process and executes the copied
uninstaller. To sim ulate this process, use:

InitPluginsDir
CopyFiles $INSTDIR\uninstaller.exe $PLUGINSDIR
ExecWait '"$PLUGINSDIR\uninstaller.exe" _?=$INSTDIR' $0
DetailPrint "uninstaller set error level $0"

If you don't do this, you'll only be able to know if the uninstaller failed
copying itself to the tem porary directory.

D.2 Add uninstall inform ation to Add/Rem ove
Program s
Create a key with your product nam e under
HKLM \Software\M icrosoft\W indows\CurrentVersion\Uninstall to add
entries to the "Add/Rem ove Program s" section in the Control Panel. For
W indows NT (NT4/2000/XP), it's also possible to create the key in the
HKCU hive, so it will only appear for the current user. There are several
values you can write to key to give inform ation about your application and
the uninstaller. W rite a value using the W riteRegStr com m and (for
strings) or W riteRegDW O RD com m and (for DW O RD values).

Exam ple:

WriteRegStr HKLM "Software\Microsoft\Windows\CurrentVersion\Uninstall\MyProduct" "DisplayName" "Application Name"
WriteRegStr HKLM "Software\Microsoft\Windows\CurrentVersion\Uninstall\MyProduct" "UninstallString" '"$INSTDIR\uninst.exe"'

Required values

DisplayNam e (string) - Nam e of the application 
UninstallString (string) - Path and filenam e of the uninstaller. You should
alw ays quote the path.

O ptional values

Som e of the following values will not be used by older W indows versions.

InstallLocation (string) - Installation directory ($INSTDIR) 
DisplayIcon (string) - Path, filenam e and index of the icon that will be
displayed next to your application nam e

Publisher (string) - (Com pany) nam e of the publisher

M odifyPath (string) - Path and filenam e of the application m odify program
InstallSource (string) - Location where the application was installed from

ProductID (string) - Product ID of the application 
RegO wner (string) - Registered owner of the application 



RegCom pany (string) - Registered com pany of the application

HelpLink (string) - Link to the support website 
HelpTelephone (string) - Telephone num ber for support

URLUpdateInfo (string) - Link to the website for application updates 
URLInfoAbout (string) - Link to the application hom e page

DisplayVersion (string) - Displayed version of the application 
VersionM ajor (DW O RD) - M ajor version num ber of the application 
VersionM inor (DW O RD) - M inor version num ber of the application

NoM odify (DW O RD) - 1 if uninstaller has no option to m odify the installed
application 
NoRepair (DW O RD) - 1 if the uninstaller has no option to repair the
installation

If both NoM odify and NoRepair are set to 1, the button displays
"Rem ove" instead of "M odify/Rem ove".

D.3 Calling an external DLL using the System .dll
plug-in
Som e installers need to call functions in third-party DLLs. A prim e
exam ple of this is when installing a Palm (TM ) conduit.

Som e background about System .dll The System .dll plug-in enables
calling of external DLLs by using its 'Call' function. There are a num ber of
other functions provided by System .dll but they will not be covered here.
For m ore details about the other functions, lock the doors, take the phone
off the hook, screw your head on *real* tight and head on over to the
System  readm e.

Data Types 
System .dll recognises the following data types:

v - void (generally for return)
p - pointer (includes void*, HANDLE, HW ND, UINT_PTR and so on)
i - int (a 32bit integer)
l - large integer (also known as int64)
t - text, string (LPTSTR, pointer to first character)
k - callback. See Callback section in system .htm l.
* - pointer specifier -> the proc needs the pointer to type, affects next
char (param eter) [ex: '*i' - pointer to int]

M apping System .dll variables to NSIS script variables 
There's not m uch point in being able to call an external function if you
can't get any data back. System .dll m aps function variables to NSIS
script variables in the following way:

NSIS $0..$9 becom es System .dll r0..r9 NSIS $R0..$R9 becom es
System .dll r10..r19

Each param eter is specified by type, input and output. To skip input or
output use a dot. Exam ples:

String (pointer to a character array), input is 'happy calling':



t 'happy calling'

String (pointer to a character array), input is taken from  $5 and changes
to the array m ade by the callee are saved into $R8:

t r5R8

Pointer to an integer, value taken from  $1 and put into $2:

*i r1r2

Pointer to a 64-bit integer, output pushed on stack, no input:

*l .s

Using System .dll::Call To call a function in a third party DLL, the Call
function is used like this:

System::Call 'YourDllName::YourDllFunction(i, *i, t) i(r0, .r1, r2) .r3'

The '(r0, .r1, r2) .r3' section at the end are the param eters that are
passed between your DLL and your NSIS script. As can be seen in this
param eters list, type and input/output can be separated. Each block of "
(parm s list) return value" overrides and/or adds to the last one. In this
case, the first block specifies the types and the second specifies input
and output.

Before starting to code the NSIS script 
Before you start to code any NSIS code you need to know the full
prototype of the function you are going to call. For the purposes of this
exam ple, we will use the 'Cm G etHotSyncExecPath' function from  the
Palm  'CondM gr.dll'. This function is used to return the full path of
'HotSync.exe'.

Function Definition 

int __stdcall Cm G etHotSyncExecPath(TCHAR *pPath, int *piSize);

where



pPath is a pointer to a character buffer. Upon return, this is the path
& file nam e of the installed HotSync m anager.
piSize is a pointer to an integer that specifies the size (in TCHAR's),
of the buffer referenced by the pPath param eter.

return values:

0: No error
-1: A non-specific error occurred
ERR_REG ISTRY_ACCESS(-1006):Unable to access the Palm
configuration entries
ERR_BUFFER_TO O _SM ALL(-1010): The buffer is too sm all to hold
the requested inform ation
ERR_INVALID_PO INTER(-1013):The specified pointer is not a valid
pointer

Also, if the buffer is too sm all the value in *int is the size (in TCHARs) that
the buffer should be.

This function definition m aps to the following System .dll definition:

Cm G etHotSyncExecPath(t, *i) i

i.e. It takes a text variable, a pointer to int, and returns an int value.

Using the external dll function 
Now that we've sorted out what the function does and how it m aps to the
System .dll form at we can use the function in a NSIS script.

First you have to change the output directory to that where the DLL you
want to use is. It m ay also work if the DLL is in the system  path but this
hasn't been tested.

The following code fragm ent will install 'condm gr.dll' to a tem porary
directory, execute the Cm G etHotSyncExecPath function and display
returned data. Save this script

; **** snip ****
Function loadDll



  SetOutPath $TEMP\eInspect             ; create temp directory
  File bin\CondMgr.dll                  ; copy dll there
  StrCpy $1 ${NSIS_MAX_STRLEN}          ; assign memory to $0
  System::Call 'CondMgr::CmGetHotSyncExecPath(t, *i) i(.r0, r1r1).r2'
  DetailPrint 'Path: "$0"'
  DetailPrint "Path length: $1"
  DetailPrint "Return value: $2"

FunctionEnd
; **** snip ****

and this function produces the following output in the 'details' page:

O utput folder: c:\windows\TEM P\eInspect 
Extract: CondM gr.dll 
Path: "C:\Dave\palm \Hotsync.exe" 
Path length: 24 
Return value: 0

W ritten by djc

Acknowledgem ents & Thanks 
Lots of thanks go to kichik and Sunjam m er for spending a lot of tim e
assisting in solving this problem . Also to brainsucker for creating the
System .dll plug-in in the first place. G ood Luck!

D.4 Dum p Content of Log W indow  to File
This function will dum p the log of the installer (installer details) to a file of
your choice.

To use it, push a file nam e and call it. It will dum p the log to the file
specified. For exam ple:

GetTempFileName $0
Push $0
Call DumpLog

Here is the function:

!define LVM_GETITEMCOUNT 0x1004
!define LVM_GETITEMTEXTA 0x102D

Function DumpLog # Written by KiCHiK
  Exch $5
  Push $0
  Push $1
  Push $2
  Push $3
  Push $4
  Push $6

  FindWindow $0 "#32770" "" $HWNDPARENT
  GetDlgItem $0 $0 1016
  StrCmp $0 0 error
  FileOpen $5 $5 "w"
  StrCmp $5 0 error
    SendMessage $0 ${LVM_GETITEMCOUNT} 0 0 $6
    System::StrAlloc ${NSIS_MAX_STRLEN}
    Pop $3
    StrCpy $2 0
    System::Call "*(i, i, i, i, i, i, i, i, i) p \



      (0, 0, 0, 0, 0, r3, ${NSIS_MAX_STRLEN}) .r1"
    loop: StrCmp $2 $6 done
      System::Call "User32::SendMessageA(p, i, p, p) i \
        ($0, ${LVM_GETITEMTEXTA}, $2, r1)"
      System::Call "*$3(&t${NSIS_MAX_STRLEN} .r4)"
      FileWrite $5 "$4$\r$\n"
      IntOp $2 $2 + 1
      Goto loop
    done:
      FileClose $5
      System::Free $1
      System::Free $3
      Goto exit
  error:
    MessageBox MB_OK error
  exit:
    Pop $6
    Pop $4
    Pop $3
    Pop $2
    Pop $1
    Pop $0
    Exch $5
FunctionEnd

Here's the function to generate a UTF-16LE file if you're building a
Unicode installer.

!define LVM_GETITEMCOUNT 0x1004
!define LVM_GETITEMTEXTW 0x1073

Function DumpLog # Written by KiCHiK, modified by Jim Park
  Exch $5
  Push $0
  Push $1
  Push $2
  Push $3



  Push $4
  Push $6

  FindWindow $0 "#32770" "" $HWNDPARENT
  GetDlgItem $0 $0 1016
  StrCmp $0 0 error
  FileOpen $5 $5 "w"
  FileWriteWord $5 0xfeff ; Write the BOM
  StrCmp $5 0 error
    SendMessage $0 ${LVM_GETITEMCOUNT} 0 0 $6
    System::StrAlloc ${NSIS_MAX_STRLEN}
    Pop $3
    StrCpy $2 0
    System::Call "*(i, i, i, i, i, i, i, i, i) p \
      (0, 0, 0, 0, 0, r3, ${NSIS_MAX_STRLEN}) .r1"
    loop: StrCmp $2 $6 done
      System::Call "User32::SendMessageW(p, i, p, p) i \
        ($0, ${LVM_GETITEMTEXTW}, $2, r1)"
      System::Call "*$3(&t${NSIS_MAX_STRLEN} .r4)"
      FileWriteUTF16LE $5 "$4$\r$\n"
      IntOp $2 $2 + 1
      Goto loop
    done:
      FileClose $5
      System::Free $1
      System::Free $3
      Goto exit
  error:
    MessageBox MB_OK error
  exit:
    Pop $6
    Pop $4
    Pop $3
    Pop $2
    Pop $1
    Pop $0
    Exch $5



FunctionEnd

Previous | Contents | Next



Previous | Contents | Next



D.5 How  to Read REG _M ULTI_SZ Values
KiCHiK wrote this script to help rpetges in this forum  thread. It reads a
registry value of the type REG _M ULTI_SZ and prints it out. Don't forget
to edit where it says "Edit this!" when you test this script. The values m ust
point to a REG _M ULTI_SZ value or the exam ple will spit out an error.

OutFile "REG_MULTI_SZ Reader.exe"
Name "REG_MULTI_SZ Reader"
ShowInstDetails show

!define HKEY_CLASSES_ROOT        0x80000000
!define HKEY_CURRENT_USER        0x80000001
!define HKEY_LOCAL_MACHINE       0x80000002
!define HKEY_USERS               0x80000003
!define HKEY_PERFORMANCE_DATA    0x80000004
!define HKEY_PERFORMANCE_TEXT    0x80000050
!define HKEY_PERFORMANCE_NLSTEXT 0x80000060
!define HKEY_CURRENT_CONFIG      0x80000005
!define HKEY_DYN_DATA            0x80000006
!define KEY_QUERY_VALUE          0x0001
!define KEY_ENUMERATE_SUB_KEYS   0x0008
!define REG_NONE                 0
!define REG_SZ                   1
!define REG_EXPAND_SZ            2
!define REG_BINARY               3
!define REG_DWORD                4
!define REG_DWORD_LITTLE_ENDIAN  4
!define REG_DWORD_BIG_ENDIAN     5
!define REG_LINK                 6
!define REG_MULTI_SZ             7

!define RegOpenKeyEx     "Advapi32::RegOpenKeyExA(i, t, i, i, *i) i"
!define RegQueryValueEx  "Advapi32::RegQueryValueExA(i, t, i, *i, i, *i) i"
!define RegCloseKey      "Advapi32::RegCloseKeyA(i) i"



####### Edit this!

!define ROOT_KEY         ${HKEY_CURRENT_USER}
!define SUB_KEY          "Software\Joe Software"
!define VALUE            "Strings"

####### Stop editing

Section "Read"
  StrCpy $0 ""
  StrCpy $1 ""
  StrCpy $2 ""
  StrCpy $3 ""
  System::Call "${RegOpenKeyEx}(${ROOT_KEY}, '${SUB_KEY}', \
    0, ${KEY_QUERY_VALUE}|${KEY_ENUMERATE_SUB_KEYS}, .r0) .r3"
 
  StrCmp $3 0 goon
    MessageBox MB_OK|MB_ICONSTOP "Can't open registry key! ($3)"
    Goto done
goon:

  System::Call "${RegQueryValueEx}(r0, '${VALUE}', 0, .r1, 0, .r2) .r3"
 
  StrCmp $3 0 read
    MessageBox MB_OK|MB_ICONSTOP "Can't query registry value size! ($3)"
    Goto done
 
read:
 
  StrCmp $1 ${REG_MULTI_SZ} multisz
    MessageBox MB_OK|MB_ICONSTOP "Registry value no REG_MULTI_SZ! ($3)"
    Goto done
 
multisz:
 
  StrCmp $2 0 0 multiszalloc
    MessageBox MB_OK|MB_ICONSTOP "Registry value empty! ($3)"



    Goto done
 
multiszalloc:

  System::Alloc $2
  Pop $1
 
  StrCmp $1 0 0 multiszget
    MessageBox MB_OK|MB_ICONSTOP "Can't allocate enough memory! ($3)"
    Goto done
 
multiszget:
 
  System::Call "${RegQueryValueEx}(r0, '${VALUE}', 0, n, r1, r2) .r3"
 
  StrCmp $3 0 multiszprocess
    MessageBox MB_OK|MB_ICONSTOP "Can't query registry value data! ($3)"
    Goto done
 
multiszprocess:
 
  StrCpy $4 $1
 
  loop:
 
    System::Call "*$4(&t${NSIS_MAX_STRLEN} .r3)"
    StrCmp $3 "" done
    DetailPrint $3
    StrLen $5 $3
    IntOp $4 $4 + $5
    IntOp $4 $4 + 1
    Goto loop
 
done:
 
  System::Free $1
 



  StrCmp $0 0 noClose
    System::Call "${RegCloseKey}(r0)"
 
noClose:

SectionEnd

D.6 Predefined M acros for Unicode support
There are two m acros that can help you write scripts that work for both
Unicode and ANSI installers. To figure out if the script is being com piled
to generate a Unicode installer, use !ifdef to check for
${NSIS_UNICO DE}. To see what the size of a character is, use
${NSIS_CHAR_SIZE}. It will be 1 for ANSI and 2 for Unicode installers.

Appendix E: Useful Headers
File Functions Header

Introduction
Locate
G etSize
DriveSpace
G etDrives
G etTim e
G etFileAttributes
G etFileVersion
G etExeNam e
G etExePath
G etParam eters
G etO ptions
G etO ptionsS
G etRoot
G etParent
G etFileNam e
G etBaseNam e
G etFileExt
BannerTrim Path
DirState
RefreshShellIcons

Text Functions Header
Introduction
LineFind
LineRead
FileReadFrom End
LineSum
FileJoin
TextCom pare
TextCom pareS
ConfigRead
ConfigReadS
ConfigW rite
ConfigW riteS
FileRecode



Trim NewLines
W ord Functions Header

Introduction
W ordFind
W ordFindS
W ordFind2X
W ordFind2XS
W ordFind3X
W ordFind3XS
W ordReplace
W ordReplaceS
W ordAdd
W ordAddS
W ordInsert
W ordInsertS
StrFilter
StrFilterS
VersionCom pare
VersionConvert

E.1 File Functions Header

E.1.1 Introduction

Include header:

!include "FileFunc.nsh"

Call functions:

Section Install
     ${GetFileExt} "C:\My Downloads\Index.html" $R0
     ; $R0="html"
SectionEnd

Section un.Install
     ${GetParent} "C:\My Downloads\Index.html" $R0
     ; $R0="C:\My Downloads"
SectionEnd

E.1.2 Locate

Find files, directories and em pty directories with m ask and size
options.

Syntax:

${Locate} "[Path]" "[Options]" "Function"

"[Path]"      ; Disk or Directory
              ;
"[Options]"   ; /L=[FD|F|D|DE|FDE]
              ;     /L=FD    - Locate Files and Directories (default)
              ;     /L=F     - Locate Files only
              ;     /L=D     - Locate Directories only
              ;     /L=DE    - Locate Empty Directories only
              ;     /L=FDE   - Locate Files and Empty Directories


